
Forord
Bogen er en introduktion til computeralgebraprogrammet Maple. Eksemplerne er gennemregnet i Maple 13, men det 
meste vil også kunne bruges i tidligere og (formoder vi) senere versioner.
Det anbefales, at man først læser kapitel 1, der indeholder det allermest nødvendige. 
Kapitlerne 1-9 indeholder det væsentligste af det man får brug for på det indledende matematikkursus på civil- eller 
diplom-ingeniørstudiet.


Maple er et overmåde omfattende system, så selv hvor omtalen synes lang, kunne den have været meget længere. Man 
må selv skaffe sig flere detaljer ved flittig brug af programmets egen hjælp. Det er meget langt fra, at alle de 
tilgængelige procedurer og kommandoer nævnes. 


Bogen begyndte som noter skrevet af Chr. Pommer 1994 i Wordperfect for DOS-versionen af MapleV Release 2. De 
blev derefter af mig omskrevet i Maple først til Release 3, senere til Release 4, 5, 6, 7, 8, 9, 9.5, 11 og nu 13. De talrige 
omskrivninger har efterhånden ændret det meste og mangedoblet sideantallet. 


Over årene fra Release 2 til Release 13 er mange ting forbedret, nogle ting ændret og nye ting kommet til. Der er bl.a. 
sket store ændringer (forbedringer) i interface. 


Med Maple 9 og senere følger en grafisk brugerflade (GUI) skrevet i Java. Denne er mere computerressourcekrævende 
end den klassiske brugerflade, der også følger med Mapleinstallationen. Fordelen ved Java-GUI'en skulle være, at den 
virker ens på forskellige operativsystemer (Windows, MacIntosh, Linux, Unix). Da vi i denne bog næsten ikke kommer 
ind på detaljer angående den grafiske brugerflade, er det for det meste uden betydning, om man har valgt at bruge den 
ene GUI frem for den anden.
Det bør nævnes, at man helt kan undgå en grafisk brugerflade ved at bruge Command Line Maple, der ligeledes følger 
med Mapleinstallationen.


Vi kommer slet ikke ind på tekstbehandlingsdelen af Maple, formelpaletter, regneark, smartplot og filhåndtering . Vi 
opfordrer læseren til selv at eksperimentere sig frem. Disse ting er også GUI-afhængige.


Disse noter er først skrevet som et antal Maple-worksheets i Windowsversionen af Java-Worksheet-Maple 13. De er 
derefter enkeltvis eksporteret som PDF-filer.


Der er skrevet mange bøger om Maple. Fortegnelser findes på web-adressen: http://www.maplesoft.com/books/


Manualerne fra Maplesoft til Maple 13 kan hentes gratis på adressen:
http://www.maplesoft.com/documentation%5Fcenter/


På Internettet kan man finde adskilligt om Maple.  Specielt bør der gøres opmærksom på
-Nyhedsgruppen comp.soft-sys.math.maple 


-Maple's egne sider findes på http://www.maplesoft.com/


Ethvert computeralgebraprogram indeholder fejl. Maple er ingen undtagelse. At finde fejl er ingen kunst. I bogen er  
medtaget eksempler, hvor det kan konstateres, at Maple begår en fejl.  Maple er et værktøj, der bør bruges med 
omtanke. Hvis liv, velfærd eller renommé afhænger af resultatet, så kontrollér det! Maple kan i stort omfang bruges ved 
kontrollen. I sidste ende er det brugeren, der er ansvarlig for resultatet.


Juni 2009             Preben Alsholm            


Institut for Matematik, Danmarks Tekniske Universitet







Vejledning i brug af Maple 13 sammen med Maplebogen


Hvis man bruger Classic Worksheet Maple 13, skal man ikke foretage sig noget. Classic Worksheet Maple har den 
gamle brugerflade, der er udmærket, men som ikke længere opdateres. Det er dog eksakt det samme, der kører bag 
skærmen. 


Bruger man derimod Maple 13 med den Java-baserede brugerflade (betegnes kun med Maple 13 ), så gør man følgende: 


• Gå ind i 
1. Tools/Options/Display/Input display 
2. Vælg Maple Notation. 
3. Klik på Apply Globally. 


• Gå ind i 
1. Tools/Options/Interface/Default format for new worksheets. 
2. Vælg Worksheet. 
3. Klik på Apply Globally. 


• Ønsker man desuden at undgå nummering af output gør man også følgende: 
1. Gå ind i Tools/Options/Display/Show equation labels. 
2. Fravælg. 
3. Klik på Apply Globally. 


Til allersidst åbner man et nyt worksheet. Dette vil nu være af det rigtige format.

1. Indledning


Dette kapitel giver en hurtig første introduktion til Maple. Det anbefales, at man læser hele kapitlet 
fra ende til anden.
Mange emner vil blive taget op i større detaljer i senere kapitler.


Anbring vha. piletasterne cursoren (en blinkende lodret streg) på en input-linie, dvs. en linie med 
symbolet  >  yderst til venstre. Skriv en ordre til Maple (som f.eks. 2+2; ). Tryk derefter på returtasten
(ENTER). Maple udfører da ordren (kommandoen) på den pågældende linie, hvis denne slutter med 
semikolon. 
Cursoren kan anbringes hvor som helst på linien. 
Cursoren  går automatisk videre til næste input-linie. Den hopper altså over linier, der kun indeholder
forklarende tekst. 


2+2;
4


Enhver kommando skal afsluttes med et semikolon (;) eller med et kolon (:). Bruges kolon vil 
kommandoen blive udført, men ikke vist:


(17-2)/5:


At der alligevel skete noget, ses ved at bede Maple om det sidstudregnede resultat.
I Maple refererer et procenttegn, %, til det sidstudregnede resultat. To procenttegn, %%, refererer til 
det næstsidste og %%% til det trediesidste:


%; 
3


%%%;
4


Det er meget bekvemt at kunne bruge % for det sidstudregnede resultat, men hopper man rundt på 
skærmen med musen eller piletasterne, så er sidstudregnede resultat ikke nødvendigvis resultatet i 
linien ovenover. Det anbefales derfor, at brugen af % minimeres. Giv udtrykkene navne.


Et udtryk kan gemmes i en variabel (gives et navn) ved brug af tilordningssymbolet, der er 
sammensat af et kolon efterfulgt af et lighedstegn:
 


abc:=14*x^3+7/sin(x*y);


Herefter kan abc bruges i stedet for  . Eksempelvis har vi


abc/7;

Gangetegnet * kan aldrig udelades i input. I output vises gangetegnet derimod ikke. 
Maple skelner mellem store og små bogstaver. Bemærk resultatet af følgende input bestående af en 
følge af tre udtryk:


ABC,aBC,abc;


Der må gerne stå flere kommandoer på samme linie:


expand( (a+b)^2 );   factor( 2*x^3+7*x-6*x^2-21 );


ligesom en kommando godt må fylde flere linier. Her bør ved linieskift trykke Shift+Enter.


expand(


(a+b)^3


);


Glemmer man semikolon (eller skal kommandoen fylde flere linier), sker følgende:


2+5


Warning, inserted missing semicolon at end of statement
7


Ønsker man, at en kommando skal fylde flere linier, og vil man undgå advarslen, kan man som sagt 
ved linieskift trykke Shift+Enter. 


Lad os løse ligningen  ved hjælp af Maple. Vi vælger først at gemme ligningen i en 


variabel, som vi her kalder ligning:


ligning:=12/(x-1)-x=2;


Hermed får vi at se, om ligningen er tastet rigtigt ind. Havde vi glemt parenteserne, ville ligningen se 


således ud i output:  . 


For at løse ligningen eksakt bruges proceduren solve:


solve(ligning,x);

Der er to løsninger. Disse er adskilt ved et komma. Man har en følge af løsninger. En udtryksfølge 
(expression sequence) i Maple er en række udtryk adskilt ved kommaer. 
Kommandoen solve( ligning, x);  vil vi sige er et kald til proceduren solve.
Ethvert kald af en procedure består af et procedurenavn (som her solve) efterfulgt af et sæt parenteser,
indenfor hvilke procedurens argument(-er) anbringes. 
Den nyttige kommando restart skal ikke have parenteser. Kommandoen restart;  får Maple til at 
glemme alt:


restart;


abc,ligning;


Som det ses, har Maple glemt tilordningerne til abc og ligning.
Hvis alt synes at gå i skuddermudder, så prøv kommandoen restart, eller i værste fald: Gå ud af 
Maple og ind igen.


Tegnet # efter (semi-)kolon kan anbringes, hvis en kommentar ønskes vedføjet kommandoen:


f:=sin(Pi*x)*exp(-x); # Her defineres udtrykket f


Det er normalt bedre at have forklarende tekst skrevet over eller under kommandoen, som det jo også
er gjort i disse noter, der er skrevet i Maple. En inputlinie kan (i Windows) omdannes til et tekstfelt 
ved at holde tasten Ctrl nede og så trykke t. 
En ny input-linie kan indsættes over eller under en given linie ved at holde Ctrl-tasten nede og så 
taste henholdsvis k eller j. Analogt fjernes en hel linie ved tastekombinationen Ctrl+Delete. 


Vi tegner grafen af f på intervallet [0, 4], som i Maple skrives 0..4 (mindst 2 punktummer):


plot(f,x=0..4);


x


1 2 3 4
0


Bemærk ved definitionen af udtrykket f  ovenfor, at der blev skrevet sin(x), ikke sin x. I Maple skal 
alle funktionskald og (som nævnt) procedurekald have parenteser. Bemærk resultatet af følgende 


input, hvor vi forsøger at finde :

sin Pi;


Error, missing operator or `;`


Prøver man det selv, vil man se, at cursoren står og blinker ved Pi. Der er en syntaksfejl. Et navn kan 
ikke indeholde mellemrum, så sin Pi kan ikke tolkes som et navn. Maple ser dermed to navne adskilt 
af et mellemrum, nemlig sin og Pi. Den ser ingen instruktion om hvad den skal gøre ved disse. Maple
konkluderer, at der mangler noget. Prøver man uden mellemrum, fås


sinPi;
sinPi


Her opfatter Maple blot bogstavkombinationen sinPi som et navn, som iøvrigt ikke af os eller af den 


er blevet tillagt nogen værdi. Vil man finde , skal man skrive


sin(Pi);
0


Mellemrum kan uden ødelæggende virkning indsættes flere steder. I et procedurenavn eller en fast 
konstruktion af tegn kan der dog ikke indsættes mellemrum:


sin     (    Pi     /  2 ); 
1


Dette går ikke:


s i n ( Pi/2 ); 


Error, missing operator or `;`


Vi forsøger at sætte q til 7:


q: =7;


Error, `=` unexpected


Mellemrummet mellem kolonnet og lighedstegnet må ikke være der. Maple ser ovenfor to 
kommandoer på samme linie, nemlig  q:  og   =7;  Den første er lidt uinteressant, men helt iorden. 
Den anden giver ingen mening, derfor fås den viste fejlmelding. 


Som sædvanligt i matematik bruges almindelige buede parenteser ( ) til omkredsning af et udtryk, der
ønskes beregnet før kontakt med omverdenen uden for parentesen:
  


a*(b+c);


a*(8+7^2);


I output vises gangetegnet ikke. Det gør man normalt heller ikke, når man ellers skriver matematik. 
Udelader man gangetegnet i input, dvs. skriver

a(b+c);


så bemærker man en forskel i udseende mellem de to output. I sidste tilfælde opfattes bogstavet a 
som navnet på en funktion eller en procedure, der bliver anvendt på summen b+c. At denne funktion
a så er ukendt, har den effekt, at Maple returnerer inputtet uevalueret. 
I Maple har man valgt at følge sædvanlig matematisk parentesbrug og må så acceptere, at gangetegn 
aldrig kan udelades. I computeralgebraprogrammet Mathematica har man valgt at bruge kantede 
parenteser ved funktionsanvendelse. Eksempelvis skrives i Mathematica  Sin[ Pi/2 ], når man vil 


have . I Mathematica kan gangetegn dermed oftest udelades. 


I Maple bruges kantede parenteser bl.a. ved definitionen af en liste. 
Tuborgparenteser { } bruges som mængdeklammer ganske som i sædvanlig matematisk skrivemåde. 
Sædvanlige parenteser kan ikke erstattes af [ ] eller { }. Har man mange parenteser indeni hinanden, 
kan man altså ikke udskifte nogle med [ ] eller { }, men man kan indsætte mellemrum eller endog 
linieskift:


a*(    (  b + (c-Pi/2)^2)^3  + 4  )^2;


Differentiation af udtrykket f som vi definerede ovenfor:


f;


sker ved brug af diff:


diff( f, x);


Integration sker ved brug af int. Ubestemt integration (stamfunktionsbestemmelse) fås ved 
kommandoen


int( f,x);


Ved differentiation af dette udtryk skulle vi gerne få f tilbage:


diff(%,x); 


Resultatet ligner ikke f, men vi kan simplificere det:

simplify(%);


Det bestemte integral  findes således


int(f, x=0..4);


En decimalbrøkstilnærmelse hertil fås ved proceduren evalf :


evalf(%);
0.2837317959


Flere tilordninger kan foretages på én gang. Nye tilordninger overskriver gamle. 


På den følgende inputlinie får f tillagt værdien , og  g får tillagt værdien  


. Hermed er den gamle f-værdi glemt:


f,g:=exp(x)-1-sin(x),1-cos(4*x);


Vil man nu finde grænseværdien for udtrykket    for , kan man gøre således:


limit( f/g, x=0);
1


16


Hjælp til en procedure (eksempelvis plot ), hvis navn man allerede har skrevet på en input-linie, kan 
fås ved at anbringe cursoren på procedurenavnet og derefter trykke på F2-tasten. Man kan også gå 
ind i programmets Hjælp-menu. Eller man kan prøve følgende:


?plot


Det er her en forudsætning, at man kender procedurens navn og stavemåde. Proceduren factor 
faktoriserer udtryk. Staver man proceduren med k, sker følgende:


faktor( 2*x^3+7*x-6*x^2-21 );


Maple returnerer uevalueret, simpelthen fordi den intet kendskab har til proceduren faktor. Den 
brokker sig ikke, men giver blot et kønt ekko. Man får intet ud af  ?faktor. Tast i stedet Ctrl+F1, 
eller gå ind i hjælpemenuen, vælg Maple Help. I boksen øverst til venstre skrives så det man leder 







Kapitel 1: Indledning


Kapitel 1 side 7


efter. Skriver man faktor, svares No matches found, men så kan man evt. prøve med kun fa, så 
kommer en lang liste af ord, der begynder med bogstaverne fa, bl.a. factor. Nu har man jo nok fundet
ud af, at man blot stavede forkert, men hjælpesiden for factor kommer frem ved at klikke på factor. 


Den bedste forståelse af en procedures brug fås i eksemplerne, der altid er anbragt nederst på 
hjælpesiden. Disse eksempler kan hentes ind i ens eget worksheet på sædvanlig måde (copy and 
paste) eller ved i hjælpesiden at gå ind i menuen Edit og der vælge Copy Examples. Eksemplerne kan 
så anbringes i ens eget worksheet ved tastetrykket Ctrl v.


I Maple kan programmeres. Der kan skrives løkker og betingede sætninger. 


Som et eksempel på en løkke beder vi Maple gange udtrykket  ud for n = 2, 3, 4, 5, 6. 
 


for n from 2 to 6 do expand((a+b)^n) end do; 


Har man et specielt formål, hvortil en egnet Maple-procedure ikke allerede findes, kan man selv 
skrive en. Her har man selvfølgelig alle Maples egne procedurer til rådighed. Vi vil i et senere kapitel
komme nærmere ind herpå. 
Imidlertid er i Maple en funktion også en procedure. 


Funktionen f givet ved forskriften 


for alle x > 0, defineres i Maple på følgende måde:
 


f:=x->x^2*ln(x); 


Pilen er sammensat af et minus fulgt af ulighedstegnet >. Herefter kan vi bede om eksempelvis f(1) 
og f(2):


f(1);
0


f(2);


eller blot om f(t):


f(t);


En procedures argumenter evalueres normalt fuldt ud før proceduren laver sit arbejde. Derfor går 
følgende ikke godt:

x:=7:


int(x^4/(1+x^2), x=0..1);


Error, (in int) integration range or variable must be specified in 


the second argument, got 7 = 0 .. 1


Problemet er, at x er sat til 7, så Maple læser kommandoen som int( 7^4/(1+7^2), 7=0..1);  
Problemet kan løses ved at rense x for tilordningen:


x:='x'; 


hvor der bruges apostroffer (ikke accenter). Herved sættes variablen  x  tilbage til sin oprindelige 
værdi, som er dens eget navn.
Herefter finder vi


int(x^4/(1+x^2), x=0..1);


Men vil man have Maple til at glemme alt, bruger man som sagt kommandoen restart:
 


restart;

2. Differentiation


Til differentiation af algebraiske udtryk benyttes diff. Til differentiation af funktioner (procedurer) 
benyttes differentialoperatoren D.


Differentialoperatoren diff


Vil man differentiere det algebraiske udtryk  med hensyn til , gør man således


 


diff(x^2*sin(a*x),x);


Vi giver næste udtryk et navn, så vi kan se, om det er tastet rigtigt ind, inden det skal 
differentieres:


udtryk:=x^7*sin(x)+3*x/(1+x);


diff( udtryk, x); 


Skal udtrykket differentieres tre gange, skrives


diff( udtryk, x,x,x);


eller lidt kortere 


diff( udtryk, x$3);


idet x$3 blot er en følge af tre x'er:  x, x, x:


hurra$9;


x'erne kan også, om man vil anbringes i listeklammer:


diff(udtryk,[x,x,x]);

Denne syntaks tillader den tomme liste, der giver den nul'te afledede, altså funktionen selv:


diff(udtryk,[]);


Proceduren diff tillader differentiation af et ubekendt udtryk y(x) i variablen x. Naturligvis fås 
ingen egentlig udregning, men Maple skriver udtrykket pænt:


diff(y(x), x);


Bemærk, at kommandoen  diff( y, x);  giver nul, når y ikke er sat til noget. Dette skyldes, at Maple
ikke kan se noget x, hvorfor y antages ikke at afhænge heraf.
Vi er nu i stand til at formulere en differentialligning i den ubekendte y(x):


ligning1:=diff(y(x),x,x)-3*diff(y(x),x)+2*y(x)=x*exp(x);


I kapitel 6 behandler vi løsning af differentialligninger.


En inaktiv version af diff er Diff:


Diff(x*sin(x),x);


Bemærk, at d'erne er grå, når der bruges Diff. 
Den inaktive Diff kan omdannes til den aktive diff ved brug af value:


value(%); 


En elegant version fås som følger. Bemærk, at der er to kommandoer på samme linie. Resultatet 
fra den første vises ikke, men bruges i form af  %  i den næste:


Diff(sin(x*exp(x)),x):  %=value(%);


Differentialoperatoren D


Differentialoperatoren D benyttes ved differentiation af funktioner:

D(sin);
cos


D(ln);


Bemærk, at Maple her ikke kunne svare med et navn på funktionen  , da et sådant ikke 


findes. 
Lad funktionen f være defineret ved


f:=x -> x^2*sin(3*x);


så fås den afledede af f, altså f ' , ved kommandoen


D( f );


Bemærk, at D som input accepterer en funktion (eksempelvis f eller sin) og som output også 
leverer en funktion (f ' eller cos). 
Output fra D kan så netop bruges som en funktion bruges:


D(sin)(Pi/6);


D(f)(Pi);


Den 5. afledede af funktionen f fås som følger, idet snabel a-tegnet @ bruges ved sammensætning 
af funktioner og operatorer:


(D@D@D@D@D)(f); 


Det er dog lettere at skrive således:


(D@@5)(f); 


Den nul'te afledede (funktionen selv) kommer frem således:


(D@@0)(f);
f


(D@@0)(f)(x);


Udtrykket  skal vi ikke tænke på som en funktion, men kun som et udtryk 


indeholdende variablen x eller som værdien af funktionen f  i  x.
Vi finder til sammenligning den femte afledede af udtrykket

f(x);


Dette gøres ved hjælp af diff:


diff(f(x),x$5);


Samme resultat kunne opnås således


(D@@5)(f)(x);


idet (D@@5)(f) giver os den femte afledede af f. Herefter anvendes den femte afledede så på x.


Vi kan benytte operatoren D på en ubekendt funktion som f.eks. y:


D(y);


Dette er nyttigt, når differentialligninger med begyndelsesbetingelser skal løses. 
Begyndelsesbetingelserne y(7) = 4,  y'(7) = -2,  skrives i Maple:  y(7) = 4, D(y)(7) = -2. 
Se iøvrigt kapitel 6 om differentialligninger.


diff eller D ?


Forsøger man at definere den afledede funktion fmærke af en funktion f ved brug af diff på 
følgende måde, får man et problem. 


Lad eksempelvis funktionen have forskriften .


Vi forsøger en definition af den afledede således:


fmaerke:= x -> diff( x^2*sin(Pi*x), x);


Vi afprøver definitionen:


fmaerke(x), fmaerke(u);


Alt er tilsyneladende i orden. Men se nu, hvad der sker:


fmaerke(3);


Error, (in fmaerke) invalid input: diff received 3, which is not 


valid for its 2nd argument


fmaerke(Pi);


Error, (in fmaerke) invalid input: diff received Pi, which is not


valid for its 2nd argument

Den gik altså ikke! Forklaringen skal søges i den måde, hvorpå Maple opfatter proceduren
fmaerke:


lprint(eval(fmaerke));


proc (x) options operator, arrow; diff(x^2*sin(Pi*x), x) end proc


Hvis proceduren fmaerke får et 3-tal vil den først overalt ombytte den formelle parameter x med 
den aktuelle værdi, nemlig 3. Derefter vil den forsøge at udregne diff( 3^2*sin(Pi*3), 3). Men 
dette vil diff protestere over, idet det ingen mening giver at differentiere mht. 3. 


Når den afledede ønskes som funktion, bør bruges D  på en funktion. Dette gøres enten på 
funktionens navn, hvis den har fået et, eller på dens definition vha. en pil som i dette eksempel:


D(  x-> x^2*sin(Pi*x)  );


Alternativt kan man ved brug af unapply lave en funktion ud af diff( x^2*sin(Pi*x), x).


fm:=unapply( diff( x^2*sin(Pi*x), x), x);


fm(3);


Partielle afledede


Lad g være følgende funktion af to variable:


g:=  (x,y) -> x*y^2*exp(-x^2-3*y^2);


Hvis vi vil differentiere udtrykket g(x,y) partielt mht. x,  skriver vi:


diff( g(x,y), x );


Hvis vi i stedet vil differentiere g(x,y) partielt mht. y,  skriver vi:


diff( g(x,y), y );


Hvis de partielle afledede skal findes som funktioner (ikke udtryk), gøres således, hvor 2-tallet 
betyder differentiation mht. variabel nummer 2:


D[2](g); 

Således fås  som funktion:


D[2,1](g); 


Da D[2,1] blot betyder, at der først skal differentieres mht. variabel nr. 2, dernæst mht. variabel 
nr. 1 kan samme resultat opnås ved


(D[1]@D[2])(g); 


 
Maple antager for en ukendt funktion, at de blandede afledede er ens. Det er de jo også næsten 
altid:


D[2,1](F)-D[1,2](F);
0


Bruges diff på det ukendte funktionsudtryk F(x,y), så krølles d'erne i output, idet Maple jo kan se, 
at udtrykket afhænger af flere variable:


diff(F(x,y),x);


De variable, der skal differentieres mht. kan anbringes i listeklammer, om man vil:


diff( x^2+7*x*y+cos(x),[x,y]);
7


Denne syntaks tillader den tomme liste, hvorved man får udtrykket tilbage:


diff( x^2+7*x*y+cos(x),[] );


Gradienten af en funktion af flere variable kan f.eks. findes således (vi bruger g fra før): 


diff~(g(x,y),<x,y>);


Her er brugt den elementvise version af diff nemlig diff~ (elementvise operationer af denne slags 
kom med Maple 13).
Alternativt kan man bruge map2:


map2(diff,g(x,y),<x,y>);


Her betyder <x, y> søjlevektoren med elementerne x og y. Resultatet kan laves til en funktion 
således

G:=unapply(%,x,y):


G(1,2);


I VectorCalculus-pakken findes desuden en Mapleprocedure for gradienten.


Differentiation af stykkevist defineret funktion


Til visse formål er D næsten uundværlig, bl.a. virker D også på funktioner defineret ved en 
Tuborg:


f:= x -> if x<1 then x^2 else x^4 end if;


D(f);


Derimod går det ikke så godt, hvis vi forsøger med diff:


diff( f(x),x); 


Error, (in f) cannot determine if this expression is true or 


false: x < 1


Problemet med diff( f(x), x) ) er, at Maple først forsøger at evaluere f(x). Det kan den imidlertid 
ikke, da den ikke kan afgøre, om x < 1 eller ej. 


f(x);


Error, (in f) cannot determine if this expression is true or 


false: x < 1


Det kan i øvrigt anbefales, at man ved definitionen af f bruger piecewise i stedet: 


w:=piecewise(x<-2,7,x<1,x^2,x^4);


Så kan også diff bruges:


diff(w,x);

Vi laver en funktion ud af w:


f:=unapply(w,x);


På denne funktion kan D anvendes:


D(f);


D(f)(x);


Implicit differentiation


Ligningen


ligning:=x^2*y-sin(x+y)=0;


sammenknytter x og y på en måde, der gør, at det er umuligt at finde et konkret formeludtryk for y 
udtrykt ved x og faktisk heller ikke omvendt:


solve( ligning, y);


solve(ligning, x);


Her skal et RootOf-udtryk indeholdende variablen _Z forstås som løsningen for _Z når udtrykkes 
sættes lig nul. I realiteten har Maple altså ikke kunnet finde et formeludtryk i de to tilfælde 
ovenfor.


Kurven med den givne ligning kan tegnes vha. implicitplot fra plots-pakken.
Først gør vi pakken klar til brug med with:


with(plots);

Herefter er alle procedurerne i plots-pakken umiddelbart tilgængelige, bl.a. implicitplot:


implicitplot(ligning, x=-1..1,y=-3.5..3.5);


x


0 1


y


1


2


3


Følgende kommando viser, at kurven går gennem punktet (0, ):


eval(ligning,{x=0,y=Pi});


I omegnen af punktet (0, ) er kurven graf for en funktion, som det ses af figuren ovenfor. Denne 


funktions afledede kan findes ved brug af implicitdiff. Nedenfor findes y'(x):


implicitdiff( ligning, y, x);


eval(%, {x=0, y=Pi});


Hermed har vi fundet, at tangenthældningen i punktet (0, ) er -1. I kapitlet om 


differentialligninger er vist en kommando, der giver Taylorpolynomier for den implicit givne 
funktion. 

3. Integration


Integralerne  og 


Ubestemt integration af udtrykket


udtr:=(x^3+a*x)/(x^2+2);


udføres ved at benytte proceduren int:


int(udtr, x);


Vi kan kontrollere resultatet ved differentiation:


diff( %, x);


Det ligner ikke det oprindelige udtryk, så vi sætter på fælles brøkstreg:


normal(%);


Vil man gerne se integralet skrevet op uden udregning, kan man bruge den inaktive Int:


Int( udtr, x);


Bemærk, at integraltegnet er gråt.
Den inaktive version kan omdannes til den aktive int ved brug af value:


value(%);


Pænt opskrevet:


Int( udtr, x): %=value(%);

Her er et andet eksempel. Funktionen g er givet ved:


g:=x->x*exp(x^2);


En stamfunktion til g findes således:


int(g(x), x);


Resultatet kan igen let kontrolleres ved differentiation: diff( %, x);


Bestemt integration kan af Maple gøres på både algebraiske udtryk og funktioner (defineret ved 
pile).
Det bestemte integral over intervallet [1 , 2] af funktionen  g  fås enten ved syntaksen: 


int(g(x), x=1..2); 


eller ved den variabelfrie syntaks


int( g, 1..2);


Bemærk, at der skal bruges mindst 2 punktummer ved intervalangivelsen.


Integrationsvariablen i Maple en global variabel. Dette har følgende konsekvens: 


Er x i kommandoen  int(g(x), x=1..2) tidligere tillagt en talværdi, vil Maple give en 


fejlmelding:


x:=78:


int(g(x),x=1..2);


Error, (in int) integration range or variable must be specified 


in the second argument, got 78 = 1 .. 2


Vi renser x for tilordningen ovenfor:


x:='x':


En numerisk tilnærmelse til resultatet af en udregning af det bestemte integral:


int( x/ln(x),x=2..3);


fås ved efterfølgende brug af 


evalf(%);

Output ser imaginært ud, men er jo i realiteten reelt:


simplify(%);
2.753652661


Et bestemt integral med ikke-konkrete grænser, som f.eks.  udregnes ikke nødvendigvis 


uden videre:


int(1/t,t=1..x);


Warning, unable to determine if 0 is between 1 and x; try to use 


assumptions or use the AllSolutions option


Vi må gøre antagelser om x, tilføje et tredie argument eller sætte _EnvAllSolutions til værdien true:


int(1/t,t=1..x) assuming x>0;


int(1/t,t=1..x, AllSolutions);


_EnvAllSolutions:=true:


int(1/t,t=1..x);


Et alternativt tredie argument er 'continuous' som instruerer int om ikke at lede efter 
diskontinuiteter for integrand og stamfunktion:


int(1/t,t=1..x,continuous);


Tilføjelsen 'continuous' skal man dog kun gøre, hvis man selv har styr på situationen.


Følgende stamfunktion til  er diskontinuert i :  


fu:=int(1/(2+sin(t)),t);


limit(fu,t=Pi,left)<>limit(fu,t=Pi,right);

Derfor kan det ikke undre, at man får det forkerte resultat ved kommandoen 


int(1/(2+sin(t)),t=0..2*Pi,continuous);
0


Det korrekte resultat fås uden tilføjelse:


int(1/(2+sin(t)),t=0..2*Pi);


Et ukompliceret integral som dette


restart;


Int(sin(t)/(1+cos(t)^2), t=0..x):  % = value(%);


udregnes heldigvis, som det ses, uden antagelser eller tilføjelser.


Numerisk integration


Ønskes brugt en numerisk metode til udregning af det bestemte integral 


g:=x->ln(x)/(1+x);


kan bruges en af følgende to kommandoer: 


evalf( Int( g(x), x=1..2) );
0.1472206770


evalf( Int( g, 1..2) );
0.1472206770


eller med Maple 13 også en af følgende:


int(g(x), x=1..2, numeric);
0.1472206770


int(g, 1..2, numeric);
0.1472206770


Int er den inaktive version af int. Brugen af Int i stedet for int  får Maple til ikke at forsøge eksakt
udregning først. 
Bruger man evalf(int(...)) (med lille i), og er ingen af integrationsgrænserne decimaltal, så 
forsøger Maple først at udregne integralet eksakt. Hvis dette lykkes, så giver evalf  en 
decimalbrøkstilnærmelse til dette eksakte resultat. Kun hvis det ikke lykkes at udregne integralet 
eksakt, vil en numerisk metode blive brugt på integralet. 
Vil man kontrollere det eksakte resultat af en udregning, kan det f.eks. gøres således:


Int(g(x),x=1..2): %=value(%);

evalf(%);


Decimaltallet til venstre er resultatet af en numerisk udregning af integralet, hvorimod 
decimaltallet til højre er decimaltalstilnærmelsen til det eksakte resultat. Når afrundingsfejl tages i
betragtning er de to tal lige store i dette tilfælde. Vi tillader os at konkludere, at det eksakte 
resultat er korrekt.


Bemærk, at et bestemt integral, hvor mindst den ene integrationsgrænse er et decimaltal, sendes 
til den numeriske integrator først. Dette betyder, at ovenstående integral også kan udregnes ved en
numerisk metode således:


int( g(x), x=1.0..2);
0.1472206770


int( g, 1.0..2);
0.1472206770


For mange procedurer kan man sætte informationsniveauet passende højt, så man under 
eksekveringen får udskrevet information om hvad der foregår. Således kan vi let verificere, at 
numerisk integration bruges ovenfor ved at sætte informationsniveauet for den numeriske 
integrationsprocedure til et passende højt niveau. For illustrationens skyld er det nok med det lave
niveau 1. 
Man skal ved denne kontrol være opmærksom på, at evalf husker resultater, så prøv hellere en 


anden funktion. Vi finder her  simpelthen  :


infolevel[`evalf/int`]:=1:


int( 3*x^2, x=0..1.0);


Control: Entering NAGInt


Control: trying d01ajc (nag_1d_quad_gen)


d01ajc: result=1.
1.


At proceduren ̀ evalf/int  ̀overhovedet skriver information ud, viser, at integralet sendes til den 
numeriske integrator. NAG står for Numerical Algorithm Group.


Den numeriske integrator kan selvfølgelig ikke klare næste eksempel, hvor øvre grænse er 
symbolet a, så efter besøg hos ̀ evalf/int  ̀sendes integralet til den eksakte integrator:


int( g(x), x=1.0..a);


Vi sætter informationsniveauet tilbage til 0:


infolevel[`evalf/int`]:=0:


 
Vi kan anbefale, at kommandoen evalf( Int(...)) eller int(..., ..., numeric) bruges, hvis man 
insisterer på en numerisk integration. For hjælp til udnyttelse af de valgfri muligheder ved 

numerisk integration prøv


?int,numerical


Variabelfri syntaks og for tidlig evaluering


Variabelfri syntaks er undertiden en nødvendighed for at forhindre for tidlig evaluering.  


Betragt som eksempel herpå den omvendte funktion til funktionen  defineret på 


intervallet . Den omvendte er defineret på værdimængden for   som er [0, 1]. 


Den omvendte (som vi her kalder f ) kan bestemmes numerisk ved:


f:= t-> fsolve(sin(x)/x=t,x=0..Pi);


Nu kan f bestemmes i ethvert konkret tal mellem 0 og 1:


f(.9);
0.7866830720


Hvis t kun er bogstavet t, så kan f(t) ikke udregnes:


f(t);


Error, (in fsolve) t is in the equation, and is not solved for


Vi kan imidlertid både plotte grafen for f og finde det bestemte integral , vel at mærke 


hvis vi bruger variabelfri syntaks:


plot(f, 0..1);


0 1


1


2


3


evalf(Int( f, 0..1));
1.851937052


Resultatet kan i dette tilfælde let kontrolleres, idet en arealbetragtning giver 

:


int(sin(x)/x,x=0..Pi): %=evalf(%);


Stykkevist definerede funktioner


For tidlig evaluering giver (som også nævnt ovenfor) sommetider problemer ved numerisk 
integration. Som et andet eksempel definerer vi en funktion ved en Tuborg:


f:= x -> if x<2 then x^2 else 6-x end if; 


evalf(Int(f(x), x=0..6) );


Error, (in f) cannot determine if this expression is true or 


false: x < 2


Problemet er, at Maple tror den skal evaluere f(x) for bogstavet x. Den kan selvfølgelig ikke afgøre
om x < 2 eller ej, sålænge x kun er et bogstav. Den følgende syntaks bruges på procedurer:


evalf(Int(f, 0..6) );
10.66666667


Bemærk procedurenavn uden variabel og ingen variabel ved angivelse af integrationsintervallet.
Her virker også alternativet


evalf(Int( 'f(x)', x=0..6));
10.66666667


hvor apostrofferne blev brugt til at forhale evaluering:


'f(x)';


Denne forhaling betyder, at evalf/int kan komme i gang med sit arbejde før Maple brokker sig 
over at den ikke kan afgøre om bogstavet x er mindre end 2 eller ej. 
 
Bemærk dog, at der ingen redning er for det eksakte integral. Ingen af følgende to virker:


int('f(x)',x=0..6);


Error, (in f) cannot determine if this expression is true or 


false: x < 2


int(f,0..6);


Error, (in f) cannot determine if this expression is true or 


false: x < 2


Ved definition af en Tuborg-funktion bør i stedet bruges piecewise:







Kapitel 3: Integration


Kapitel 3 side 8


f:= x -> piecewise(x<2, x^2, 6-x);


f(x);


int( f(x), x=0..6);
32


3


Her er et lidt mere indviklet eksempel:


udtryk:=piecewise(x<Pi/4,x/cos(x)^2,x<Pi,sin(x),x-Pi);


int(udtryk,x=0..2*Pi);


Maple regner undertiden galt!


Maple's eksakte integrator regner undertiden galt, så man bør kontrollere Maples integrationer 
ved også at foretage en numerisk integration vha.  evalf(Int(...)) (det store I er afgørende!) eller
int(...., numeric).


I enhver version af Maple vil man kunne finde bestemte integraler, som Maples eksakte integrator 
udregner forkert. Dette skal egentlig ikke opfattes som en kritik af Maple, men blot som en 
konstatering: Det er svært det her!
Men der sker naturligvis hele tiden forbedringer. Vi har i flere udgaver af disse noter brugt 
eksemplet


Int( ln( (5+3*sin(x))/(5+4*sin(x))),x=0..2*Pi): %=combine


(value(%));


evalf(%);


evalf udregner her begge sider. I venstre side udregnes integralet ved en numerisk metode, i højre 
sider findes en decimalbrøkstilnærmelse til det eksakte result.
Integralet blev undertiden (specielt irriterende, dette 'undertiden') regnet forkert ud i Maple 8 og 9.

Fra og med Maple 9.5 ser det ud til at gå godt.
 
Vi må således hele tiden finde nye eksempler på fejlagtige udregninger af bestemte integraler.


Vil man prøve en udregning flere gange, må man enten begynde med et restart eller bede Maple 
om at glemme resultatet af tidligere integrationer. Dette sidste kan gøres ved brug af proceduren
forget på formen forget( int ); Herved fjernes husketabellen fra underproceduren ̀ int/int`, der 
husker udregnede resultater indtil næste "indsamling af affald". Proceduren `int/int` har nemlig
option remember, system. Det betyder, at proceduren husker tidligere udregnede resultater, men 
tilføjelsen system gør, at husketabellen fjernes, når Maple indsamler affald (garbage collection), 
hvilket sker med passende mellemrum. Husketabellen fås ved kommandoen op(4, eval(`int/int`))
;  men bemærk, som sagt, at tabellen med mellemrum fjernes. 


Her er et eksempel, hvor integralet udregnes forkert i Maple 13 (og også i alle versioner i 
hvertfald siden Maple 5.1):


restart;


A:=Int(exp(-x)/(1-exp(-x)),x=0..infinity);


Beder man nu om value(A) fås Maples bud på det eksakte resultat: 


value(A);


  er Eulers konstant, der med 10 betydende cifre er


evalf(gamma);
0.5772156649


Integralet er imidlertid divergent på grund af integrandens singularitet i 0, hvilket den numeriske 
integrator da også opdager:


evalf(A);


Sjovt nok klares integralet på intervallet [0, 1] helt fint:


B:=Int(exp(-x)/(1-exp(-x)),x=0..1);


value(B);


Det samme gælder integralet på [1,


Int(exp(-x)/(1-exp(-x)),x=1..infinity): %=value(%);


evalf(%);

Følgende går ikke godt i Maple 13:


int(sin(x)/x^3, x = 0 .. infinity);


Den numeriske integration opdager heldigvis, at integralet jo er divergent:


int(sin(x)/x^3, x = 0 .. infinity, numeric);


Planintegral


Kan man selv omskrive et planintegral til et dobbeltintegral, så kan det udregnes af Maple. Vi 
giver et par eksempler. Vi har valgt at bruge Int i stedet for int, så vi kan se, om vi har tastet 
dobbeltintegralet rigtigt ind, før vi beder Maple om udregningen. 


Først udregner vi i polære koordinater planintegralet  over det område S i første kvadrant,


der er begrænset af x-aksen og kurven  . 


plot(cos(theta)^2,theta=0..Pi/2,coords=polar,scaling=


constrained,filled=true,color=gray);


0 1
0


A:=Int(Int( r^4*cos(theta)*sin(theta)^2,r=0..cos(theta)^2),


theta=0..Pi/2);


Her kommer det færdige resultat:


value(%);
256


45045


Følgende syntaks med kun ét Int (eller int) kan (i Maple 13) også bruges:

Int( r^4*cos(theta)*sin(theta)^2,r=0..cos(theta)^2,theta=0..


Pi/2);


Vil man se mellemregninger, kan man gøre således, idet første operand i dobbeltintegralet netop 
er det indre integral:


applyop(value,1,A);


value(%);
256


45045


Et andet eksempel:


B:=Int( Int( x/sqrt(1-y^2), y=0..sin(x)), x=0..Pi/2); 


Her kommer ufrivilligt en mellemregning af sig selv:


value(B);


simplify kan udnytte, at x er integrationsvariablen og derfor må ligge mellem 0 og  :


simplify(%);


Numerisk Udregning af planintegral


Maple har procedurer til direkte numerisk udregning af et integral over et rektangel i flere 
dimensioner (altså over et cartesisk produkt af intervaller). Desuden kan dobbelt- (og flerdobbelt-)
integraler udregnes ved brug af numeriske integrationsmetoder for endimensionale integraler.  
Vi giver nogle eksempler.







Kapitel 3: Integration


Kapitel 3 side 12


Dette dobbeltintegralintegral kan Maple 13 ikke udregne eksakt


int(x^2*sin(x^2+cos(y)),y=0..x,x=0..2);


men det går udmærket numerisk:


int(x^2*sin(x^2+cos(y)),y=0..x,x=0..2,numeric);


eller med en lidt anderledes syntaks (listeklammerne om intervallerne kan undværes):


evalf(Int(x^2*sin(x^2+cos(y)),[y=0..x,x=0..2]));


Her er et integral over et 4-dimensionalt rektangel: [0,1]x[0,1]x[0,1]x[0,1]. Integranden er


f:=(x,y,z,w)->1/(1+x+y+z+w);


Her udregnes integralet med den forenklede syntaks:


evalf(Int(f(x,y,z,w),[x=0..1,y=0..1,z=0..1,w=0..1]));
0.3471439323


int(f(x,y,z,w),[x=0..1,y=0..1,z=0..1,w=0..1],numeric);
0.3471439323


Integralet kan iøvrigt udregnes eksakt. Vi får:


Int(f(x,y,z,w),[x=0..1,y=0..1,z=0..1,w=0..1]): %=value(%);


evalf[12](%);


Et tredie eksempel:


B:=Int( Int( (x+1)*y/sqrt(x^2+y^2+2), y=0..2*sqrt(1-x^2)), 


x=-1..1); 


evalf(B);
1.403741149

Planintegralet (en DTU-eksamensopgave fra 1955) kan udregnes eksakt:


combine(value(B));


radnormal(%);


evalf(%);
1.403741149

4. Grænseværdi, Taylorudvikling, Sum, Følge og Produkt


Grænseværdi


Grænseværdier af udtryk findes vha. limit, hvis tilsvarende inaktive version hedder Limit.
Lad u være udtrykket


u:=(x*sin(x)-arctan(x^2) ) / x^4;


Grænseværdien af u for  fås således:


limit( u, x=0);


Man bemærker, at der skal bruges et lighedstegn, ikke en pil.
Her er samme resultat stillet smukt op:


Limit( u, x=0): %=value(%);


Ensidige grænseovergange fås ved at tilføje et tredie argument:


limit( x*ln(x), x=0, right);
0


Her er et eksempel på grænseværdi i :


limit( (1+1/n)^n, n=infinity);
e


Taylorudvikling


Der findes en særlig procedure til brug ved Taylorudvikling af et udtryk f(x).  Vi tager som 
eksempel følgende funktion


f:=  x -> x^2*(ln(x)-1);  


Bemærk, at 3. argument i taylor-proceduren er trunkeringsordenen, ikke udviklingsordenen:

taylor(f(x), x=1, 5); 


whattype(%);
series


Output er af en særlig datatype: series (series betyder række).  Restleddet i Taylors formel 


repræsenteres af . Vi er interesseret i det fjerde Taylorpolynomium og må derfor 


omdanne vores udtryk af type series til et polynomium. Dette sker ved en convert-kommando:


p:=convert(%%,polynom);


Hermed er p et polynomiumsudtryk. Det er dog tit mere praktisk at arbejde med Taylorpolynomiet
i form af en funktion, så vi bruger unapply:


P4:=unapply(p,x);


Vi tegner graferne for f og P4 i samme koordinatsystem. Den stiplede kurve (linestyle=2) er 
grafen for P4:


plot([P4,f],0..2.5,linestyle=[2,0],color=[red,blue],


thickness=[3,1],


axes=boxed,legend=["P4","f"]);


P4 f


0 1 2


0


Afvigelserne mellem f og P4 kan evt. plottes:


plot( f-P4, 0..2.5,caption=typeset("Forskellen  ",'f'-P[4]))


;

1 2
0


Forskellen  


Taylorudviklingens orden (eller rettere trunkeringsorden) kan udelades. Så vil Maple i stedet 
bruge værdien af variablen Order som ordenen. Med mindre man selv har ændret denne, har den 
værdien 6:


Order;
6


Her er ordenen udeladt:


taylor(exp(x),x=0);


Den orden, der angives som tredie argument til taylor, er faktisk den orden, som Maple bruger 
ved rækkeudvikling af de enkelte bestanddele af det givne udtryk. Man kan derfor komme ud for, 
at ordenen i resultatet bliver lavere, som i næste eksempel.


Vi begynder med at bede Maple om at slette indholdet i husketabellen for series:


forget(series);


udtryk:=(sin(x)-x+x^3/3!)/x^5;


taylor( udtryk, x=0, 8);


Ordenen er 5 mindre end forventet pga. nævneren. 
Maple husker rækkeudviklingen. Dette kan ses, hvis man beder om at se husketabellen for 
proceduren series. 
Kommandoen

op(4,eval(series));


viser tabellens indhold, der kan fylde en del (derfor forget-kommandoen ovenfor).
Man ser af tabellen, at Maple bl.a. har noteret sig udviklingerne for 


.


Kommandoen 


op(4,eval(series))[udtryk,x];


viser det, som tabellen husker om udviklingen af vores udtryk i variablen x.
Hvis vi nu beder om udviklingen til en lavere orden, så fås enten resultatet til samme orden som 
før eller til den "forventede" orden (den mindste af de to):


taylor( udtryk, x=0, 6);


taylor(udtryk, x=0, 3);


taylor(udtryk,x=0, 2);


Taylorpolynomier for funktioner af flere variable kan også udregnes. Proceduren hedder mtaylor.
I modsætning til situationen for taylor er outputtet ikke af type series, der kommer intet O-led,
men det er stadig trunkeringsordenen (den orden, som O-leddet ville have), der skal skrives som 
sidste argument. 

I det følgende eksempel findes det 3. Taylorpolynomium. Sidste argument skal da være 3+1 = 4. 


p:=mtaylor( exp(x*y)*sin(y), [x=0, y=Pi], 4); 


type(p,series);
false


mtaylor virker også på funktioner af én variabel og har altså den fordel frem for taylor, at der 
intet O-led kommer:


mtaylor(x*ln(x),x=1,4);


En abstrakt Taylorudvikling kan også foretages:


mtaylor(g(x,y),[x=a,y=b],3);


mtaylor virker i realiteten som følgende konstruktion, der også normalt bruges ved beviset for 
Taylors sætning i flere variable:


subs(t=1,h=x-a,k=y-b,


convert(taylor(g(a+t*h,b+t*k),t=0,3),polynom)


);


Der findes en procedure coeftayl, der finder en given koefficient til Taylorpolynomiet. Den kan 
bruges på funktioner af én eller flere variable: 
  


coeftayl(sin(x),x=0,5);
1


120


Hvis vi ønsker koefficienten til  i Taylorudviklingen for  med 


udviklingspunkt ( ) kan den fås således:


coeftayl(exp(x*y)*sin(y), [x,y]=[0,Pi], [2,1]);







Kapitel 4: Grænseværdi, Taylorudvikling, Sum, Følge og Produkt


Kapitel 4 side 6


Bemærk, at syntaksen i andet argument afviger fra syntaksen for mtaylor.


Generel rækkeudvikling


Mere generel rækkeudvikling end Taylorudvikling klares af proceduren series:


udtryk:=(x^2+3)/(x-2)^2/(x+1);


Vi ønsker at udvikle udtryk i positive og negative potenser af   og skriver derfor:


series( udtryk, x=2, 7); 


Som for taylor kan ordenen udelades. Maple bruger da værdien af variablen Order, der ved start 
er 6.


Asymptotisk udvikling
Den asymptotiske udvikling for et udtryk findes ved hjælp af asympt.
Her følger nogle eksempler:


asympt(arctan(x),x,5);


Her udnyttede Maple, at


arctan(x)+arctan(1/x)=simplify(arctan(x)+arctan(1/x)) 


assuming x>0;


således at


arctan(x)=Pi/2-subs(t=1/x,taylor(arctan(t),t=0,5));


asympt(x*ln(x)/(1+x),x);

Bemærk ovenfor, at der intet er sket med logaritmen. I øvrigt burde store O-leddet også indeholde
ln(x).


f:=int(exp(-t^2),t=0..x);


asympt(f, x,4);


convert/polynom skal ikke tages for bogstaveligt:


f4:=convert(%,polynom);


plot([f,f4],x=1..3,legend=["f","f4"]);


f f4


x


1 2 3


Rækker og summer


Et konkret antal led i en sum   findes ved brug af add:


add(n^2, n = 0..100 );
338350


Er antallet ikke konkret, som i rækken  , hvor øvre grænse blot er navnet N, så skal bruges


sum:


sum(n^2, n = 0..N );

sum vil forsøge at finde en formel for summen, og dette lykkedes i vores eksempel. Hvis det ikke 
lykkes, så returnerer sum uevalueret:


sum(n^2/(1+exp(n)), n = 0..N );


sum kan også bruges, hvor N er konkret (eksempelvis 100), men i sådanne tilfælde er add bedre, 
se nedenfor.   er en tilladelig øvre grænse i sum: 


sum(1/n^2,n=1..infinity);


sum har en inaktiv version, der hedder Sum, ligesom int, diff og limit har inaktive versioner Int, 
Diff og Limit:


Sum(n*exp(-n), n=0..N); 


value omdanner Sum til sum:


value(%);


simplify(%);


Den inaktive Sum kan benyttes til udstilling, som f.eks. i denne uendelige række:


Sum(1/n^6, n=1..infinity):  % = value(%); 


I sum-proceduren er summationsvariablen en global variabel. Det er derfor vigtigt, at den brugte 
summationsvariabel (ovenfor kaldet n) ikke tidligere er blevet tildelt nogen værdi. Dette kan være 
sket ved direkte tilordning (f.eks. n:=7;) eller n kan have været brugt i en for n from .. to .. do ... 
end do sætning:


n:=5:  


sum(n^10,n=1..N);


Error, (in sum) summation variable previously assigned, second 


argument evaluates to 5 = 1 .. N

Løsningen er nem:  


n:='n': 


sum(n^10,n=1..N);


Man kan også bruge følgende syntaks, hvor man skal bemærke sig apostrofferne, der forhindrer 
evaluering (her af n til 8):


n:=8:  


sum('n^3', 'n'=1..N); 


sum('2^n', 'n'=0..N); 


En sum af et konkret antal led (eksempelvis 37, ikke N) fås mest effektivt vha. add. 


A:=add(1/k, k=1..1000);


Denne sum kan selvfølgelig også fås vha. sum: 


B:=sum(1/m ,m=1..1000); 


Maple's sum-procedure lægger ikke blot tallene sammen ét for ét, men prøver at finde en generel 
formel for summen. sum kan derfor være langsommere end add. Sidstnævnte kan dog kun bruges,
når et konkret antal led skal adderes. I eksemplet ovenfor har sum  netop fundet en formel, som 

det også fremgår af følgende kommando: 


sum(1/m, m=1..N);


Konstanten  er Eulers konstant, der er defineret som følgende grænseværdi: 


Limit(Sum(1/k,k=1..N)-ln(N),N=infinity): %=value(%);


Digamma-funktionen  er givet ved , hvor  er gamma-funktionen, se iøvrigt 


hjælpen:


?Psi


eller prøv kommandoen FunctionAdvisor(Psi);  evt. med den ekstra tilføjelse definition:


FunctionAdvisor(Psi,definition);


add generes ikke af tilordninger til summationsvariablen: 
 


n:=1:


add(n^2,n=1..10);
385


Bemærk ovenfor, at heller ikke første argument til add blev evalueret til 1 før summationen!
Sammenlign resultatet af de to følgende add-kommandoer: 


restart;


u:=k^2:  k:=2:  u;
4


Selv om u ved fuld evaluering bliver til tallet 4 fås


add(u,k=1..10); 
385


Vi ser, at add er klog nok til at indse, at når der skal summeres over k, så kan det ikke være 
meningen, at k i udtrykket u skal evalueres til 2.
Fuld evaluering af argumenterne til en procedure før proceduren selv evaluererer er det generelle i
Maple. Dette betyder altså, at add har specielle evalueringsregler. Ellers ville add ovenfor få 
argumenterne 4, 2 = 1..10. Derimod følger sum den generelle regel, jf. ovenfor. Derfor fås en 
fejlmelding:.

sum(u,k=1..10); 


Error, (in sum) summation variable previously assigned, second 


argument evaluates to 2 = 1 .. 10


Hedder summationsvariablen  i stedet n, fås


add(u,n=1..10); 
40


hvor add nu er klar over, at u skal evalueres helt, hvilket som sagt er det normale i Maple.


Sammenlign også resultatet af følgende to add-kommandoer: 


restart;


u:=k^2:  k:=p^3:


add(u,p=1..10); # Her opfattes u som p^6
1978405


add(u,k=1..10); # Her opfattes u som k^2
385


At add da overovedet kan se k skyldes, at man kan få Maple til kun at evaluere ét niveau:


eval(u,1);


Indeholder de størrelser, der skal adderes ikke en heltalsvariabel over hvilken der skal summeres, 
kan summeringen ske ved følgende syntaks, hvor der summeres over operanderne (elementerne, 
byggestenene, se kapitel 14) i udtrykket efter lighedstegnet:


add( s^2, s={a,b,c,d,56}); 


add( s^2, s=[a,b,c,d,56]);


Man kan også bruge følgende syntaks med in:


add( s^2, s in {a,b,c,d,56});


Uden for enhver sammenhæng er følgende måske en smule bizart, men altså tilladeligt:


add( exp(s), s=a*b*c/d);


Forklaring: Operanderne i a*b*c/d er


op(a*b*c/d);


Numerisk summation af uendelige rækker

Summen af en uendelig række kan også findes numerisk, analogt med at integraler kan 
udregnes ved en numerisk metode.
Vi giver først et eksempel, hvor vi kender summen. Vi sætter printlevel til 5, så vi kan se lidt 
af det, der sker:


restart;


printlevel:=5:


evalf(Sum(1/n^2,n=1..infinity));


{--> enter evalf/Sum, args = 1/n^2, n = 1 .. infinity


<-- exit evalf/Sum (now at top level) = 1.644934067}
1.644934067


Vi ser, at en procedure ̀ evalf/Sum` bliver kaldt. Det er en numerisk procedure. Mange flere 
detaljer fås ved at sætte printlevel til 20.
Vi sætter printlevel tilbage til 1:


printlevel:=1:


Den eksakte værdi af summen er


sum(1/n^2,n=1..infinity);


En decimalbrøksapproksimation hertil er


evalf(%);
1.644934068


Numerisk summation er selvfølgelig mere interessant i de tilfælde, hvor sum returnerer 
uevalueret som i følgende eksempel:


sum(n^2/(1+exp(n)),n=1..infinity);


evalf(%);
1.800908249


Følger


En følge af et konkret antal elementer fås bedst med proceduren seq. Som et eksempel på dens 
normale anvendelse finder vi de første 10 kvadrattal:


seq( k^2, k=1..10);


Variablen i seq behandles ganske som illustreret ovenfor for variablen i add.
Ligesom for add kan følgende syntaks undertiden være bekvem. Index-variablen (i første 
eksempel k) gennemløber samtlige operander (elementer eller "byggestene", se kapitel 14) i det 
udtryk, der står efter lighedstegnet:


seq( k^2, k = [1,3,7]);

seq( Int(f(x),x)=int(f(x),x) , f=[sin, cos, tan, cot]);


I Maple 13 kan et lignende resultat opnås ved brug af de tre elementvise operationer Int~, int~, =
~ således: 


Int~([sin, cos, tan, cot](x),x) =~ int~([sin, cos, tan, cot]


(x),x);


seq( sin(k), k=a+2*b-c^2);


Ligesom for add kan bruges in:


seq( 'D'(f)=D(f), f in [sin,cos,exp,sinh,arctan]);


Følger kan også frembringes vha. dollartegnet $:


n^2$n=1..5;


Bemærk, at i $-operatoren evalueres variablen (hér n) helt som for sum-proceduren, men modsat 
situationen for seq. Den må altså ikke evaluere til andet end et navn:


kk:=7:  


kk^2$kk=1..5;


Error, invalid input: `$` expects its 2nd argument, range, to be 


of type {numeric, algebraic, name = literal .. literal, name = 


algebraic .. algebraic}, but received 7 = 1 .. 5


Dollartegnet er godt, når det skal være kort:


$1.7..9.9; 


Ovenstående har samme virkning som


x$x=1.7..9.9; 


I forbindelse med differentiation ved diff har vi haft glæde af følgende variant:


x$5;

der også lidt mere udførligt fås ved kommandoen


x$j=1..5;


diff(x^7,x$3);


$-operatoren opfører sig også i andre henseender anderledes end seq. Den tillader således 
ukonkrete grænser. Der returneres i så fald blot uevalueret, dvs. at man blot får et ekko af input:


n^2$n=a..b; 


seq derimod vil protestere:


seq(n^2, n=a..b); 


Error, unable to execute seq


Den tomme følge kan fås i Maple ved kommandoen NULL:


3+4;
7


NULL;


%;
7


x,y,22/7,NULL,pebernød;


Man bemærker, at NULL helt ignoreres og at det ekstra komma i sidste input også er væk i 
output. NULL er nyttig i programmeringssammenhæng.
Den tomme følge er resultatet af følgende kommando:


seq(k^2,k=3..2);


men kan også være resultatet fra en solve-kommando, der ikke finder en løsning:


solve(exp(x)=0,x);


Produkter


Procedurerne product og Product er ganske analoge til sum og Sum:


product(m^2, m=2..N);

Product(m^2, m=1..N):  %=value(%);


Vi prøver et uendeligt produkt:


product((m^3-1)/(m^3+1),m=2..infinity); 
2


3


Svarende til additionsproceduren add er der en multiplikationsprocedure mul, der finder produktet
af et konkret antal faktorer:


mul(k/(k+1),k=1..1000);
1


1001


mul(k^2, k=[a,b,c]);


Numerisk udregning af uendelige produkter er mulig ligesom for summer. Vi tager et eksempel, 
hvor vi først finder den eksakte værdi:


product(1+1/m^3, m=1..infinity);


evalc bringer udtrykket (som jo her bør være reelt) på rektangulær form:


evalc(%);


En decimalbrøksapproksimation til den eksakte værdi:


evalf(%);
2.428189796


Numerisk udregning (bemærk stort P i Product):


evalf(Product(1+1/m^3, m=1..infinity));
2.428189792


convert/Sum
restart;


Maple kan udtrykke sine kendte funktioner (der kan fås ved at indtaste ?inifcns på en inputlinie) 
som uendelige rækker, når dette da i øvrigt er muligt:

convert(exp(x),Sum);


Det kan være en god idé at spørge FunctionAdvisor om en sådan sumform eksisterer:


FunctionAdvisor(ln,sum_form);


Warning: when the "sum" form of a function is requested, only the


function name (e.g., ln) - is expected. Extra arguments are being


ignored.


convert(ln(x),Sum) assuming abs(x-1)<1;


Hen til kommoden og tilbaws igen: Vi erstatter Sum med sum (lille s) ved brug af value:


value(%);


Indgår funktionerne som en del af et større udtryk konverteres de også:


convert(sin(x)+4*x^5*cos(x),Sum);


Dette kan simplificeres:


simplify(%);


Kan man ikke lide Maples _k1 kan dette erstattes af et andet symbol, f.eks. bare k:


subs(_k1=k,%);

Som en lille kontrol forsøger vi at gå tilbage igen:


res:=value(%);


simplify(res);

5. Løsning af Ligninger


Ligningsløseren solve


Ligninger løses eksakt vha. solve:


ligning1:=3*x^3+x^2+x+3=0;


Vi ønsker at løse ligning1 og opbevare løsningerne i L:


L:=solve(ligning1,x); 


Vi ser, at Maple har fundet 3 løsninger. Ved den brugte syntaks fås resultatet som en følge af 3 tal.


Vil man have løsning nr. 2, så beder man om


L[2]; 


I dette eksempel er der ingen tvivl om, hvad den ubekendte hedder. Derfor kan Maple da også 
undvære en angivelse heraf:


solve(ligning1);


Hvis vi i solve-kommandoen sætter Tuborg-klammer om den ubekendte eller om ligningen, så fås 
resultatet på en anden form:


LTuborg:=solve(ligning1,{x});


Vi får stadig en følge, men en følge af 3 mængder. Denne form er velegnet, hvis man i et eller 
andet udtryk ønsker at erstatte x med løsningen . Ønsker vi således at kontrollere løsningerne, kan 
det gøres således:


subs(LTuborg[2], ligning1); 

expand(%);


Kontrol af alle rødderne kan gøres ved lidt simpel programmering. Man kan enten bruge en løkke 
af typen for .. do ...end do, man kan bruge en seq-kommando eller man kan bruge map. I det 
første og det tredie tilfælde har vi brugt LTuborg, i det andet har vi for eksemplets skyld brugt L:


for i from 1 to 3 do expand( subs(LTuborg[i], ligning1) ) 


end do;


seq( expand( subs(x=L[i], ligning1) ), i=1..3);


map(subs,{LTuborg},ligning1): expand(%);


Bruger man listeklammer (kantede parenteser) om den ubekendte, kommer resultatet som en liste 
af lister (ikke som måske forventet en følge af lister): 


solve(ligning1,[x]);


Denne syntaks er mere interessant, når der er tale om flere ligninger med flere ubekendte. Se 
nedenfor.


Ligninger af grad 5 eller derover kan generelt ikke løses eksakt, dvs. de (komplekse) løsninger,
der jo dog findes, kan ikke udtrykkes ved en formel kun indeholdende koefficienterne, rodtegn og 
tegnene  *, /, +, -. 


solve( 7*x^5+2*x^4-x^3+3*x^2+4*x-1,x);


Følgen af de 5 rødder er udtrykt ved RootOf med et index, der holder styr på rødderne.


evalf(%);


Et polynomium af n'te grad har n rødder. Det ved Maple også.
Situationen er værre for andre funktioner end polynomier. Maple kommer derfor tit i den 
situation, at den ikke kan løse en given ligning eksakt. 
Skal Maple løse en ikke-polynomial ligning, f.eks. en trigonometrisk ligning, så finder den 
normalt kun én løsning:


solve(sin(x)+2*cos(x)=0,{x});  

Samtlige løsninger?


Da cos og sin er periodiske med samme periode (nemlig ), må der selvfølgelig være 


uendeligt mange løsninger til ligningen ovenfor. Hvis man sætter variablen _EnvAllSolutions 
til true, så findes alle løsninger:


_EnvAllSolutions:=true;


fuldL:=solve(sin(x)+2*cos(x)=0,{x});


Resultatet skal forstås på den måde, at _Z1~ er et vilkårligt helt tal. Ved løsning af andre 
ligninger kan forekomme en variabel af formen _B~ eller _NN~. Variablen _B~ kan kun 
antage værdierne 0 og 1 og _NN~ kun ikke-negative heltallige værdier. Tegnet ~ (tilde) efter 
_Z betyder, at Maple har gjort antagelser om _Z. 
Vi kontrollerer resultatet ved indsættelse i ligningen:


subs(fuldL,sin(x)+2*cos(x)=0);


expand(%);


Bemærk, at variablen _Z1~ er  "en lokal variabel, der er undveget fra en procedure"  (se 
kapitel 15 om programmering). 
Dette har som konsekvens, at man ikke har direkte adgang til at tildele _Z1~ en værdi.
Den lokale variable _Z1~ kan dog fiskes ud fra fuldL ved brug af indets (= indeterminates). 
Som andet argument kan vi angive en type. Her har vi brugt typen local.
Bemærk, at ordet local må omgærdes af accent grave, da ordet har en anden funktion i 
programmeringssammenhæng, hvor det bruges ved erklæring af lokale variable.


indets(fuldL,`local`);


Mængdeklammerne kan fjernes ved at bede om operanderne i udtrykket. Her betyder det 
elementerne i mængden:


z:=op(%);


Nu har vi fanget den undvegne lokale variable.


about(z);


Originally _Z1, renamed _Z1~:


  is assumed to be: integer

Vi kan nu eksempelvis erstatte _Z1~ med 77 i løsningen:


subs(z=77,fuldL);


Lad os sætte _EnvAllSolutions tilbage til startværdien, nemlig dens eget navn:


_EnvAllSolutions:='_EnvAllSolutions';


z:='z': # Det samme gør vi for z


Ingen løsninger?


Har en ligning ingen løsninger, så returneres ingenting:


solve(exp(x)=0,x);


solve(exp(x)=0,{x});


Ved begge syntakser ovenfor forventes en følge af løsninger. I dette tilfælde fås den tomme 
udtryksfølge: NULL.
Ved brug af syntaks med listeklammer om den ubekendte fås altid en liste, men listen er her 
tom:


solve(exp(x)=0,[x]);


Kan Maple ikke finde nogen løsning (selv om der faktisk er mindst en) kan resultatet være 
NULL:


solve(sin(x)=cos(x^2),x);


Warning, solutions may have been lost


men der blev her printet en advarsel til skærmen. Bemærk, at det der printes til skærmen ikke 
er output og det kan derfor ikke hentes frem ved brug af %:


%;


Resultatet fra solve kan også være et RootOf-udtryk::


solve(sin(x)=x-1,x);


Ved at sætte informationsniveauet for solve til 1 fås en advarsel, når solve ikke kan finde en 
løsning:


infolevel[solve]:=1:


solve(exp(x)=0,x);


solve: Warning: no solutions found


Vi sætter infolevel[solve] tilbage til 0:


infolevel[solve]:=0:


Parametrisk løsning af af en ligning i to ubekendte


Maple kan finde en parametrisk løsning til en ligning i to ubekendte:

solve(x^2+y^2=1,{x(t),y(t)});


Bemærk, at cirklen ikke er blevet parametriseret med den sædvanlige trigonometriske 
parametrisering.
Næste eksempel er cardioiden givet i cartesiske koordinater:


solve(x^2+y^2=sqrt(x^2+y^2)+x,{x(t),y(t)});


plot(map(subs,[%],[x,y,t=-20..20]),title="Cardioiden");


0 1


1


Cardioiden


Numerisk ligningsløsning: fsolve


Vi betragter nu en ligning, om hvilken det vides, at løsningerne ikke alle kan udtrykkes ved de til 
rådighed værende funktionsudtryk:


ligning2:=2*sin(x)=x;


solve(ligning2,x);


Vi fik et svar, men dette ligner nærmest en omformulering af spørgsmålet. Vi må derfor løse 
ligningen numerisk.

Der er to former: 


fsolve(ligning2,x);


0.


fsolve(ligning2,{x});


Det bemærkes, at vi kun får én løsning. De andre kan fås ved enten at angive et søgeinterval eller 
et startgæt.  Igen kan de to forskellige former med og uden { } bruges. Først prøver vi at angive et
søgeinterval:


fsolve(ligning2, x=1..2);
1.895494267


fsolve(ligning2, {x=-2..-1});


Her prøver vi med et startgæt:


fsolve(ligning2, x=2); 
1.895494267


fsolve(ligning2, {x=-3});


Man kan fortælle Maple, at den skal undgå en løsning, f.eks. hvis der (som her) er tale om en 
triviel løsning, som vi ikke behøver Maple til at finde:


fsolve(ligning2, x, avoid={x=0});


fsolve returnerer uevalueret, når den ikke kan finde en løsning (selv om der muligvis er): 


fsolve(ligning2,x=2..3);


men vil returnere NULL, når den konstaterer, at der ingen løsning er:


fsolve(exp(x)=0,x);


Ved en smule programmering lykkes det hér nedenfor at finde samtlige løsninger til ligning2:


S1:={}:  # S1 sættes først til den tomme mængde


for k to 5 while not has(S1,fsolve) do 


S:=S1; S1:=fsolve(ligning2,{x},avoid=S) union S 


end do:


S;

Man kan blive klogere på, hvad der foregik ovenfor, hvis man ændrer kolon efter end do til et 
semikolon.


Det er en god idé at plotte i forbindelse med løsning af ligninger, eksempelvis kunne man plotte 
venstre- og højresiden af ligning2 ovenfor:


plot( [lhs(ligning2), rhs(ligning2)], x=-3..3);


x


0 1 2 3


1


2


3


fsolve accepterer følgende varianter med søgeinterval foruden de tilsvarende med Tuborg-
klammer:


fsolve( sin , 4..7);
6.283185307


fsolve( sin(x), x=4..7);
6.283185307


fsolve( sin(x)=0, x=4..7);
6.283185307


fsolve( sin(x), x, 4..7);   # 4..7 kan erstattes af x=4..7
6.283185307


Brug af RootOf
Vi betragter nu ligningen , som vi vil løse for den ubekendte x. 


ligning3:=sin(x)=a*x-2;


Ligningen har præcis én løsning, hvis a ligger i intervallet ]  og hvis a ligger i intervallet ]


[, hvor a1 og a2 er bestemt ved at forlange tangering mellem højre og venstre side af 


ligningen:


fsolve({ligning3,map(diff,ligning3,x)},{a,x=-4});

a1:=subs(%,a);


fsolve({ligning3,map(diff,ligning3,x)},{a,x=3});


a2:=subs(%,a);


plot([sin(x),seq(a*x-2,a=[-3,-2,-1,a1,a2,1,2,3])],x=-2*Pi.


.3*Pi,-2..2,color=[red,blue$8]);


x


0 2 4 6 8


1


2


Vi løser ligningen:


X:=solve(ligning3,x);


Selv om svaret blot ligner en omformulering af spørgsmålet, kan Maple faktisk udmærket regne 
med det i stort omfang:


plot(X,a=-5..a1):


plot(X,a=a2..5):


plots:-display(%,%%);


a


0 2 4


1


2


3

diff(X,a);


Dette differentiationsresultat svarer til det vi får ved brug af implicit differentiation af ligningen:


implicitdiff(ligning3,x,a);


subs(x=X,%);


Taylorpolynomiet af orden 2 med udviklingspunkt 1 kan også bestemmes:


P2:=mtaylor(X,a=1,3);


evalf(%);


plot([X,P2],a=a2..2.3,legend=["X","P2"]);


X P2


a


2


3


Systemer af (ulineære) ligninger. Eksakt og numerisk løsning


Maple løser også et sæt af ligninger med flere ubekendte. Er ligningssystemet lineært og 
oprindeligt givet på matrixform, bør man bruge lineær algebra. Se kapitel 7. Vi giver her et 
ulineært eksempel, som vi forsøger at løse eksakt ved hjælp af solve.


lign1:={ 9*x^2+4*y^2=72 , 2*y+z*x=0 , 9*x+2*z*y=0 };

L:=solve(lign1,{x,y,z});


Maple har fundet fire løsninger. Hvis løsningerne ikke ser så pæne ud, som de gjorde i dette 
eksempel, kan Maple finde på at bruge sin RootOf-notation, som i følgende modificerede version 
af eksemplet ovenfor:


lign2:={ 9*x^2+4*y^2=36 , 2*y+z*x=0 , 9*x+2*z*y=0 };


L:=solve(lign2,{x,y,z});


allvalues({L}); 


Se hjælpen om allvalues. Ønsker man i størst muligt omfang at undgå RootOf-udtryk kan man 
sætte _EnvExplicit til true:


_EnvExplicit:=true:


solve(lign2,{x,y,z});


Vi sætter _EnvExplicit tilbage til ikke at have en tilordnet værdi:


_EnvExplicit:='_EnvExplicit':


Sætter vi i stedet _EnvExplicit til værdien false kan vi få endnu flere RootOf'er end uden en 
værdi.


Den numeriske ligningsløser fsolve løser også ligningssystemer. Vi giver tre varianter af 
syntaksen:


lign3:={2*x+y+ln(x+y^2)=0,3*x^2-y^4+sin(Pi*(x+y))=0};


fsolve(lign3,{x,y});


Med startgæt:

fsolve(lign3,{x=0,y=1});


Med søgeintervaller:


fsolve(lign3,{x=-1..0,y=1..2});


Det er naturligvis en god idé at bruge implicitplot først, så man kan se, hvor der er 
skæringspunkter:


plots[implicitplot](lign3,x=-1..1,y=-1..2);


x


0 1


y


1


Også her kan bruges avoid:


fsolve(lign3,{x,y});


fsolve(lign3,{x,y},avoid=%);


Løsning af uligheder


Maple løser også visse typer uligheder, samt uligheder og ligninger kombineret:


ligning1:=3*x^3+x^2+x+3=0;


solve(ligning1,x);


solve({ligning1,x<0},x);

At der bruges et ulighedstegn er nok til at sikre, at der kun kommer reelle løsninger:


solve({ligning1,x<100},x);


solve({(x-1)*(x-2)*(x-3)*(x-4)=0, x>=2,x<4},x);


solve( (x-1)*(x-2)*(x-3)*(x-4)<0,x);


solve(ln(x)>x-2,x);


Løsning af identiteter


Hvis det vides, at en vis formel gælder for alle x, f.eks. , hvor a er en 


konstant, så kan Maple finde a.
En sådan formel kaldes også en identitet:


solve( identity( sin(2*x)=a*sin(x)*cos(x),x ), {a});


Et andet eksempel: Vi ønsker at skrive udtrykket


udtryk:=3*sin(x)+4*cos(x);


på formen :


form:=A*sin(x+phi);


solve( identity( udtryk=form, x), {A,phi});


eval(form,%[1]);


Kontrol:


expand(udtryk-%);
0


RootFinding-pakken
RootFinding-pakken er en numerisk pakke.

with(RootFinding);


Analytic finder komplekse nulpunkter for en analytisk funktion. Input kan også gives i form af en
ligning, hvor den omtalte funktion så blot er givet ved forskellen mellem højre og venstre side.
Vi afprøver først Analytic på en ligning, hvis løsninger let kan findes ved håndkraft, nemlig  


. Vi søger de løsninger, der ligger i det akseparallelle rektangel, hvis ene diagonal forbinder 


-2-7i og 2+7i:   


Analytic(exp(z)=5,z=-2-7*I..2+7*I);


Vha. identify kan vi forsøge at identificere disse decimaltal:


identify([evalf(%)]);


evalf er her brugt for at reducere antal decimaler, så identify ikke generes af afrundingsfejl. 


convert(%,ln);


Kontrol af løsningerne:


map(exp,%);


simplify(%);


Vi afprøver et eksempel, som vi også brugte ovenfor med fsolve. 


infolevel[Analytic]:=1:


Analytic(2*sin(z)=z,z=-2-I..2+I);


f = 2*sin(z)-z, re = -2. .. 2., im = -1. .. 1..


Homotopy finder løsninger til n polynome ligninger med n ubekendte på formen poly = 0, men 
kun polynomierne skal angives (selv om hjælpen siger noget andet). 
Vi tager som eksempel to polynomier med to variable. Først tegner vi vha. implicitplot de to 
kurver, der er givet ved poly = 0:


plots[implicitplot]([20*x^2*y-1=0, x^2+2*y^2+x*y/2-1=0],x=


-2..2,y=-1..1,


color=[red,blue],linestyle=[1,2]);

x


1 2


y


1


p1:=%:


Homotopy([20*x^2*y-1, x^2 + 2*y^2+x*y/2-1]);


simplify(%);


Reelle løsninger:


L:=remove(has,%,I);


Vi laver en liste bestående af koordinaterne [x, y] for de 4 punkter:


pkt:=map(subs,L,[x,y]);


p2:=plot(pkt,style=point,symbol=circle,color=blue,


symbolsize=20):


plots[display](p1,p2);

x


1 2


y


1


Har vi netop 2 polynomier i 2 variable kan BivariatePolynomial også bruges:


BivariatePolynomial([20*x^2*y-1, x^2 + 2*y^2+x*y/2-1],[x,y])


;


Vi fjerner afrundingsfejl med fnormal og derefter fjerner simplify den imaginære del af tal af 
formen  :


simplify(fnormal([%]));


Nu fjerner vi imaginære løsninger:


remove(has,%,I);


fsolve kan også løse de to ligninger, men finder kun en løsning af gangen. Ved brug af avoid og 
startgæt kan resten findes som vist ovenfor ved lidt programmering: 


S1:={}:


for k to 5 while not has(S1,fsolve) do 


S:=S1; 


S1:={fsolve({20*x^2*y-1=0, x^2 + 2*y^2+x*y/2-1=0},{x=-1.5.


.1.5,y=0..1},avoid=S)} union S 


end do:


S;

Andre ligningsløsere: rsolve, isolve, msolve
Til løsning af differensligninger bruges rsolve:


restart;


ligning:=x(k)+x(k-1)+4*x(k-2)+4*x(k-3)=0;


rsolve(ligning,x(k));


rsolve({ligning,x(0)=1,x(1)=0,x(2)=0},x(k));


Til at finde heltallige løsning i ligninger bruges isolve.
Vi afprøver på en gammel traver:


ligning:=x^2+y^2=z^2;


isolve(ligning);


Vi kan her sætte _Z1, _Z2 og _Z3 til vilkårlige heltallige værdier. ( igcd betyder integer greatest 
common divisor).
Vi tager nogle tilfældige tal (_Z1, _Z2, _Z3 er globale variable):


subs(_Z1=7,_Z2=9,_Z3=13,%);


%;


subs(%,ligning);


Til løsning af ligninger modulo et helt tal bruges msolve.
Vi bruger her samme ligning som ovenfor:


msolve(ligning,2);







Kapitel 5: Løsning af Ligninger


Kapitel 5 side 17


map(subs,[%],ligning);


% mod 2;


map(lhs-rhs,%);


msolve(ligning,3);


Kontrol af løsningerne


map(subs,[%],ligning);


% mod 3;


map(lhs-rhs,%);


Andre lignende værktøjer
?isolate


?eliminate


?realroot


?sturm


?roots

6. Løsning af Differentialligninger


Maple kan løse ordinære differentialligninger, såvel som systemer af differentialligninger, symbolsk 
eller numerisk. 


I en differentialligning, hvor den ubekendte funktion hedder x og den uafhængige variable hedder t, er
der tradition for at udelade t'et i x(t) og kun skrive x, som i følgende eksempel: 


 


I Maple skal altid skrives x(t).  Ovenstående differentialligning skrives derfor således i Maple


diff( x(t), t) +2*x(t)/(1+t) = sin(t)
 


Eksakt løsning af differentialligning


Differentialligningsløseren hedder dsolve. Vi giver tre eksempler:


Eksempel 1


Vi betragter en lineær differentialligning af første orden, som vi let selv kan løse, men her 
bruger vi Maple:


ligning1:=diff( x(t),t) +2*x(t)/(1+t)=sin(t);


Den fuldstændige løsning findes således:


dsolve(ligning1); 


Det kan være ønskeligt at smukkesere lidt på resultatet. I dette tilfælde samles led med sin og
cos, og koefficienterne hertil faktoriseres:


collect(%,[sin,cos],factor); 


I ganske særlige tilfælde kan det være nødvendigt at give Maple oplysning om den uafhængige
variable t og den afhængige variable x gennem tilføjelsen x(t). Det er dog altid tilladt:

dsolve(ligning1,x(t)); 


Ved tilføjelsen useInt bruges inaktiv integration (altså Int). Derved fås et mellemresultat:


res:=dsolve(ligning1,useInt); 


Man ser Panserformlen i brug. Udregningen af integralerne sker ved value:


value(res);


Vil man kun have integralerne indeholdt i eksponentialfunktionerne regnet ud, kan man bruge
evalindets (eller subsindets), der her bruger value på operander af typen specfunc(anything, 
exp):  


evalindets(res,specfunc(anything,exp),value);


Herefter kan vi igen finde det endelige resultat:


value(%);


Er der foruden differentialligningen også givet en begyndelsesbetingelse, eksempelvis 


,  så gøres som følger:


L:=dsolve( {ligning1, x(Pi/2)=2 }); 


Også her kan man bruge useInt: 


dsolve( {ligning1, x(Pi/2)=2},useInt); 

Vi smukkeserer løsningen:


Lsmuk:=collect(L,[sin,cos],factor);


Løsningen kan (og bør!) kontrolleres. Her sker det ved indsættelse i ligningen vha. subs:


subs( Lsmuk, ligning1);


simplify(%);


Kontrol kan også ske vha. proceduren odetest. Kommer der et nul som output, er løsningen 
korrekt (ifølge Maple):


odetest(L,ligning1); 
0


DEtools-pakken indeholder en procedure odeadvisor, der kan fortælle til hvilken type den 
givne differentialligning hører:


with(DEtools): 


odeadvisor(ligning1);


En hjælpeside fås således:


odeadvisor(ligning1,help); 


Løsningen til differentialligningen gives af Maple i form af en ligning, enten (som ovenfor) på 
formen x(t)=konkret udtryk i t (dette kaldes en eksplicit representation) eller på formen F(t, x(t)
)=0 (dette kaldes en implicit repræsentation). 

Er der givet en begyndelsesbetingelse, vil Maple altid forsøge at finde en eksplicit 
repræsentation. Hvis dette mislykkes, er outputtet NULL (dvs. intet). Vi så ovenfor, at man 
også kan få en eksplicit repræsentation, selv om ingen begyndelsesbetingelse er stillet. Maple 
vil forsøge at isolere den ubekendte funktion, men hvis dette f.eks. vil resultere i løsning af en 


ligning af formen , hvor r er en brøk, så vil den give resultatet på implicit form. 


Man kan få Maple til (at prøve) at finde en eksplicit (eller en implicit) repræsentation ved at 
tilføje ordet explicit (eller implicit) som et ekstra argument til dsolve-proceduren (se 
Eksempel 2 nedenfor).


Ønskes løsningen defineret som en funktion kaldet f, kan man gøre ét af to: 
(1) Man kan lave højresiden af Lsmuk til en funktion af  t  vha. unapply:


f:=unapply( rhs(Lsmuk), t); 


Lad os afprøve funktionen:


f(Pi/2);


simplify(%);
2


(2) Den anden metode til definition af f ser lidt mere uigennemskuelig ud, men idéen er god at 
bide mærke i:


f:=unapply( subs(Lsmuk,x(t)), t);


Vi skal senere se, at en lignende syntaks er nyttig ved numerisk løsning af 
differentialligninger.


Eksempel 2


Det første eksempel var en lineær differentialligning. Vi tager nu en simpel ulineær 
differentialligning. Betragt for t > 0 følgende differentialligning:


ligning2:=diff(x(t),t)=3/2/t*(x(t)-x(t)^(1/3));


imp:=dsolve(ligning2);

odeadvisor(ligning2);


odetest(imp,ligning2);
0


Vi ser, at den fuldstændige løsning til denne separable differentialligning blev givet på implicit
form, og at odetest kan kontrollere implicit givne løsninger. Bemærk i øvrigt, at løsningen x(t) 
= 0 (for alle t) ikke blev fundet. Ved tilføjelsen explicit fås løsningerne på eksplicit form:


eksp:=dsolve(ligning2,  explicit);


I dette tilfælde er det dog så som så med det eksplicitte.
Løses et begyndelsesværdiproblem kommer løsningen altid på eksplicit form - men evt. ved 
brug af RootOf-notationen.


dsolve({ligning2, x(1)=8});


I Maple 13 er resultatet en følge af 3: Den første er absurd, den næste er den korrekte 
(begyndelsesværdiproblemet har kun én løsning), den sidste er imaginær og opfylder ikke 
differentialligningen, men dog begyndelsesbetingelsen.
I Maple 12 var resultatet korrekt.
 


Eksempel 3


Som vores næste eksempel tager vi en lineær andenordens differentialligning.


ligning3:=t*diff(x(t),t,t) - (2*t+1)*diff(x(t),t) + (t+1)*


x(t) = 2*t^2*exp(t);


Den fuldstændige løsning fås ved dsolve som ovenfor:


dsolve(ligning3);


hvor _C1 og _C2 er arbitrære konstanter. Ønsker vi den partikulære løsning, der opfylder 
begyndelsesbetingelserne  x(1) = e  og  x'(1) = e  skriver vi: 


q:=dsolve({ligning3, x(1) = exp(1) , D(x)(1) = exp(1)});

idet vi erindrer, at exp(1) skal bruges for e. Løsningen som funktion:


f:=unapply( subs(q,x(t)), t);


Løsning af randværdiproblemer lader sig også gøre:


dsolve({ligning3, x(1) = exp(1) , x(3) = 0});


Løsning ved rækkeudvikling


En rækkeløsning af differentialligningen kommer specielt på tale, når Maple ikke kan løse 
ligningen eksakt. Vi giver her et eksempel, hvor Maple i realiteten må opgive at finde en eksakt 
løsning:


ligning4:=diff(x(t),t)=sqrt(1+x(t)^3)*sin(t);


dsolve({ligning4, x(Pi/2)=1});


Resultatet af kommandoen


has(%,Int);
true


viser, at integralet er et inaktivt integral (et Int-integral). Vi kan bede om at få det regnet ud:


res:=value(%%);

Vi er selvfølgelig ikke sluppet fri af RootOf, så vi er egentlig lige vidt. Vi finder derfor en 


rækkeløsning udfra begyndelsestidspunktet, altså her . Dette kan gøres direkte ud fra 


differentialligningen:


dsolve({ligning4, x(Pi/2)=1},x(t),type=series);


Bemærk her, at dsolve/series forlanger kendskab til den ubekendte gennem tilføjelsen x(t). (Prøv 
at fjerne x(t)).
type = series kan erstattes af blot series og ordenen kan angives på formen order = n (mærkeligt 
nok ikke nævnt i hjælpen for dsolve/series).
Hvis ikke ordenen angives har den værdien af den globale variabel Order, der er 6, når maskinen 
startes, men kan ændres til eksempelvis 13 ved kommandoen Order:=13; 


dsolve({ligning4, x(Pi/2)=1},x(t),series,order=4);


Vi fjerner store-O-leddet:


p3:=convert( rhs(%),polynom);


Som vi nu skal se, kan Maple faktisk godt tegne grafen af den eksakte løsning selv om denne er 
givet ved et RootOF-udtryk. Beregningerne nødvendige for plottet tager dog lang tid (nogle få 
minutter). Vi tegner grafen sammen med grafen for det 3. Taylorpolynomium:


plot([rhs(res),p3],t=0..Pi,title="Løsningen og dens 3.


Taylorpolynomium",


legend=["Eksakt løsning","3. Taylorpolynomium"],linestyle=


[1,3]);

Løsningen og dens 3.Taylorpolynomium


Taylorpolynomier for implicit given funktion


Som en anvendelse af dsolve/series vil vi vise en metode til bestemmelse af Taylorpolynomier 
for en implicit given funktion af én variabel.
Lad funktionen være givet implicit ved ligningen 
 


ligning:=x^2*y-sin(x+y)=0;


Idéen er, at bruge implicitdiff til at udtrykke  ved x og y. Herved opnås en 


differentialligning, som dsolve/series så kan anvendes på:


difflign:=diff(y(x),x) = subs(y=y(x), implicitdiff


(ligning,y,x));


Punktet (x, y) = ( ) opfylder ligningen ligning:


eval(ligning,{x=0,y=Pi});


 Følgende kommando giver så Taylorudviklingen med udviklingspunkt 0:


dsolve({difflign,y(0)=Pi},y(x),type=series, order=5);


Taylorpolynomiet  fås nu således:


convert(rhs(%), polynom);

Numerisk løsning af differentialligning


Maple kan løse både begyndelsesværdiproblemer og randværdiproblemer numerisk.
Hvis vi ønsker en numerisk beregning af løsningen, tilføjer vi type= numeric (eller blot numeric) 
som et ekstra argument i dsolve-kommandoen. 
Vi tager som første eksempel begyndelsesværdiproblemet bestående af differentialligningen 


ligning4 med  . 


ligning4;


Maple's svar kan fås på 5 forskellige former, enten som en listprocedure, procedurelist, operator, 
piecewise eller en Array/array. 
Vi betragter kun 3 af de 5: listprocedure, procedurelist (default) og Array/array.


output=listprocedure


Først tager vi den, der er lettest at arbejde med:


Lp:=dsolve({ligning4, x(Pi/2) =1} , type=numeric, output=


listprocedure);


Lp er hermed en liste bestående af procedurer (hér 2). Det er den sidste af disse, der er vores 
numeriske løsning. Den fås meget elegant som en funktion (en procedure) som følger:


f:=subs(Lp, x(t)); 


Afprøvning:


f(2);
1.78697734035338840


Procedurerne i output-listen fra dsolve/numeric/listprocedure returnerer uevalueret, hvis 
input ikke er et tal. Det gør, at vores procedure f returnerer uevalueret i følgende to tilfælde:


f(h),f(t);


Dette er en afgjort fordel bl.a. ved plotning, idet vi da er fri til at benytte både syntaksen plot( f
(t), t=a..b); og den variabelfri: plot( f, a..b);


plot(f(t),t=0..3*Pi,caption=typeset("Numerisk løsning af 


begyndelsesværdiproblemet \n",ligning4,"  med ", x(Pi/2) =


1));

Numerisk løsning af begyndelsesværdiproblemet 


  med 


Der findes i plots-pakken en særlig procedure indrettet specielt på plotning af resultatet af
dsolve, numeric. Den kaldes odeplot. Den forudsætter ikke, at man først har defineret 
løsningen som en funktion:


with(plots): 


odeplot(Lp,[t,x(t)],t=0..3*Pi,caption=typeset("Numerisk 


løsning af begyndelsesværdiproblemet \n",ligning4,"  med 


", x(Pi/2) =1));    

Numerisk løsning af begyndelsesværdiproblemet 


  med 


odeplot kan også plotte funktioner af løsningen. Her plotter vi højresiden af 
differentialligningen som funktion af t:


odeplot(Lp,[t,sqrt(1+x(t)^3)*sin(t)],t=0..3*Pi,numpoints=


100,caption=typeset("Højresiden ",rhs(ligning4),"  som jo 


er ", diff(x(t),t)));

Højresiden   som jo er 


Tilføjes et valgfrit argument af formen frames=positivt helt tal fås en animation som i følgende
eksempel:


odeplot(Lp,[t,x(t)],t=0..5*Pi,frames=50,numpoints=200);


t

output=procedurelist


Hvis man ikke skriver noget om output, så har man valgt output=procedurelist (som altså er
default-værdien):


pL:=dsolve({ligning4, x(Pi/2) = 1} ,  type=numeric); 


derved bliver der dannet en procedure pL, der for enhver værdi af  t giver en liste: pL er hermed
i Maple's sprog en procedurelist:


pL(2);

altså er x(2) = 1.78697749374652548. 


Som ved output=listprocedure kan løsningen plottes ved brug af odeplot fra plots-pakken 
(som vi gjorde tilgængelig ovenfor):


odeplot(pL,[t,x(t)],t=0..3*Pi); # Ikke vist


Som for output=listprocedure, returnerer output-proceduren fra 
dsolve/numeric/procedurelist uevalueret, hvis input ikke er et tal. Det gør, at output-formen
output=procedurelist i visse tilfælde er endnu mere besværlig, end den var før. Desværre er 
den default-metoden. Problemet er her, at pL(2) er en liste, hvorimod pL(t) er et uevalueret 
funktionskald (type function i Maple):


pL(t);


pL(2);


Vi anbefaler derfor output-formen output=listprocedure. 


output=Array/array


Ønsker vi en tabel med funktionsværdier x(t) svarende til en given liste af t-værdier, kan vi 
bruge output = Array (eller array).
Lad os sige, at vi ønsker løsningens værdier udregnet i følgende liste af t-værdier:


T:=[seq(Pi/4*k,k=2..8)];


Så er syntaksen følgende:


A:=dsolve({ligning4, x(Pi/2) = 1} ,  type=numeric, output=


Array(T)); 


Resultatet er en 2x1-matrix, hvis (2,1)-element er matricen


A[2,1];

En liste af sammenhørende (t, x) - værdier kan nu fås ved en convert-kommando:


L:=convert(%,listlist);


plot(L,style=point,symbolsize=25,labels=["t","x"],


tickmarks = [spacing(Pi/4), default],gridlines=true);


t

Eksempel på numerisk løsning af randværdiproblem


Betragt følgende andenordensdifferentialligning:


ligning5:=diff(x(t),t,t)+t*x(t)=sin(t);


Vi ønsker at bestemme den løsning, der opfylder randbetingelserne . 


Vi prøver først eksakt løsning:
 


res:=dsolve({ligning5, x(0)=0, x(Pi)=1});

Det gik godt selv om resultatet indeholder uudregnede integraler indeholdende Airy-funktioner.
 
Vi prøver også numerisk:


L:=dsolve({ligning5, x(0)=0, x(Pi)=1},type=numeric,output=


listprocedure);


f:=subs(L,x(t));


display(


plot(f,0..Pi),


plot([[0,0],[Pi,1]],style=point,symbol=circle,symbolsize=


20,color=blue)


,caption=typeset("Numerisk løsning af randværdiproblemet 


\n",ligning5,"  med ",x(0)=0," og ",x(Pi)=1));

Numerisk løsning af randværdiproblemet 


  med  og 


Bemærk, at man kun kan få beregnet f(x)  for værdier af x i intervallet mellem de to 


randpunkter (hér 0 og ).


Den eksakte løsning, der blev fundet ovenfor kan selvfølgelig plottes også på større intervaller:


plot(rhs(res),t=0..2*Pi,caption=typeset("Eksakt løsning af


randværdiproblemet \n",ligning5,"  med ",x(0)=0," og ",x


(Pi)=1));


t

Eksakt løsning af randværdiproblemet 


  med  og 


Metodevalg ved numerisk løsning


Maple bruger ved begyndelsesværdiproblemer en såkaldt Fehlberg 4.-5.-ordens Runge-Kutta-
metode (rkf45), hvis man ikke instruerer den om andet. Det kan man imidlertid gøre. 
Vi viser her to metoder i sving, dels den allersimpleste metode, nemlig Eulers metode, dels den 
meget brugte Runge-Kutta-metode af 4. orden. Disse metoder hører til de metoder, som Maple 
kalder klassiske. Så vi skal skrive method=classical. Hvis vi ikke skriver andet, så fås Eulers 
metode. Vil vi have den klassiske Runge-Kutta-metode af 4. orden, så skriver vi method=classical
[rk4]. Eulers metode fås også ved at skrive method=classical[foreuler]    (for forward Euler ).
Mere information fås ved at bruge hjælpen:


?dsolve,numeric

Eksempel på brug af method=classical


ligning:=diff(x(t),t)=1/(t+x(t)+3);


Vi løser først ligningen med x(1) = 2 ved brug af Eulers metode med skridtlængde 0.25 (ret 
stor!):


L_euler:=dsolve({ligning, x(1) = 2} , type=numeric,output=


listprocedure,method=classical,stepsize=0.25);


f_euler:=subs(L_euler,x(t)); #Løsningen som funktion


Derefter løses ligningen ved brug af en Runge-Kutta metode af 4.orden og med samme 
skridtlængde:


L_rk4:=dsolve({ligning, x(1) = 2} , type=numeric,output=


listprocedure,method=classical[rk4],stepsize=0.25);


f_rk4:=subs(L_rk4,x(t));


Til sammenligning udregnes dels, hvad disse to metoder giver, dels hvad eksakt løsning giver 


for : 


f_euler(3),f_rk4(3);


Den eksakte løsning er


dsolve({ligning, x(1) = 2});


g:=unapply(subs(%,x(t)),t);


evalf[18](g(3));
2.28218704267880438


Et logaritmisk plot af fejlen ved de to metoder viser tydeligt, at Runge-Kutta er Euler langt 
overlegen:


logplot([f_euler-g,f_rk4-g],1..3,color=[red,blue],legend=


["euler - eksakt","rk4 - eksakt"]);

euler - eksakt rk4 - eksakt

Den eksakte løsnings graf vist sammen med Euler-løsningen for forskellige trinstørrelser i 
form af en animation:


baggrund:=plot(g,1..3,color=blue):


p:=s->odeplot(dsolve({ligning, x(1) = 2} , type=numeric,


output=listprocedure,method=classical,stepsize=s),


[t,x(t)],1..3):


animate(p,[s],s=1..0.1,background=baggrund);


t

Systemer af differentialligninger 
Maple kan også løse et sæt af sammenhørende differentialligninger. 
Vi begynder med et simpelt lineært og homogent system med konstante koefficienter


Lineært og homogent system med konstante koefficienter


Er A en kvadratisk matrix med konstante koefficienter, kan systemet løses 


ved hjælp af MatrixExponential  fra LinearAlgebra-pakken. Kommandoen

MatrixExponential(A,t) giver  altså .


A:=Matrix([[-5, -4, 0], [3, 2, 0], [-8, -12, -3]]);


with(LinearAlgebra):


F:=MatrixExponential(A,t);


Løsningen med begyndelsesværdivektoren x(0) = <2, -1, 3 > fås da således:


X:=F.<2,-1,3>;


plot(X,t=0..5);


t

Opskrives de 3 differentialligninger hver for sig, kan systemet også løses ved brug af dsolve:


X:=<seq(x[k](t),k=1..3)>;


map(diff,X,t)=A.X;

sys:={seq(lhs(%)[k]=rhs(%)[k],k=1..3)};


Systemet kan også laves vha. elementvis differentiation diff~ og elementvis =~ således


diff~(X,t) =~ A.X;


convert(%,set);


dsolve(sys union {x[1](0)=2,x[2](0)=-1,x[3](0)=3});


Løsningen på vektorform:


X=subs(%,X);


Brusselatoren: Et ulineært eksempel


Brusselatoren er en simpel matematisk model for et svingende kemisk system. Den er opstillet 
af Prigogine og Lefever i 1968.
Der foregår følgende reaktioner i en omrørt tank uden gennemstrømning:


Det antages, at A og B er tilstede i så stor mængde, at deres koncentrationer kan antages at 
være konstante. Med passende valg af enheder kan systemets udvikling i tiden t da beskrives 

ved følgende to differentialligninger i koncentrationerne af X og Y ( x(t) og  y(t) ).
Differentialligningerne opbevarer vi i variablen brussel:


brussel:={diff(x(t),t)=1-(beta+1)*x(t)+alpha*x(t)^2*y(t), 


diff(y(t),t)=beta*x(t)-alpha*x(t)^2*y(t)};


Maple kan ikke løse dette system eksakt, selv om kommandoen dsolve(brussel); faktisk giver 
et resultat. Dette er dog slet ikke konkret nok til vi kan bruge det til noget. Vi prøver numerisk 


løsning. Så må de to konstanter  og  gives konkrete værdier, og vi må opgive et sæt 


begyndelsesbetingelser. Vi tager som illustration ,   og  x(0)=1 og y(0)=1:


brussel_konkret:=subs(alpha=1,beta=2.5,brussel);


begynd:={x(0)=1,y(0)=1};


Lb := dsolve(brussel_konkret union begynd, type=numeric,


output=listprocedure);


Vi definerer løsningerne som to funktioner (procedurer):


f:=subs(Lb,x(t));  g:=subs(Lb,y(t));


f(3),g(3);


Banekurven med parameterfremstillingen  (x,y) = ( f(t), g(t) ) over t-intervallet [0,10]:


plot( [f, g, 0..20] , view=[0..3, 0..4], 


   title= "Brusselatoren i faseplanen",labels=["x","y"]);

Brusselatoren i faseplanen


Graferne for f og g kan tegnes i samme koordinatsystem. På tegningen vil vi dog hellere kalde 
dem x og y:


tp:=textplot([[4,.9,"x"],[10,2.9,"y"]]):


fgp:=plot( [f, g], 0..20 , 0..4, 


           title= "Graferne for f og g"):


display(tp,fgp);

Graferne for f og g


I stedet for det lidt omstændelige textplot, kan man benytte tilføjelsen legend=["x","y"], der 
anbringer en signaturforklaring under figuren.


odeplot kan også tegne banekurven:


odeplot(Lb, [x(t),y(t)], 0..20,numpoints=200); # Ikke vist


og også graferne for x og y (som vi ovenfor kaldte f og g):


odeplot(Lb, [[t,x(t)],[t,y(t)]], 0..20, legend=["x","y"]);







Kapitel 6: Løsning af Differentialligninger


Kapitel 6 side 22


x y


t


0 5 10 15 20


x, y


1


2


3


En animering af banekurven med tiden t som animationsparameter kan fås direkte i odeplot 
således:


odeplot(Lb, [x(t),y(t)], 0..20,numpoints=200,frames=50,


color=blue); # Ikke vist


Vi kan gemme filmen til senere brug:


banefilm:=%:


Nedenfor produceres en animation af banekurven i faseplanen med  som 


animationsparameter og med .


Først defineres p som en funktion (procedure), der for et givet input a producerer et plot af 


banekurven med . Derefter bruges animate til at animere plotteproceduren p. 


Bemærk, at begyndelsesværdierne er givet for t = 0, men at banerne kun tegnes for t = 88 .. 
100. 
Det der tegnes er for hver værdi af a  tilnærmelsesvis den lukkede kurve (en grænsecykel), som
banerne konvergerer imod.


p:=a-> odeplot(


dsolve(subs(alpha=a,beta=2.5,brussel) union begynd, type=


numeric,output=listprocedure,range=0..100),


[x(t),y(t)],t=88..100,refine=1):


animate(p,[a],a=0.5..1.5);

Lorenz-attraktoren: Et 3-dimensionalt eksempel med kaos
Systemet hvor parametrene antages positive.
Først højresiderne:


f:=(x,y,z)->sigma*(y-x);


g:=(x,y,z)->rho*x-y-x*z;


h:=(x,y,z)->-beta*z+x*y;


lorenz:={diff(x(t),t)=f(x(t),y(t),z(t)),diff(y(t),t)=g(x


(t),y(t),z(t)),diff(z(t),t)=h(x(t),y(t),z(t))};


Ligevægtspunkter:


_EnvExplicit:=true:


solve({f(x,y,z)=0,g(x,y,z)=0,h(x,y,z)=0},{x,y,z});


pkt:=map(subs,[%],[x,y,z]);


Vi vælger følgende konkrete parameterværdier til brug ved illustrationen nedenfor.


param:={sigma=3,beta=1,rho=30};


Begyndelsesbetingelser:


begbet:={x(0)=0,y(0)=5,z(0)=0}:

Vi løser systemet numerisk:


R:=0..100:


L:=dsolve(subs(param,lorenz) union begbet,{x(t),y(t),z(t)}


,type=numeric,output=listprocedure,range=R);


F:=subs(L,[x(t),y(t),z(t)]);


with(plots): 


lp:=pointplot3d(subs(param,pkt),symbol=solidsphere,


symbolsize=20,color=black):


spacecurve(F,R,numpoints=4000,axes=boxed,labels=["x","y",


"z"]):


display(%,lp);


En animation:


odeplot(L,[x(t),y(t),z(t)],R,numpoints=4000,frames=100,


axes=boxed):


display(%,lp);

En lidt sjovere animation:


T:='T':


baggrund:=display(odeplot(L,[x(t),y(t),z(t)],0..25,


numpoints=1000,axes=boxed,color=blue),lp):


animate(odeplot,[L,[x(t),y(t),z(t)],T-.1..T,axes=boxed,


thickness=3],T=0.1..25,frames=200,background=baggrund);


Egentlig burde det røde tog have været et punkt lavet med pointplot3d. Dette kræver imidlertid
en udbedring af en skavank i animate. 
Denne udbedring sker i de næste par linier:


EVAL:=(x,y)->eval(y,x):


minanimate:=subs(subs=EVAL,eval(`plots/animate`)):


minanimate(pointplot3d,[[F(T)],symbol=solidsphere,


symbolsize=25,color=red,axes=boxed],T=.1..25,frames=200,


background=baggrund);

DEtools-pakken


Pakken DEtools indeholder forskelligt af interesse. Vi tager igen Brusselatoren som eksempel:


with(DEtools):


Vi minder om systemet:


brussel_konkret;


Proceduren DEplot tegner løsninger, når den får differentialligningerne og 
begyndelsesbetingelserne. Begyndelsesbetingelserne gives i en liste (eller mængde) af lister, enten
på formen [ [x(t0) = x0, y(t0) = y0], [x(t1) = y1, y(t1) = y1], ....] eller på formen [ [t0, x0, y0], [t1, 
x1, y1],.....]. I eksemplet nedenfor tegnes den tidligere sete banekurve plus én mere. Desuden 
tegnes retningsfeltet med mindre man tilføjer arrows=none:


DEplot( brussel_konkret, [x,y], t=0..20, [ [x(0)=1,y(0)=1], 


[x(0)=1,y(0)=2.4] ], x=0..3,y=0..4, stepsize=0.05,linecolor=


[blue,maroon]);

Vil man kun have retningsfeltet, kan man fjerne begyndelsesværdilisten:


DEplot( brussel_konkret, [x,y], t=0..20, x=0..3,y=0..4,


arrows=large,color=grey);


Vi kan evt. vise den animation af banekurven, som vi lavede tidligere, sammen med retningsfeltet:


display(banefilm,%); #Ikke vist


Beder man om hjælp til dsolve får man en fortrinlig oversigt, der på det varmeste kan anbefales.


?dsolve

7. Lineær Algebra


LinearAlgebra-pakken


Fra og med Release 6 har Maple to pakker med lineær algebra: LinearAlgebra og den gamle
linalg.  
Her vil linalg kun blive omtalt af hensyn til forvekslingsmuligheden: Man bør eksempelvis være 
opmærksom på, at matrix er stavemåden for en matrix i linalg, men i LinearAlgebra skal skrives
Matrix. 
Procedurenavnene i de to pakker kan skelnes fra hinanden ved at der i den nye pakke bruges store 
forbogstaver i navnene, og at der ikke bruges forkortelser. Ved sammensatte ord begynder alle 
delene med stort, eksempelvis JordanBlockMatrix.


with(LinearAlgebra): 


Definition af matrix og vektor


En matrix A kan defineres på følgende måde, hvor man giver Maple en liste af lister, nemlig 
en liste af rækkerne i matricen. Hver række er selv en liste: 


A:=Matrix([[0,1,3],[1,2,-5],[3,1,-4]] );


Man kunne i stedet begynde med at definere søjlevektorerne ved brug af <...>. Her defineres 
første søjlevektor:


s1:= <0,1,3>;


På én gang kan alle tre søjler defineres således:


s1,s2,s3:= <0,1,3>,<1,2,1>,<3,-5,-4>;


Matricen A kunne nu i stedet defineres som matricen bestående af de tre søjlevektorer s1, s2 
og s3, enten ved brug af Matrix, eller ved brug af <...>. Den lodrette streg  |  skal så benyttes 
til at adskille søjlerne:

Matrix([s1,s2,s3])= <s1|s2|s3>;


Ved brug af udelukkende symbolerne <, > og |  kan matricen A defineres direkte som følger, 
hvor vi dog undertrykker output, da vi efterhånden har set matricen tit nok:


<<0,1,3>|<1,2,1>|<3,-5,-4>>:


En søjlevektor kan også defineres ved brug af Vector (eller mere udførligt Vector[column] ) 
således:


Vector([7,9,13]);


Vil man definere en rækkevektor kan man enten bruge tegnene <, > og | således:


<4|0|a|-23>;


eller man kan bruge  Vector[row] således:


Vector[row]([4,0,a,-23]);


Man kan angive en funktion, der fortæller hvordan elementerne i matricen (eller vektoren) skal
regnes ud. Vil man eksempelvis definere en 2x7-matrix i hvilket elementet i positionen (i, j) 


skal være tallet  , så gør man således


Matrix(2,7,(i,j)->j-i^2);


Hvis man ønsker en kvadratisk matrix, eksempelvis en 3x3, behøver man kun give 
rækkeantallet:


Matrix(3,(i,j)->j-i^2);


De første 10 kvadrattal som elementerne i en rækkevektor kan fås således:

Vector[row](10,i->i^2);


En matrix kan også defineres ved at definere dens elementer et for et:


C:=Matrix(2,3);


C[2,3]:=7: C[1,2]:=a:  C;


Eller således, hvor fill- tilføjelsen fortæller, hvad de resterende pladser skal fyldes ud med 
(hvis ikke det er med nul):


C:=Matrix(2,3,{(1,1)=5,(2,3)=13},fill=-14);


En matrix bliver vist, hvis blot den bliver nævnt. Dette forudsætter dog (til at begynde med), at
den ikke er større end 10x10 . Er matricen større end 10x10, så vil matricen som default ikke 
blive vist. Her følger et eksempel på en 11x11-matrix:


B:=Matrix(11,(i,j)->i+j-1);


Prøv nu at højreklikke på output. Vælg Browse. Her kommer nu en Matrix-Browser frem. Man
kan ændre den øvre grænse på 10x10 ved en interface-kommando. Vil man f.eks. ingen grænse
have, kan man gøre således:


interface(rtablesize=infinity);
10


Output fra ovenstående kommando er den tidligere øvre grænse, altså 10. Herefter vil matricen
B blive vist efter kommandoen B; Det vil vi af pladshensyn ikke gøre, men vi sætter i stedet
rtablesize tilbage til 10:


interface(rtablesize=10);


Har man brug for eksempler på matricer, kan man bede Maple om at finde en tilfældig matrix. 
Vi definerer her en 2x5-matrix med elementer valgt som tilfældige hele tal mellem -9 og 9:

C:=RandomMatrix(2,5,generator=-9..9);


Løsning af ligningssystemer


Løsning af lineære ligningssystemer foregår ved hjælp af LinearSolve. 


Antag, at ligningssystemet er givet på formen er Ax = b, hvor A er givet ved:


A:=<0,2,3;1,5,6;2,-1,-3>;


og hvor søjlevektoren b er givet ved


b:= <-3,0,3>;


Systemet af ligninger kan skrives ud ved brug af GenerateEquations:


sys:=GenerateEquations(A,[x1,x2,x3],b);


Systemet kunne så løses vd brug af den generelle procedure solve


solve(sys,[x1,x2,x3]);


Det er imidlertid en omvej. Systemet bør løses ved brug af LinearSolve, der kun som input 
skal have A og b eller i stedet totalmatricen for systemet: 


LinearSolve(A,b);


Systemets totalmatrix kan simplest dannes ud fra A og b ved brug af tegnene <, > og |


T:=<A|b>;

men kan også dannes således


Matrix([A,b]);


Får LinearSolve kun ét argument, så tolkes dette som totalmatricen for et lineært 
ligningssystem med en vektor (den sidste søjle) som højresiden:


LinearSolve(T);


Vil man løse et system af formen  AX = B, hvor B er en matrix (med mere end én søjle), så 
skal man bruge syntaksen med to argumenter:


B:=<b|<3,2,-2>>;


LinearSolve(A,B);


Gauss-elimination


Ønsker vi at følge løsningen af systemet Ax = b via Gauss-elimination gør vi som følger.
Først dannes totalmatricen:


T:=<A|b>;


Resultatet af Gauss-elimination fås således:

GaussianElimination(T);


Resultatet af Gauss-Jordan-elimination (dvs. reduktion til reduceret echelonform):


ReducedRowEchelonForm(T);


Herefter kan den reducerede matrix igen omskrives til det tilsvarende ligningssystem. I vores 
eksempel er det så simpelt, at det egentlig ikke er nødvendigt, men vi gør det alligevel:


GenerateEquations(%,[x1,x2,x3]);


RowOperation udfører de enkelte række- eller søjleoperationer:


Først ombyttes række 1 og 2:


T1:=RowOperation(T,[1,2]);


Dernæst rækkeoperationen  :


T2:=RowOperation(T1,[3,1],-2);


Rækkeoperationen  bringer så matricen på echelonform:


T3:=RowOperation(T2,[3,2],11/2); 


Man kan i stedet bruge Pivot, hvor kommandoen Pivot( A, i, j); skaber nuller over og under (i, 
j)-elementet i den j'te søjle. Ved tilføjelse af et fjerde argument kan man styre i hvilke rækker, 

der skal skabes nuller. 


Pivot(T1,1,1);


Pivot(%,2,2,3);


hvorimod følgende kommando skaber nuller både over og under (2, 2)- elementet:


Pivot(%%,2,2);


Eksempel på ligningssystem med løsning indeholdende en fri parameter


Vi kan billigt lave et eksempel på et ligningssystem med løsning indeholdende en fri 
parameter ved at fjerne eksempelvis første række i A, altså betragte undermatricen:


A1:=A[2..3,1..3];


Lad b1 være højresiden:


b1:=<1,2>;


Vi løser systemet  A1 x = b1. Ved tilføjelsen free = t bestemmer vi, at en (evt.) fri parameter 
kommer til at hedde t med et index:


LinearSolve(A1,b1,free=t);

Mellemregningerne følger. 
Totalmatricen:


T1:=<A1|b1>;


GaussianElimination(T1);


ReducedRowEchelonForm(T1);


Det tilsvarende ligningssystem:


GenerateEquations(%,[x1,x2,x3]);


x3 er åbenbart en fri parameter og vi sætter den til 11t for at undgå brøker i resultatet:


subs( x3=11*t,%);


Dette simple ligningssystem skal nu løses for x1 og x2. Dette kan gøres vha. af solve. Derfor 
gør vi således:


solve(%, {x1,x2});


Endelig kan vi skrive resultatet på vektorform således:


res:=subs(%, x3=11*t, <x1,x2,x3>);


Med lidt programmering (se kapitlet herom) kan man lave sig en procedure, der skriver 
resultatet ovenfor på formen u + t*v, hvor dog det sædvanlige gangetegn `* ` er erstattet af 
`&*` for at forhindre, at Maple automatisk sætter det omhyggeligt opsplittede resultat tilbage 
igen. 
Tegn der begynder med & accepteres af Maple. De fungerer i realiteten blot som skilletegn: 

Idéen i proceduren er følgende.
Med Maple 13 kan man ved tilføjelse af et tilde ~ til funktioner og operatorer (så som `+`, `*`, 
`=`) få disse til at virke elementvis på matricer, vektorer, lister og mængder ("containere").


Her bruger vi coeff~ til at finde koefficienten til  på hver af elementerne i vektoren res:


v:=coeff~(res,t,1);


Hermed er u givet ved


u:=res-t*v;


Den ønskede opsplitning er nu


u+t&*v;


Proceduren kan håndtere flere frie parametre:


ExpandVector:=proc(v::Vector) local d,S,n,w,i,v0;


uses LinearAlgebra;


d:=Dimension(v);


#


#Vi finder navnene på parametrene og deres antal (n):


S:=indets(v,name);


n:=nops(S);


#


#Vi finder koefficientvektorerne w[i] til parametrene:


for i to n do w[i]:=coeff~(v,S[i],1) end do;


(*


Vi bestemmer den vektor, der ingen parameter indeholder


og hvis denne er nulvektoren, sætter vi den til 0, 


så den forsvinder helt i output:


*)


v0:=v-add(S[i]*w[i],i=1..n);


if Equal(v0,ZeroVector(d)) then v0:=0 end if;


#Her kommer så output:


v0+add(S[i]&*w[i],i=1..n);


end proc:

ExpandVector(res);


Det oprindelige resultat fås igen ved at erstatte `&*`med `*`:


eval(%,`&*`=`*`);


Matrixprodukt


Lad A være matricen


A:=Matrix([[0,1,3],[1,2,-5],[3,1,-4]] );


Vi lader B være en matrix med elementer valgt som tilfældige hele tal mellem -9 og 9:


B:=RandomMatrix(3,3,generator=-9..9);


Produktet mellem A og B fås således, hvor prikken er et punktum:


A.B;


Lad b være søjlevektoren:


b:=<1,-2,6>;

Produktet Ab fås ligesom produktet AB vha. punktum:


A.b; 


Ved multiplikation med en scalar bruges det sædvanlige gangetegn *, men punktum kan 
bruges, når skalaren er konkret, altså faktisk er et tal, eksempelvis 7. Så må man dog sørge for,
at der er et mellemrum mellem 7 og punktum for at forhindre fortolkning som decimaltal:


s*A, 7*A;


Uden mellemrum fås en fejlmelding:


7.A;


Error, missing operator or `;`


Nu med mellemrum:


7 .A;


Transponering. Skalarprodukt.
Den transponerede matrix til en matrix A fås således:


Transpose(A);


Matrixproduktet  vil altid være en symmetrisk matrix:


Transpose(A).A;







Kapitel 7: Lineær Algebra


Kapitel 7 side 12


Vi ser, at matricen er symmetrisk.


Skalarproduktet mellem to vektorer kan fås ved brug af DotProduct:


v:=RandomVector[row](3);


w:=RandomVector[row](3);


DotProduct(v,w);


DotProduct ignorerer forskellen mellem række- og søjlevektorer, hvilket jo er rimeligt nok:


u:=RandomVector(3);


DotProduct(u,v);
1694


Skalarproduktet to søjlevektorer eller mellem to rækkevektorer kan fås ved brug af punktum 
ligesom matrixproduktet. 
Her tager vi som eksempel de to rækkevektorer v og w:
Da de to vektorer forstået som 1x3-matricer ikke kan ganges sammen, tillader Maple at v.w 
tolkes som skalarproduktet af v og w:


v.w;


Skalarproduktet kan også fås ved matrixmultiplikation således:


v.Transpose(w);


Bemærk, at i omvendt rækkefølge fås en 3x3-matrix, da vi har produktet af en 3x1- og en 1x3-
matrix:


Transpose(w).v;


Invers matrix

Den inverse til A fås således:


A^(-1);


Man kan også bruge MatrixInverse:


MatrixInverse(A);


Det korteste er selvfølgelig at taste A^(-1), hvilket også kan anbefales. 
For matricer, der ikke har nogen invers, vil MatrixInverse returnere den såkaldte Moore-
Penrose-pseudoinverse matrix, se herom i hjælpen til MatrixInverse.


Kontrol af den inverse består i udregning af , der gerne skulle give enhedsmatricen:


IdentityMatrix(3);


Vi prøver:


A.A^(-1);


Vi fik enhedsmatricen, hvilket viser, at den inverse blev regnet rigtigt ud.


Nulrum for matrix

Lad A være matricen


A:=Matrix([[1,1,-2,1,2],[0,0,2,2,1]]);


En basis for nulrummet for:


NA:=NullSpace(A);


Kontrol af at de 3 vektorer tilhører nulrummet kan gøres ved udregning af Av for hver vektor i
NA. Dette kan evt gøres i et hug vha. map, der her bruger funktionen  på hver vektor i 
NA:


res:=map( v->A.v, NA);


Bemærk, at der i mængden ovenfor er 3 kopier af nulvektoren. Her er matrix-datastrukturen 
lidt speciel, idet vi jo eksempelvis har


{a,a,a};


Resultatet af følgende er 3 gange true:


map(v->Equal(v,<0,0>),res);


Ved "håndkraft" kan en basis for nulrummet findes således:
Vi danner først totalmatricen for det homogene system Ax = 0:


T:=<A|<0,0>>;


Vi beder om reduceret echelonform:


RRA:=ReducedRowEchelonForm(T);


Frie variable er x2, x4 og x5.

Vi nedskriver det tilsvarende ligningssystem:


GenerateEquations(RRA,[x[1],x[2],x[3],x[4],x[5]]);


Vi løser for de basale variable vha. solve: 


solve(%,{x[1],x[3]});


Her fås så løsningerne til Ax = 0 udtrykt ved de frie variable x2, x4 og x5:


subs(%,<x[1],x[2],x[3],x[4],x[5]>);


Ved brug af vores hjemmelavede procedure ExpandVector fås


ExpandVector(%);


En basis for nulrummet udgøres nu af de 3 vektorer, der står klinet op ad . 


Basisvektorerne kan pilles ud herfra ved følgende kommando, om man ønsker:


indets(%,Vector);


LinearSolve vil producere en anden (men ligeså god) basis: 


LinearSolve(A,<0,0>,free=x):

ExpandVector(%);


Basisvektorerne kan pilles ud som ovenfor:


NA2:=indets(%,Vector);


At de to baser er lige gode kan kontrolleres ved at vise, at vektorerne i den ene kan skrives 
som en linearkombination af vektorerne i den anden og omvendt. Og dette kan evt. gøres 
således:


V:=<NA[1]|NA[2]|NA[3]>:


U:=<NA2[1]|NA2[2]|NA2[3]>:


Vi løser matrixligningen  VX = U for X, hvorved vi undersøger om vektorerne i NA2 kan 
skrives som en linearkombination af vektorerne i NA:


ReducedRowEchelonForm(<V|U>);


X:=%[1..3,4..6];


Der var altså en løsning, og vi bemærker, at X er invertibel, så VX = U medfører  . 
Altså kan vektorerne i NA også skrives som en linearkombination af vektorerne i NA2.

Equal og Copy


Man kan med rette undre sig over, at resultatet res af kontrollen af resultatet fra NullSpace 
ovenfor ikke resulterede i mængden af en enkelt nulvektor, som tilfældet normalt er med 
dubletter i mængder:


{ a, a, 7,9,0,0,0};


Sagen er, at de 3 vektorer opbevares 3 forskellige steder i maskinens hukommelse:


map(addressof,res);


Derfor anses de ikke umiddelbart for at være ens. Men Equal vil afsløre om vektorerne er 
elementvis ens - og derfor ens i matematisk henseende:


Equal(res[1],res[2]);
true


Equal(res[1],res[3]);
true


Matricer og vektorer er lidt specielle, hvad angår opbevaring i hukommelsen. 
Ved sædvanlige tilordninger i Maple:  


m:=7; n:=m;


vil m 'pege' på 7, og n vil også pege på 7, idet fuld evaluering af m i den anden tilordning gør, 
at n faktisk sættes til 7.


m,n;


Laver vi m om, vil n ikke ændres:


m:=8: n;
7


Lad nu matricen M være givet ved


M:=Matrix([[1,2],[3,4]]);


Til senere brug noterer vi os dens maskinadresse:


addressof(M);
155746216


Definerer vi nu N til at være M, vil N faktisk blive eksakt det samme som M: Hvis den ene af 
de to ændres, så gør den anden også:


N:=M;







Kapitel 7: Lineær Algebra


Kapitel 7 side 18


Vi ændrer på M:


M[1,1]:=76:


og vi ser, at N også ændres:


M,N;


Vi ændrer på N:


N[2,1]:=-99:


og også M ændres:


M,N;


De to matricer har begge stadig den adresse den oprindelige matrix M havde:


addressof(M),addressof(N);


Alt i alt ser vi, at der ikke er nogensomhelst forskel på M og N. Vi ser desuden, at 
maskinadressen for en matrix bevares ved elementvise ændringer.


Vil man derfor lave en kopi N af matricen M, således at man kan ændre på M uden dermed at 
ændre på N, må man bruge Copy:


N:=Copy(M);


Nu har N fået en ny maskinadresse:


addressof(M),addressof(N);


og den ene af de to kan ændres uden at den anden ændres:


M[1,1]:=45:


M,N;


addressof(M),addressof(N);


Basis for underrum. Søjlerum.
Lad der være givet nogle vektorer (det kunne være søjlerne i en matrix):


v1,v2,v3,v4:=<1,-3,2,-4>,<-3,9,-6,12>,<2,-1,4,2>,<-4,5,-3,


7>;

Vi vil udtynde den givne mængde af vektorer til en basis for Span( v1, v2, v3, v4 ):


Basis([v1,v2,v3,v4]);


Håndkraft:


V:=Matrix([v1,v2,v3,v4]);


GaussianElimination(V);


Pivoteringssøjlerne er 1, 3 og 4. Altså er {v1, v3, v4} en basis for Span( v1, v2, v3, v4 ).


Vil vi blot finde en basis for søjlerummet for V (altså en basis for Span( v1, v2, v3, v4 ) ), men
ikke nødvendigvis en basis, der er fundet ved udtynding, kan vi bruge ColumnSpace:


ColumnSpace(V);


Denne specielt simple basis for søjlerummet er opnået ved at lave søjleoperationer på V, 


hvilket jo svarer til rækkeoperationer på 
En søjleoperation ændrer ikke på det underrum, der udspændes af søjlerne: 


res:=Transpose(ReducedRowEchelonForm(Transpose(V)));

Basis for søjlerummet består nu af de n søjler, der er forskellige fra nulvektoren.
n er rangen af V ( = rangen af res):


n:=Rank(res);


Column(res,1..n);


Rang af matrix
Rangen af en matrix kan f.eks. defineres som det maksimale antal lineært uafhængige søjler 
(eller rækker!) i matricen, altså som dimensionen af søjlerummet. Rangen kan også defineres 
som antal fra nulrækken forskellige rækker i en echelonform for matricen.
Den findes let i Maple. Vi laver først en matrix vi kan afprøve ud af matricen A fra tidligere:


A;


M:=<A,-3*A>;


Rank(M);
2


Kontrol:


GaussianElimination(M);

Antal fra nulrækken forskellige rækker i echelonformen er 2, hvilket stemmer med Maples 
svar på rangen.


En basis for søjlerummet består da også netop af 2 vektorer:


ColumnSpace(M);


Determinanter


Vi tager en tilfældig 3x3-matrix:


A:=RandomMatrix(3);


Determinanten af A findes simpelthen ved kommandoen


Determinant(A);


Nogle lærebøger definerer determinanten rekursivt ved udvikling i komplementer langs første 
række. Vi illustrerer her denne definition.
Undermatricen der fås ved at stryge række 1 og søjle 2 i A er


Minor(A,1,2,output=matrix);


Underdeterminanten, der fås ved at tage determinanten af denne matrix er


Minor(A,1,2);
7013


Her følger så en definition af determinanten. 
Definitionen er rekursiv: Determinanten af en nxn-matrix defineres ud fra kendskabet til 
determinanter af orden (n-1)x(n-1) ved udvikling langs første række:


add((-1)^(1+i)*A[1,i]*det(Minor(A,1,i,output=matrix)),i=1.


.3);


Vi har brugt en af Maple ukendt funktion det. Det er gjort for at vi kan se, hvad der foregår.
Vi ved allerede, hvordan 2x2-matricer udregnes, så determinanten af A fås til:

eval(%,det=Determinant);


Eller lidt mere direkte:


add((-1)^(1+i)*A[1,i]*Minor(A,1,i),i=1..3);


Egenværdier og egenvektorer


Eksempel 1 (matrix med pæne egenværdier)


Lad matricen A være givet ved


A:=Matrix([[7,54,36],[8,-24,-3],[-20,90,23]]);


Det karakteristiske polynomium for matricen A kan bestemmes ved kommandoen


Determinant(A-lambda*IdentityMatrix(3));


Men der findes også en procedure CharacteristicPolynomial, der direkte giver 


polynomiet  (proceduren CharacteristicMatrix giver matricen ) :


CharacteristicPolynomial(A,lambda);


Egenværdierne for matricen A er rødderne i karakterpolynomiet:


solve( %=0, lambda);


Egenværdierne kan bestemmes direkte ved brug af Eigenvalues:


Eigenvalues(A);


Egenværdier og egenvektorer bestemmes på én gang ved brug af Eigenvectors. 
Output er som default en følge bestående af en vektor indholdende egenværdierne og en 
matrix, hvis søjler er egenvektorer hørende til egenværdierne og angivet i samme 

rækkefølge:


Eigenvectors(A);


Tilføjer man det valgfri argument output=list, fås resultatet som en liste af lister hver 
indeholdende egenværdien, dens algebraiske multiplicitet og en basis for egenrummet:


Eigenvectors(A,output=list);


Det er selvfølgelig nødvendigt, at output fra Eigenvectors består af både egenværdier og 
egenvektorer, idet man skal være i stand til at afgøre hvilken egenvektor hører til hvilken 
egenværdi. 
Man kan ikke stole på, at egenværdierne kommer i samme rækkefølge næste gang man 
prøver!
Dette betyder også, at man ikke skal fastlægge sine egenværdier ved brug af Eigenvalues 
og derefter bestemme egenvektorer vha. Eigenvectors. 


Når output som i default-tilfældet er en følge (hér af længde 2), kan man lave flere 
tilordninger på én gang, her 2:


ev,P:=Eigenvectors(A);


Hermed har ev fået tilordnet vektoren af egenværdier og P er tilordnet matricen, hvis søjler 
er egenvektorer. En enkelt af egenvektorerne (eksempelvis nr. 2) kan plukkes ud fra P 
således:


Column(P,2);

Eksempel 2 (matrix med "grimme" egenværdier)


Det er ikke normalt at egenværdier og egenvektorer for en matrix er pæne, bortset fra 
matricer lavet af matematiklærere.


Lad os som eksempel betragte matricen


A:=Matrix([[0,1,3,-4],[1,2,-5,1],[3,1,-4,6],[-1,2,-1,1]


] );


Egenværdierne for matricen A kan som i Eksempel 1 beregnes ved brug af Eigenvalues, 
men svaret fylder meget, så efter vi har set, hvor dejligt indviklede de ser ud, kan vi erstatte
semikolon med et kolon, således at output ikke vises. Herefter vil vi nøjes med at vise 
decimalbrøkstilnærmelserne:


Eigenvalues(A):


evalf(%); 


Alternativt kan man lave en decimalbrøksversion af matricen før man finder dens 
egenværdier. I så fald bruger Maple en numerisk metode til bestemmelse af egenværdierne.
Man skal (bl.a. af den grund) ikke regne med, at egenværdierne kommer i samme 
rækkefølge som før.
Vi har anbragt simplify udenom kommandoen for at omdanne decimaltal som -6.2+0.I til 
-6.2:


simplify(Eigenvalues(evalf(A)));

Egenvektorerne fås som i Eksempel 1 ved brug af Eigenvectors. Den giver en følge 
bestående af en vektor med egenværdierne og en matrix, hvis søjler er egenvektorerne. Vi 
gemmer de to i henholdsvis w og P. Outputtet fylder virkelig meget, så det undertrykkes i 
første omgang:


w,P:=Eigenvectors(A): 


Decimalbrøksversionen af w:


evalf(w);


fnormal bruges i næste linie faktisk kun for at reducere antal viste decimaler, så 
decimalbrøksversionen af P ikke fylder så meget:


evalf(P):   map(fnormal,%,5);


I stedet for kommandoen map(fnormal, %, 5); kunne ( i Maple 13) være brugt 
kommandoen fnormal~(%, 5); hvor man skal bemærke tegnet ~.


Her følger de tilsvarende beregninger med udgangspunkt i en decimalbrøksversion af A. 
Her bruger Maple en numerisk metode til bestemmelse af egenværdier og egenvektorer:


fw,fP:=Eigenvectors(evalf(A)):


simplify(fw);

map(fnormal,simplify(fP),4);


I det numeriske tilfælde sørger Maple for at egenvektorerne får længde 1 (længde = 2-
norm):
Først kontrolleres dette udsagn på første søjle i fP:


Norm(Column(fP,1),2);
1.


Her kontrolleres længderne af alle 4 søjler:


seq(Norm(Column(fP,k),2),k=1..4);


Ved beregningen af P ovenfor blev ikke brugt en numerisk metode, så søjlelængderne er 
ikke (nødvendigvis) 1:


seq(Norm(Column(evalf(P),k),2),k=1..4);


Eksempel 3 (en ikke diagonaliserbar matrix)


Lad A være matricen


A:=Matrix([[12,4,-8],[-13,-2,10],[6,3,-3]]);


Eigenvalues(A);


Egenværdien 2 har altså algebraisk multiplicitet 2.
Den geometriske multiplicitet fremgår af 


Eigenvectors(A,output=list);

Vi ser, at den geometriske multiplicitet af egenværdien 2 kun er én, da basis for 
egenrummet består af én vektor. Altså er A ikke diagonaliserbar.
Bruger vi default-output:


Eigenvectors(A);


ser vi, at en af søjlerne består af lutter nuller. Dette betyder igen, at den geometriske 
multiplicitet af egenværdien 2 kun er 1. 


Lidt mere direkte ser vi det samme ved at finde


NullSpace( A-2 );


Nulrummet for A-2I er altså éndimensional, så den geometriske multiplicitet af 
egenværdien 2 er 1.


Flere matrixopbygningsmetoder


DiagonalMatrix laver en diagonalmatrix:


DiagonalMatrix([2,a,-sqrt(7)]);

men den kan også lave en blokdiagonalmatrix:


Q1:=Matrix([[0,1],[1,0]]): 


Q2:=Matrix([[1,2],[3,4]]):


DiagonalMatrix([Q1,Q2,7]);


En Matrix kan opbygges ud fra mindre matricer (eller vektorer) vha. Matrix:


W:=Matrix([[Q1,Q1^(-1)],[Q2,Q2^(-1)]]);


Og den modsatte vej:


SubMatrix(W,2..3,1..4);


eller endnu lettere:


W[2..3,1..4];


v;


Matrix([v,2*v],scan=columns) = <v,2*v>;


<Q1|Q2> = Matrix([Q1,Q2]);

Enhedsmatricen


Enhedsmatricen kan man få f.eks. ved brug af IdentityMatrix:


IdentityMatrix(2);


Bemærk følgende forskel. Lægger man q til en matrix A, så returnerer LinearAlgebra-pakken 
denne sum uevalueret, når q ikke er et konkret tal eller en matrix. Men er q et konkret tal 
lægges det til diagonalen i A:


A+q; 


A-7;


A-lambda*IdentityMatrix(3);


Hessematricen, Jacobimatricen, Eksponentialmatricen


I VectorCalculus-pakken ligger procedurer til beregning af Hesse- og Jacobi-matricerne:


Hesse:=VectorCalculus[Hessian](f(x,y),[x,y]);


Jacob:=VectorCalculus[Jacobian]([f(x,y),g(x,y)],[x,y]);

Den funktion H, der sender (x, y) over i Hesse kan laves vha. unapply. 
I det abstrakte tilfælde som her, hvor f ikke er blevet defineret, må vi dog først konvertere diff 
til D. Dette er ikke nødvendigt i et konkret tilfælde, hvor differentiationerne jo bliver udført 
allerede ved definitionen af Hesse: 


H:=unapply(convert(Hesse,D),x,y):


H(p,q);


Jacob kan selvfølgelig på samme måde laves til en funktion J.


Lad Q være matricen 


Q:=Matrix([[0,1],[-1,0]]);


Eksponential matricen  fås ved brug af MatrixExponential:


M:=MatrixExponential(Q,t); 


Eksponentialmatricen er løsning til  :


diff~(M,t)=M.Q;


Bemærk den elementvise brug af diff, der i Maple 13 opnås ved diff~.


Overbestemte lineære ligningssystemer


Betragt følgende ligningssystem bestående af 4 ligninger med 3 ubekendte. 


S:={x+2*y-4*z=5,3*x+y-z=0,5*x-y+z=3,y+z=-4};


Slist:=convert(S,list);

Koefficientmatricen og højresiden kan pilles ud som følger:


A,b:=GenerateMatrix(Slist,[x,y,z]);


LinearSolve(A,b);


Error, (in LinearAlgebra:-LA_Main:-LinearSolve) inconsistent 


system


Systemet har altså ingen løsninger. Men den vektor v = (x, y, z), der minimerer længden (2-
normen) af Av-b findes ved én af følgende to kommandoer:


LeastSquares(S,{x,y,z});


v:=LeastSquares(A,b);


Residualvektoren er (vi transponerer af pladshensyn):


Transpose(A.v-b);


Den sædvanlige euklidiske norm (2-normen) af denne vektor er:


Norm(A.v-b,2);


linalg-pakken

Foruden pakken LinearAlgebra er der også den gamle lineær algebra-pakke linalg. 
Man bør være opmærksom på dens eksistens, da fejlindtastning i LinearAlgebra-pakken 
(eksempelvis matrix skrevet med småt) kan have uheldige konsekvenser. Derfor dette korte afsnit.


restart;


with(linalg):


Definition af matrix og vektor. Matrixprodukt. Invers. Transposition. 


Matricer kan defineres ved brug af proceduren matrix. Man kan opgive størrelsen som de to 
første argumenter, og en liste (eller en liste af lister) som tredie argument:


A := matrix(3 , 3, [0,1,3,1,2,-5,3,1,-4] );


Hvis man kun giver Maple en liste af lister, tolker Maple dette som en liste af rækkerne i en 
matrix. Matricen A kan altså også defineres således:


matrix([[0,1,3],[1,2,-5],[3,1,-4]] );


En tredie mulighed er som tredie argument til matrix at angive en funktion af to variable, der 
ved input (i, j) leverer matricens (i, j)-element:


matrix(2,7,(i,j)->i^2-j);


Nævnes matricen A


A;
A


så ser vi ikke andet end navnet.  linalg-matricer evalueres kun til navnet, akkurat som for 
funktioner og procedurer (og for tabeller og arrays, se kapitel 14). Fuld evaluering kan dog 
tvinges igennem ved hjælp af evalm:


evalm(A);

Den inverse til en invertibel matrix og den transponerede til en vilkårlig matrix fås ved simple 
kommandoer:


inverse(A);


transpose(A);


Den inverse kan også fås ved kommandoen:


evalm(A^(-1)):      # Output undertrykt


En vektor kan defineres ved brug af proceduren vector, men en vektor kan også blot opfattes 
som en matrix med kun én søjle eller kun én række
Vi definerer nu v som en vektor og b som en søjlevektor. Vi lader B være en tilfældig valgt 
matrix:


v:=vector([13,26,65]);


b := matrix(3 , 1 , [13,26,65] ); 


B:=randmatrix(3,3,entries=rand(-9..9));

Produkterne Av, Ab og AB fås enten vha. multiply eller vha. &* efterfulgt af evalm:


evalm(A&*v),evalm(A&*b),multiply(A,v),multiply(A,b); 


Bemærk, at outputtet har forskellig type: A&*v er en vektor, mens A&*b er en matrix 
(søjlevektor).


evalm(A&*B);


En vektor kan ganges på en matrix fra venstre. Her opfattes den som en rækkevektor:
 


evalm(v&*A),evalm(transpose(b)&*A);

8. Diverse Maplepakker


Mange af Maples procedurer er anbragt i pakker. Der er i Maple 13 ca. 110 pakker. En oversigt fås 
ved


?index,packages


I de foregående kapitler er allerede gjort brug af forskellige pakker, bl.a. plots, DEtools og 
LinearAlgebra.
I dette kapitel omtales numapprox, orthopoly, CurveFitting, Optimization og inttrans. I 
forbindelse med omtalen af CurveFitting og Optimization vises kort brugen af procedurerne
LinearFit og NonlinearFit fra Statistics-pakken.


Har man kendskab til Matlab, bør man være opmærksom på pakkerne CodeGeneration og Matlab. 
Den sidstnævnte forudsætter, at man har Matlab installeret på sin computer, idet man fra Maple kan 
få udregninger foretaget af Matlab, men resultatet leveret i Maple. CodeGeneration-pakken 
indeholder oversættere fra Maple til Matlab, Fortran, Java, C og VisualBasic.


Gamle pakker har form af tabeller (se kapitel 14) og navnene på såvel pakke som de indeholdte 
procedurer er typisk forkortede og skrevet med småt, som f.eks. inttrans, der er en forkortelse af
integral transform. 
Nye pakker (og reviderede gamle) er moduler (se kapitel 15). Navnene på såvel pakke som 
procedurer er for nye pakker skrevet helt ud og med stort begyndelsesbogstav for hver del af et 
sammensat ord, som f.eks. GaussianElimination fra LinearAlgebra-pakken. 


En procedure pp i en pakke med navn pk kan man bruge efter kommandoen with(pk);  men man kan 
også i stedet blot bruge dens lange navn, som er pk[pp]. For nye pakker (moduler) kan også bruges
pk:-pp.
Der findes også en kommando unwith(pk); der omgør det som  with(pk); gjorde:


with(LinearAlgebra): with(inttrans):


packages();


unwith(inttrans);


Error, (in unwith) the procedure `unwith' is currently disabled due


to the presence of table-based packages


packages();


Når der er tabelbaserede pakker i brug, virker unwith dog ikke, hverken på de tabelbaserede pakker 
eller på de modulbaserede pakker:


type(inttrans,table);
true


unwith(LinearAlgebra);


Error, (in unwith) the procedure `unwith' is currently disabled due


to the presence of table-based packages


Ønsker man i en række beregninger brugt en pakke kan man bruge en konstruktion som use pakke in
..... end use;


use LinearAlgebra in

A:=RandomMatrix(3);


d:=Determinant(A);


t:=Trace(A);


end use;


restart;


numapprox-pakken: Approksimation af funktion, infnorm


Et Taylorpolynomium for en given funktion er glimrende som approksimation nær 
udviklingspunktet. Ønsker man sig derimod en polynomial approksimation, der er god på et helt 
interval, kan man bruge procedurerne minimax eller chebyshev. Disse ligger i pakken
numapprox. Her findes også andet af interesse, bl.a. infnorm, der finder den største numeriske 
værdi af en funktion på et givet interval.


with(numapprox);


numapprox-pakken er et modul:


type(numapprox,`module`);
true


Vi finder størsteværdien for den numeriske værdi af funktionen


f:=x->sin(x^2)/x;


 på intervallet [0,10].


infnorm(f(x),x=0..10);
0.8512410654


Med ekstra argument:


infnorm(f(x),x=0..10,'x0');
0.8512410654


Størsteværdien for absolutværdien antages i x0, der blev fundet til


x0;
1.079582390


Vi forsøger variabelfri syntaks:


infnorm(f,0..10);


men infnorm generes åbenbart af at funktionen ikke er defineret i 0. 

Men da


Limit(f(x),x=0): %=value(%);


kan vi enten betragte et lidt mindre interval som f.eks. [0.001, 10]


infnorm(f,0.001..10);
0.8512410655


eller vi kan vha. piecewise definere f i 0 til grænseværdien, hvorved f i øvrigt bliver kontinuert på 
[0, 10]:


f:=x->piecewise(x=0,0,sin(x^2)/x);


infnorm(f,0..10);
0.8512410654


infnorm(f,0..10,'x0');
0.8512410654


plot(f(x),x=0..10):


plot([[x0,f(x0)]],style=point,symbolsize=30,symbol=


solidcircle,color=blue):


plots:-display(%,%%,caption=typeset("infnorm fandt 


størsteværdien af ", sin(x^2)/x," \n til at være ",evalf[3]


(f(x0))," på intervallet ",[0,10]));


x


2 4 6 8 10
0


infnorm fandt størsteværdien af  


 til at være 0.853 på intervallet 


Som man kan se af de to næste resultater, så kan det være en god idé at kontrollere sine resultater:


maximize(f(x),x=0..10);
0


maximize(f(x),x=0.1..10);
0.09999833334


Begge disse resultater er jo forkerte!
Et andet forsøg:

Optimization:-Maximize(sin(x^2)/x,x=0..10);


x-værdien er operand [2, 1, 2] i resultatet:


x1:=op([2,1,2],%);


f'(x1) er ca. 0:


D(f)(x1);


f''(x1) er negativ:


(D@D)(f)(x1);


Optimization:-Maximize fandt altså et lokalt maksimum, mere lover den dog heller ikke. Se 
mere nedenfor.


minimax-proceduren beregner den bedste rationale minimax-approksimation af grad (m,n) for en 
given reel funktion f på et givet interval [a, b] mht. til en given (positiv) vægtfunktion w. Dette 


betyder, at den bestemmer dét rationale udtryk   , der minimerer den maksimale afvigelse, 


dvs. minimerer størrelsen  , når det forlanges, at  og


. Maksimum tages over intervallet [a, b].


Vægtfunktionen kan gives som fjerde argument. Hvis det udelades, bruges vægten 1.


Vil vi have en polynomial approksimation, beder vi blot om, at nævnerpolynomiet får grad 0:
 


mm:=minimax(sin(x),x=0..Pi,[5,0]);


Resultatet kommer på Horner form, der er velegnet til effektiv udregning af talværdier.
For at se, hvor god en approksimation vi har fundet, plotter vi forskellen:


plot(sin(x)-mm,x=0..Pi,caption=typeset("Forskellen mellem 


sin og minimax: ",sin(x)-evalf[3](expand(mm))));

x


1 2 3
0


Forskellen mellem sin og minimax: 


Proceduren chebyshev approksimerer et givet udtryk ved Chebyshev-polynomier, der for k = 0, 1,
2, ... er defineret ved 


T(k,x)=cos(k*arccos(x));


 chebyshev giver tilstrækkeligt mange led i den uendelige Chebyshev-udvikling af den givne 
funktion til at sikre, at afvigelsen på det givne interval er mindre end den i tredie argument givne 


tolerance. Udelades 3. argument er tolerancen den mindst mulige, nemlig  .


CH:=chebyshev(sin(x),x=-Pi..Pi,5*10^(-4));


Chebyshev-polynomierne betegnes her med T, men kan også betegnes med ChebyshevT (se 
hjælpen hertil). Proceduren til beregning af T ligger i orthopoly-pakken:


with(orthopoly); 


Nu da T er kendt kan vi udtrykke CH eksplicit som et polynomium


expand(CH);


Den største numeriske afvigelse kan bestemmes vha. infnorm. Her udregner vi forholdet mellem 
den faktiske afvigelse og den forlangte tolerance:


infnorm(sin(x)-CH,x=-Pi..Pi)/(5*10^(-4));
0.01188557574

Vi kunne altså have forlangt mere og fået samme resultat:


CH2:=chebyshev(sin(x),x=-Pi..Pi,9*10^(-5));


CurveFitting-pakken


Er et antal sammenhørende x- og y-værdier givet, kan man ønske at finde et analytisk udtryk for 
en funktion, der går gennem alle de givne punkter i xy-planen. Her kan bruges et polynomium, et 
såkaldt kollokationspolynomium, eller en spline-funktion, som er en stykkevist defineret funktion,
hvis enkelte dele er polynomier (typisk af 3. grad). 
En anden idé er at bestemme den funktion fra en bestemt klasse, der bedst approksimativt går 
gennem de givne punkter.
Begge disse idéer kræver brug af CurveFitting-pakken.


restart;


with(CurveFitting);


Interpolation, kollokation, spline.


Vi tænker os givet et antal sammenhørende x- og y-værdier. Disse givne punkter i xy-planen 
konstruerer vi for eksemplets skyld selv. Vi lader dem være punkter fra sinuskurven. Så kan vi
sammenligne bagefter.


De x-værdier, vi ønsker at bruge, gemmer vi i en liste:


Lx:=[seq(k,k=-3..3)]; 


De tilsvarende y-værdier (med 3 betydende cifre) kan fås ved at anvende evalf[3]@sin på 
hvert element i listen Lx. Dette kan gøres ved hjælp af proceduren map:


Ly:=map(evalf[3]@sin,Lx); 


Det samme resultat kan i Maple 13 opnås ved brug af den elementvist virkende (evalf[3]@sin)
~ :


(evalf[3]@sin)~(Lx);


Her kommer så kollokationspolynomiet. Proceduren hedder PolynomialInterpolation:

kollo:=PolynomialInterpolation(Lx,Ly,x);


Man kan overveje, om ikke led med meget små koefficienter bør fjernes. Dette kan ske med


fnormal, der (pr. default) erstatter decimaltal mindre end  med 0:


fkollo:=fnormal(kollo);


Vi vil selvfølgelig gerne se resultatet afbildet grafisk. Derfor danner vi vha. zip-proceduren en 
liste af  [x, y]- par (altså en liste af punkter i planen) med henblik på at afbilde disse sammen 
med grafen for kollokationspolynomiet kollo (eller evt. fkollo). 


Lxy:=zip( (x,y)->[x,y], Lx,Ly);


En kortere version (bemærk accent grave):


zip( `[]`, Lx,Ly);


Nedenfor bruges det lange navn for display, som er plots:-display:


p1:=plot(Lxy,style=point,symbol=circle,color=blue,


symbolsize=20):


p2:=plot(kollo,x=-Pi..Pi):


plots:-display(p1,p2,caption="Kollokationspolynomiet og 


punkterne");


x


0 1 2 3


1


Kollokationspolynomiet og punkterne


Da punkterne blev taget fra sinuskurven, forventer vi, at forskellen mellem sin(x) og kollo er 
lille:


numapprox:-infnorm(sin(x)-kollo, x=-Pi..Pi);
0.01987777235

Afvigelsen er så lille, at vi undlod at tegne sinuskurven sammen med grafen for 
kollokationspolynomiet. 


En spline-funktion er en stykkevist defineret funktion. På hvert delinterval er den et 
polynomium (i Maple som default et polynomium af tredie grad). Proceduren hedder Spline.


Resultatet fås således, men vi venter med at vise resultatet til vi har fjernet ubetydelige 
numeriske størrelser. I dette tilfælde vil "sætte til nul",  hvad der måtte være numerisk mindre 
end 10^(-5+2), hvor 5-tallet er andet argument til fnormal. Bemærk også, at Maple nu kun 
angiver 5 betydende cifre:


sp:=Spline(Lx,Ly,x):


fsp:=fnormal(sp,5);


Afvigelsen er denne gang ca. halv så stor:


numapprox:-infnorm(sin(x)-fsp, x=-Pi..Pi);
0.009711246285


Mindste kvadraters metode. Lineær regression.


Ofte har man langt flere punkter end nødvendigt til bestemmelse af de ukendte parametre i 
ligningen for den kurve, som man ønsker skal gå gennem disse punkter. I stedet for blot at 
smide punkter ud ønsker man at finde den kurve af den givne form, der bedst muligt tilnærmet 
går gennem alle punkterne. 
Vi kan her behandle kurver af formen 


 + ... + 


hvor ,   , ... ,  er givne og kendte funktioner, og hvor det er konstanterne ,  , ... , , der


søges. Man siger, at konstanterne optræder lineært. Der bliver tale om lineær regression. 


Metoden går ud på at minimere summen af kvadrater , hvor ( ) er det i'te 


punkt, og hvor f er den givne funktion     + ... + . Dette er


mindste kvadraters metode.
Vi skal her bruge proceduren LeastSquares fra CurveFitting-pakken. Det bør tilføjes, at
Statistics-pakken indeholder proceduren LinearFit, der kan være af speciel interesse i denne 
sammenhæng. 


Eksempel 1

Der er givet 11 punkter i xy-planen i form af to lister: En liste med x-værdier og en liste 
med y-værdier. Vi søger den rette linie, der på bedst mulig måde tilnærmet går gennem de 
11 punkter.


Lx:= [0, 1/2, 1, 3/2, 2, 5/2, 3, 7/2, 4, 9/2, 5]: 


Ly:=[0.79, 0.90, 1.15, 1.35, 1.53, 1.66, 1.85, 2.09, 


2.19, 2.38, 2.66]:


Til senere brug defineres her listen af punkter i xy-planen:


Lxy:=zip(`[]`,Lx,Ly);


Vi laver et plot af punkterne til senere brug:


p1:=plot(Lxy,style=point,color=blue,symbolsize=20,


symbol=cross):


Når den kurve, der skal forsøge at gå gennem punkterne faktisk er en ret linie, altså har 
ligningen  y = ax + b, så kan følgende korte form bruges:


res:=LeastSquares(Lx,Ly,x);


Vi kan evt. bruge match til at hente parameterværdierne ud fra resultatet:


match(res=a*x+b,x,'param');
true


param;


Lad f være den funktion, der sender x over i res:


f:=unapply(res,x);


Et plot af grafen for f:


p2:=plot(f,0..5): 


Vi kan nu vise punkter og graf sammen:


plots:-display(p1,p2,title="Lineær regression");

0 1 2 3 4 5


Lineær regression


Summen af afvigelsernes kvadrater   er


add((Ly[i]-f(Lx[i]))^2,i=1..nops(Lx));
0.01447181806


Statistics-pakken indeholder proceduren LinearFit, der også kan bruges: 


Statistics:-LinearFit([1, x], Lx, Ly, x);


For statistiske oplysninger knyttet til resultatet se hjælpen om LinearFit.


Eksempel 2


Der er igen givet 11 punkter i xy-planen i form af to lister: En liste med x-værdier og en 
liste med y-værdier. Vi forestiller os, at der er grund til at tro, at punkterne næsten ligger på
en kurve med ligningen  y = a sin x + b cos 2x.  Vi søger den kurve af denne slags, der på 
bedst mulig måde tilnærmet går gennem de 11 punkter.


Lx:= [0, 1/2, 1, 3/2, 2, 5/2, 3, 7/2, 4, 9/2, 5]: 


Ly:=[-0.5, 1.0, 3, 3.6, 3.2, 1.7, 0, -1.5, -2.3, -2.6, 


-2.6]:


Lxy:=zip((x,y)->[x,y],Lx,Ly);


Vi laver et plot af punkterne til senere brug:


p1:=plot(Lxy,style=point,color=blue,symbol=cross,


symbolsize=20):


Vi må nu angive, at vi ønsker en sammenhæng af typen y = a*sin(x) + b*cos(2*x). Dette 
anføres således:

res:=LeastSquares(Lx,Ly,x,curve=a*sin(x)+b*cos(2*x));


Værdierne af konstanterne kan hentes ud således:


match(res=a*sin(x)+b*cos(2*x),x,'param');
true


param;


Vi laver en funktion ud fra res:


f:=unapply(res,x);


og laver et plot af dens graf, hvorefter vi kan vise graf og punkter sammen:


p2:=plot(f,0..5):


plots:-display([p1,p2],title=typeset("Model: ",y=a*sin


(x)+b*cos(2*x)));


1 2 3 4 5
0


1


2


3


Model: 


Summen af afvigelsernes kvadrater   er


evalf(add((Ly[i]-f(Lx[i]))^2,i=1..nops(Lx)));
0.07738597839


Statistics-pakken indeholder proceduren LinearFit, der også kan bruges: 


Statistics:-LinearFit([sin(x), cos(2*x)], Lx, Ly, x);


For statistiske oplysninger knyttet til resultatet se hjælpen om LinearFit.


Optimization-pakken


restart;


Pakken indeholder: 


with(Optimization);

Procedurerne er numeriske procedurer. Det er umiddelbart klart, hvad Maximize og Minimize 
skal forsøge at gøre, nemlig maksimere eller minimere en funktion af en eller flere variable evt. 
underkastet visse bånd i form af ligninger eller uligheder. LSSolve er specielt beregnet til 
minimering af en sum af kvadrater på afvigelser (mindste kvadraters metode).
Vi illustrerer brugen af disse tre i eksemplerne nedenfor.


Eksempel 1


Vi vil finde største- og mindsteværdi for udtrykket 


f:=(2*x+y)*exp(-x*y);


når de variable x og y er underkastet begrænsningerne . 


Dette kan gøres således:


Maximize(f,{2*x+y<=4},assume=nonnegative);


Minimize(f,{2*x+y<=4},assume=nonnegative);


fnormal(%);


Proceduren identify prøver at finde en "pæn" (se hjælpen) konstant, hvis 
decimalbrøkstilnærmelse er lig med inputtet: 


identify(%);


Dette er dog kun et lokalt minimum. Det betaler sig at give Minimize forskellige startpunkter:


Minimize(f,{2*x+y<=4},assume=nonnegative,initialpoint={x=


1,y=.5});


Vi prøver en række tilfældigt valgte punkter.
Hertil laver vi en procedure, der vælger tilfældige decimaltal mellem 0 og 2:


r:=RandomTools[Generate](float(range=0..2,method=uniform),


makeproc);


Vi prøver nu Minimize med 10 tilfældige startpunkter valgt i rektanglet [0,2]x[0,4]:


seq(Minimize(f,{2*x+y<=4},assume=nonnegative,initialpoint=


{x=r(),y=2*r()}),k=1..10);

fnormal({%});


identify(%);


Her har vi så (hvis vi var heldige!) faktisk også fundet den globale minimumsværdi 0, der 
antages i (0,0).


Eksempel 2: Mindste kvadraters metode


Sammenhørende værdier af to fysiske størrelser t og y er givet i form af to lister:


Lt:=[seq(.5*k,k=-6..10)];


Ly:=[1.01,1.04,1.10,1.19,1.32,1.48,1.66,1.85,2.02,2.19,


2.33,2.46,2.56,2.65,2.72,2.78,2.83]:


Det antages, at sammenhængen mellem t og y er givet ved y = f(t), hvor funktionen f er givet 
ved


f:=t->a+c*exp(-exp(-b*(t-m)));


og hvor a, b, c og m er konstanter. Bemærk her, at konstanterne optræder ulineært, således at 
lineær regression ikke kan bruges. Vi skal bestemme konstanterne, så summen af kvadraterne 


på afvigelserne mellem  og  er mindst mulig. Denne sum af kvadrater (fejlen) er givet 


ved


Fejl:=add((f(Lt[k])-Ly[k])^2,k=1..nops(Lt));

Vi prøver Minimize uden startgæt.
For at få lidt information om hvad der foregår, sætter vi infolevel[Optmization] til 2:


infolevel[Optimization]:=2:


Minimize(Fejl);


NLPSolve: calling NLP solver


NLPSolve: using method=pcg


NLPSolve: number of problem variables 4


NLPSolve: trying evalhf mode


attemptsolution: number of major iterations taken 50


Det gik jo godt, men ellers er det en god idé ligesom i Eksempel 1 at give et startgæt. Der er 
også andre valgfri tilføjelser. Eksempelvis kan det maksimale antal iterationer sættes. Vi 
prøver at sætte det til en relativt lav værdi (default afhænger af metoden, men var ovenfor 50):


Minimize(Fejl,initialpoint={a=.5,b=.4,c=0.9,m=.4},


iterationlimit=25);


NLPSolve: calling NLP solver


NLPSolve: using method=pcg


NLPSolve: number of problem variables 4


NLPSolve: trying evalhf mode


Warning, limiting number of major iterations has been reached


attemptsolution: number of major iterations taken 25


Der findes i Optimization-pakken en procedure LSSolve, der direkte er beregnet til denne 
slags problemer. Den accepterer som input listen af residualer, dvs. listen af afvigelser 


:


resid:=[seq(f(Lt[k])-Ly[k],k=1..nops(Lt))];


Her uden startgæt, men først lukker vi for al information:


infolevel[Optimization]:=0:


LSSolve(resid);

Her med startgæt og for eksperimentets skyld med den mindste iterationsgrænse, for hvilken 
ingen advarsel kommer. 


LSSolve(resid,initialpoint={a=.5,b=.4,c=0.9,m=.4},


iterationlimit=46);


subs(%[2],f(t));


Ved sammenligning med Minimize anvendt på Fejl ovenfor og med samme startgæt ses, at
LSSolve når sit resultat med meget færre iterationer. Men LSSolve er jo også skræddersyet til 
denne opgave.


Statistics-pakken indeholder proceduren NonlinearFit, der også kan bruges: 


Statistics:-NonlinearFit(f(t), Lt, Ly, t);


Laplacetransformationen


restart;


Den Laplacetransformerede af en funktion f er givet ved , når den da ellers 


eksisterer.
Laplacetransformationen ( laplace ) befinder sig i pakken inttrans sammen med en række andre 
integraltransformationer: 


with(inttrans);


Syntaksen er enkel:


laplace(t^3*cos(t), t, s);


Resultatet kan selvfølgelig også findes ved direkte brug af definitionen:


Int(exp(-s*t)*t^3*cos(t),t=0..infinity): %=value(%) assuming


s>0;;


(Prøv at fjerne assuming s>0). Tilbagetransformation foregår vha. invlaplace, som det ses her i et
andet eksempel: 


u:=exp(-3*s)/s^5/(s^2+1)+2/s;

invlaplace(u, s, t);


convert(%,piecewise);


At udtrykket er undefined for t = 3 skyldes, at Heaviside-funktionen i Maple er  undefined for t =
3:


Heaviside(0);
undefined


Hvis man ønsker det, kan man selv sætte Heaviside(0) til hvad man ønsker, eksempelvis 1, således
 Heaviside(0):=1; Det gør vi ikke her.
Diracs deltafunktion hedder Dirac i Maple:


laplace(Dirac(t-7),t,s);


e


invlaplace( %,s,t); 


Som eksempel på anvendelsen af laplace og invlaplace løser vi følgende differentialligning 
ved Laplacetransformation:


ligning:=diff(x(t),t)+x(t)=Heaviside(t-1)*exp(t)+Dirac(t-1);


Som begyndelsesbetingelse tager vi


begynd:=x(0)=7; 


Først løses ved trinvis Laplacetransformation, altså simuleret håndregning.


Differentialligningen Laplacetransformeres:


lapligning:=laplace(ligning, t, s); 

I den fremkomne ligning isoleres den Laplacetransformerede af x:


solve(lapligning, {laplace(x(t),t,s)} ); 


Der kan nu inverteres:


invlaplace(%,s,t); 


Vi indsætter begyndelsesværdien:


subs(begynd,op(%)); 


Bemærk, at op blev brugt for at fjerne {}, idet op finder operanderne i et udtryk. Operanderne i en 
mængde er elementerne. Vores mængde indeholdt kun et element.
Maple giver her svaret udtrykt ved hyperbolske funktioner. Ønsker vi i stedet brugt 
eksponentialfunktioner, kan vi bruge en convert-kommando:


convert(%,exp);


collect(simplify(%),Heaviside);


convert(%,piecewise); 


res:=simplify(rhs(%));


Resultatet plottes:


plot(res,t=0..2,discont=true);

Maple kan selv lave disse operationer ved brug af dsolve med tilføjelsen method=laplace:


dsolve({ligning,begynd}, x(t), method=laplace);


Man bør ovenfor bemærke, at det er nødvendigt at medtage x(t), hvilket jo ellers for det meste 
ikke er tilfældet i dsolve. 


Faktisk går det udmærket uden tilføjelsen method=laplace og vi bebyrdes ikke med hyperbolske 
funktioner:


dsolve({ligning,begynd});


collect(%,Heaviside);


Eksempel: System med tilbagekobling og forsinkelse


Betragt et lukket system bestående af to omrørte tanke begge med rumfang V . 

F


FTank 1 volumen = V Tank 2 volumen = V


x(t) y(t)


Væske strømmer fra tank 1 til tank 2 med en hastighed på . Transporten gennem 


forbindelsesrøret tager tiden  (timer). Væske strømmer også fra tank 2 til tank 1 med en 


hastighed på F og med transporttid . I væsken er opløst et stof A, hvis koncentrationer til tiden


t i de to tanke er lig med henholdsvis  og . Det antages, at koncentrationen af stof A til 


tiden   i  forbindelsesrørene og tank 2 er lig med nul, mens koncentrationen i tank 1 er lig 


med . 


Der sker intet andet i systemet. Der er altså kun tale om en lidt kompliceret opblanding af de to


opløsninger. Vi søger  til et vilkårligt tidspunkt  t > 0.


Massebevarelse giver følgende system af differentialligninger:


ligning1:=V*diff(x(t),t)=-F*x(t)+F*Heaviside(t-tau)*y(t-


tau);


ligning2:=V*diff(y(t),t)=-F*y(t)+F*Heaviside(t-tau)*x(t-


tau);


Vi løser systemet ved Laplacetransformation, men må først fortælle Maple, at  er positiv:


assume(tau>0);


Af bekvemmelighedshensyn vil vi betegne de Laplacetransformerede af  og  med 


henholdsvis  og . Hertil bruger vi alias:


alias( X(s)=laplace(x(t),t,s), Y(s)=laplace(y(t),t,s) );

laplace({ligning1,ligning2},t,s);


solve(%,{X(s),Y(s)});


Vi indsætter de givne begyndelsesværdier, og for ikke at drukne i symboler antager vi her, at 


.


lapres:=eval(%,{y(0)=0,x(0)=x[0],V=1});


Tilbagetransformationen er ikke helt nem. Den vil da også resultere i en uendelig tuborg! 
Maple klarer den ikke uden vores hjælp. Vi foretager en rækkeudvikling med 
eksponentialudtrykket som (frosset) variabel:


subs(exp(-s*tau)=frys,lapres);


XY:=subs(%,[X(s),Y(s)]);


Vi rækkeudvikler i variablen frys:


series~(XY,frys=0,11):


convert(%,polynom):


Vi optøer den frosne eksponentialfunktion:


subs(frys=exp(-s*tau),%):


Vi tilbagetransformerer:


collect( invlaplace~(%,s,t), Heaviside):


Dette resultat er det eksakte resultat på intervallet .


Vi tegner graferne for x(t) og y(t) for  og F = 1:  


xy10:=unapply(%,F,tau,x[0],t):


xy10(F,tau,x0,t)[1];

plot(xy10(1,1,1,t),t=0..11,caption=typeset("x(t) (rød) og 


y(t) (grøn) for ",tau=1,", ",x[0]=1,", ",F=1));


t


0 2 4 6 8 10
0


1


x(t) (rød) og y(t) (grøn) for , , 


Slutværdireglen giver os den fælles koncentration, som begge tanke efterhånden nærmer sig:


limit( s*subs(lapres,X(s)),s=0),limit( s*subs(lapres,Y(s)


),s=0);


slut:=unapply(%[1],F,tau,x[0],t);


Animation med F som animationsparameter:


plots:-animate(plot,[ [op(xy10(F,1,1,t)),slut(F,1,1,t)], 


t=0..11,caption=typeset("x(t) (rød) og y(t) (grøn) for ",


tau=1,", ",x[0]=1),thickness=2],F=0..3 );

t

0


1


x(t) (rød) og y(t) (grøn) for , 


Animation med tau som animationsparameter:


plots:-animate(plot,[ [op(xy10(2,tau,1,t)),slut(2,tau,1,t)


], t=0..11*tau,caption=typeset("x(t) (rød) og y(t) (grøn) 


for ",F=2,", ",x[0]=1),thickness=2],tau=1..3 );


t


0 10 20 30
0


1


x(t) (rød) og y(t) (grøn) for , 

9. Plotning


Basale plottekommandoer


Maple kan plotte både algebraiske udtryk og funktioner (pileprocedurer og andre procedurer). 


Intervallet [a, b] angives i Maple ved  a..b . Hvis en variabel x skal gennemløbe hele intervallet 
skrives x=a..b . Der skal være mindst to punktummer. 


plot( x*sin(x^2), x=-2..2);


x


0 1 2


1


Der kan også bruges variabelfri syntaks som i følgende tilfælde. Bemærk, at der ikke må være en 
variabel i hverken funktion eller interval:


plot( sin, -2*Pi..2*Pi); 


0 2 4 6


1


I eksemplet ovenfor fandt Maple selv et passende interval på ordinat-aksen (y-aksen). Den 

forsøger at få alle de beregnede punkter med på billedet. Dette er ikke altid ønskeligt:


plot( ln(x)/x, x=0..5);


x


1 2 3 4 5
0


Det ses, at de meget store negative værdier for   for x nær 0 gør resten af grafen flad, så den


næsten er helt væk. 
Vi kan udbedre dette ved at tilføje  y=c..d  eller blot  c..d,  f.eks. således:


plot( ln(x)/x, x=0..5, y=-1..0.5);


x


1 2 3 4 5


y


0


Flere udtryk eller funktioner kan plottes i samme koordinatsystem som vist i de næste 
kommandoer. Her er også tilføjet et valgfrit argument, der angiver de farver vi selv foretrækker. 
Ellers ville de 4 farver have været rød, grøn, gulbrun ("Goldenrod") og blå. Desuden er tykkelsen 
sat til 2 (default er 1 og ikke 0, som der står i hjælpen i Maple 13. Se ?plot, options):


plot([x,x^2,x^3,x^4], x=0..1, color=["DarkSlateBlue","Red",


"Green","Maroon"],thickness=2); 

x


0 1
0


1


Der er brugt nye farvenavne. Disse er tekststrenge. De gamle farvenavne var Maple-navne (ikke 
tekststrenge) og blev skrevet med småt, eksempelvis red eller blue. De gamle navne virker stadig.
For de nye farvenavne, se hjælpen


?plot,colornames


Her plottes 3 funktioner med brug af variabelfri syntaks: 


plot([sin,cos, sin+cos], -Pi..Pi,thickness=2,gridlines=true)


;


0 1 2 3


1


I stedet for de kantede parenteser kan bruges mængdeklammer. Maple opfatter 


 som en mængde, men  som en liste. Det er en fordel at 


bruge en liste, idet Maple da respekterer rækkefølgen. I det følgende eksempel er sinuskurven 
fuldt optrukket (linestyle = 1), cosinuskurven er punkteret (linestyle = 2) og summen er stiplet 
(linestyle = 3). Der kan også tilføjes en signaturforklaring (legend): 


plot([sin,cos, sin+cos],-Pi..Pi,linestyle=[1,2,3],


color="Black",legend=["sin","cos","sumfunktionen"],


thickness=3); 

sin cos sumfunktionen


0 1 2 3


1


Er en kurve givet ved parameterfremstillingen 


hvor t gennemløber intervallet , fås et plot af kurven ved som første argument til plot at 


angive en liste, hvis første og andet element er x(t),y(t) og hvis tredie element er interval-
angivelsen, altså i dette tilfælde t=0..2*Pi:


plot( [ 3*cos(t), 2*sin(t)^3, t=0..2*Pi],


    title="Parametrisk plot",thickness=2,


    caption=typeset([x,y]=[ 3*cos(t), 2*sin(t)^3]) );


0 1 2 3


1


2


Parametrisk plot


plot-proceduren accepterer et væld af valgfri argumenter. Det følgende er igen et parametrisk 
plot, hvor (x,y) er givet ved samme udtryk, som i forrige eksempel, men hvor koordinaterne 
opfattes som polære; der tegnes ingen akser.


plot( [ 3*cos(t), 2*sin(t)^3, t=0..2*Pi], coords=polar, 


axes=none,title="Flettede hjerter",thickness=3);

Flettede hjerter


Med tilføjelsen filled=true farves området mellem grafen og x-aksen:
  


f:=sin(x): g:=cos(x):


plot(f,x=0..2*Pi,filled=true,color="DodgerBlue");


x


1 2 3 4 5 6
0


1


plot([f,g],x=0..2*Pi,filled=true,color=["MistyRose",


"DimGray"],transparency=0.2);

x


1 2 3 4 5 6
0


1


Bemærk, at f (den første af de to) ligger øverst (fjern evt. tilføjelsen om transparency, eller sæt den
til 0). Vi udnytter denne observation i følgende kunstfærdige konstruktion, der farver området 
under f  og over g gråt.


plot([min(f,0),max(g,0),f,g],x=0..2*Pi,filled=true,color=


["White","White","Gray","Gray"]);


x


1 2 3 4 5 6
0


1


Den næste konstruktion farver området mellem f  og  g med en grøn farve:


plot([min(max(f,g),0),max(min(f,g),0),f,g],x=-2*Pi..2*Pi,


filled=true,color=["White","White","SeaGreen","SeaGreen"]);

x


0 2 4 6


1


Et plot kan opbevares i en variabel. I eksemplet nedenfor er tilføjet to valgfrie argumenter, der 
gør, at plot kun finder 5 punkter på grafen. Dette er gjort af hensyn til det følgende. 
Først defineres plottestrukturen og opbevares i variablen mitplot:


mitplot:=plot( x*ln(x), x=0..1.5,numpoints=5,adaptive=false)


;


Ved blot at nævne mitplot vises plottet:


mitplot; 


x


1
0


Vil man se selve plotte-datastrukturen, der her opbevares i variablen mitplot, kan man bruge
lprint-proceduren således


lprint(mitplot); 


PLOT(CURVES([[.392348774999999983, -.367083123550222324], 


[.733729346200000077, -.227173652990711516], 


[1.11764750200000007, .124311496415864309], [1.50000000000000000,


.608197662162246576]], COLOUR(RGB, 1.00000000, 0., 0.)), 


AXESLABELS(x, ""), VIEW(0. .. 1.5, DEFAULT))

Man ser, at plottepunkterne opbevares i en liste. Hvert punkt er selv en liste bestående af de to 
koordinater. Denne liste af lister kan fiskes ud af mitplot ved hjælp af indets således: 


op(indets(mitplot,listlist));


indets finder mængden af de operander i mitplot, der er af typen listlist, og op fjerner 
mængdeklammerne.
 
Man kan sætte informationsniveauet for plot til 2 ved kommandoen


infolevel[plot]:=2:


Herefter kommer forskellig information om plot-udregningen, selv hvis plottet ikke vises, som i 
følgende eksempel:


plot( x*ln(x), x=0..1.5):


plot/adaptive: evalhf failed  [plot/adaptive numeric exception: 


division by zero]


plot/adaptive: using evalf


plot/adaptive: produced  59  output segments


plot/adaptive: using  59  function evaluations


Hardware float - beregningen (evalhf) gik galt ovenfor pga. et forsøg på beregning af ln(0). Men 
det fik blot plot til at bruge software float - beregning (evalf).
Vi prøver på et interval, der ikke indeholder 0:


plot( x*ln(x), x=0.01..1.5):


plot/adaptive: evalhf succeeded


plot/adaptive: produced  49.  output segments


plot/adaptive: using  49.  function evaluations


Vi sætter niveauet til 1 for at undgå informationen fra plot:


infolevel[plot]:=1:


Uønskede lodrette forbindelseslinier
Undertiden er det ikke ønskeligt, at plot forbinder det sidstudregnede plottepunkt med det 
næstudregnede:


plot( tan(x), x=-Pi..Pi, -10..10);







Kapitel 9: Plotning


Kapitel 9 side 9


x


0 1 2 3


5


10


Tilføjelsen discont = true gør, at Maple undersøger, om det givne funktionsudtryk er 
diskontinuert et eller andet sted (eller har en singularitet, som hér). Herved undgås uønskede 
lodrette liniestykker:


plot( tan(x), x=-Pi..Pi, -10..10, discont=true); 


x


0 1 2 3


5


10


discont er en forkortelse for discontinuous, altså diskontinuert (= ikke kontinuert).


Skal et udtryk i 2 variable eller en funktion af to variable plottes, bruges proceduren plot3d:
 


plot3d( x^2*sin(x+y^2), x=-2..2, y=-2..2,title="Klik og træk


med musen"); 

Klik og træk med musen


Efter klik på billedet kan figuren drejes ved at trække med musen. Et andet eksempel med 3 
valgfri tilføjelser. Figuren er ikke vist:


plot3d( x^2*sin(x+y^2), x=-2..2, y=-2..2, view=-1..1, axes=


boxed,transparency=.5,style=patchnogrid);


Tilføjelsen view = -1..2  begrænser z-aksen til intervallet [-1,2], mens transparency = 0.5 sætter 
gennemsigtigheden til 50%. Tilføjelsen style = patchnogrid fjerner det net ("grid"), der ellers 
vises. Proceduren plot3d accepterer masser af valgfri argumenter. En beskrivelse af mulighederne
fås ved følgende kommando:


?plot3d,options


For funktioner af to variable kan man også bruge variabelfri syntaks ganske som for funktioner af 
en variabel. Vi definerer først en funktion af to variable:


g:=  (x,y) -> x*y^2*exp(-x^2-3*y^2);


plot3d(g, -1..1, -1..1,axes=boxed,filled=true,lightmodel=

light4,style=patchnogrid); 


Flere grafer kan vises samtidigt, enten i tuborgklammer eller i kantede parenteser. Benyttes det 
sidste og er der 3 grafer,  så skal tilføjes plotlist som argument, ellers opfattes kommandoen som 
en parametrisk plottekommando. 
Her tegnes grafen for g sammen med tangentplanen i punktet (1/2, 1/2, g(1/2,1/2)):


tp:=mtaylor(g(x,y),[x=1/2,y=1/2],2);


plot3d([g(x,y),tp],x=0..1,y=0..1,style=[patch,wireframe],


orientation=[-20,85]);


En kugleoverflade med radius = 1 kan fås således


plot3d(1,phi=0..Pi,theta=0..2*Pi,coords=spherical);

men kan også fås ved brug af parametriseringen 


:


plot3d([sin(phi)*cos(theta),sin(phi)*sin(theta),cos(phi)],


phi=0..Pi,theta=0..2*Pi);


Skal i stedet tegnes graferne for de funktioner af to variable, der står i listen, må tilføjes plotlist=
true eller blot plotlist. 


plot3d([sin(phi)*cos(theta),sin(phi)*sin(theta),cos(phi)],


phi=0..Pi,theta=0..2*Pi,plotlist,style=[patch,patchnogrid,


wireframe],orientation=[35,75]);

PLOT3D-datastrukturer


Vil man se selve plot3d-datastrukturen, kan man som i det 2-dimensionale tilfælde bruge
lprint-proceduren:


mitplot3d:=plot3d(g, -1..1, -1..1,axes=boxed,numpoints=9,


transparency=.4): 


lprint(mitplot3d);


PLOT3D(GRID(-1. .. 1., -1. .. 1., Array(1 .. 4, 1 .. 4, {(1, 1)


= -0.183156388887341787e-1, (1, 2) = -0.292885709017474197e-1, 


(1, 3) = -0.292885709017474058e-1, (1, 4) = 


-0.183156388887341926e-1, (2, 1) = -0.148504754148298982e-1, 


(2, 2) = -0.237474217937020254e-1, (2, 3) = 


-0.237474217937020186e-1, (2, 4) = -0.148504754148299086e-1, 


(3, 1) = 0.148504754148298928e-1, (3, 2) = 


0.237474217937020220e-1, (3, 3) = 0.237474217937020116e-1, (3, 


4) = 0.148504754148299034e-1, (4, 1) = 0.183156388887341752e-1,


(4, 2) = 0.292885709017474266e-1, (4, 3) = 


0.292885709017474093e-1, (4, 4) = 0.183156388887341891e-1}, 


datatype = float[8], storage = rectangular, order = C_order)), 


TRANSPARENCY(.4), AXESSTYLE(BOX))


Man ser, at plottepunkterne i dette tilfælde opbevares i en GRID-struktur, der har z-værdierne 
svarende til xy-grid-punkterne opbevaret i en matrix (en Array, se kapitel 14). 
Tilføjelsen adaptive=false er ikke brugbar i det 3-dimensionale tilfælde, så numpoints=9 sikrer
kun, at mindst 9 punkter på grafen bliver fundet. Som det ses, findes i dette tilfælde 16 
punkter. Her er z-værdierne (beskåret til 5 betydende cifre af pladshensyn):


A:=op(indets(mitplot3d,Array)): map(evalf[5],%);

Vi kunne selv have fundet disse således:


evalf[5](Array(1..4,1..4,(i,j)->g(-1+(i-1)*2/3,-1+(j-1)*


2/3)));


I de næste linier producerer vi først det net af 16 punkter i xy-planen i hvilke z-værdierne 
udregnes (listen L), og listen M af punkterne (x, y, z).
Derefter laver vi et 2-d plot (p1) af punkterne L og laver en funktion tr, der kan transformere 
dette plane plot til et 3d-plot.
Så laves to 3d-plots, dels et lysegråt plot (p0) af xy-planen, dels et plot (p2) af punkterne M.
Det hele vises sammen med det oprindelige plot mitplot3d af plots:-display.


L:=[seq(seq([-1+(i-1)*2/3,-1+(j-1)*2/3],i=1..4),j=1..4)]:


M:=map(v->[op(v),g(op(v))],L):


p1:=plot(L,style=point,symbolsize=20,symbol=solidbox,


color="DarkCyan"):


tr:=plottools:-transform((x,y)->[x,y,0]):


p0:=plot3d(0,-1..1,-1..1,color="LightGray",style=


patchnogrid,transparency=.6):


p2:=plots:-pointplot3d(M,symbolsize=30,axes=boxed,symbol=


solidsphere):


plots:-display(tr(p1),p2,p0,mitplot3d);

I det næste eksempel, hvor y-intervallet afhænger af x, bliver plottepunkterne i stedet 
opbevaret i en MESH-struktur, der indeholder en 3-dimensional Array, hvis elementer A(i, j, 
1), A(i, j, 2) og A(i, j, 3) udgør x-, y- og z-koordinaterne for punktet svarende til grid-punktet 
(i, j):
 


mitplot3d:=plot3d(g(x,y),x=-1..1,y=-1..x,axes=boxed,


numpoints=9): 


Igen vil lprint vise PLOT3D-strukturen:


lprint(mitplot3d);


PLOT3D(MESH(Array(1 .. 4, 1 .. 4, 1 .. 3, {(1, 1, 1) = -1., (1,


1, 2) = -1., (1, 1, 3) = -0.183156388887341787e-1, (1, 2, 1) = 


-1., (1, 2, 2) = -1., (1, 2, 3) = -0.183156388887341787e-1, (1,


3, 1) = -1., (1, 3, 2) = -1., (1, 3, 3) = 


-0.183156388887341787e-1, (1, 4, 1) = -1., (1, 4, 2) = -1., (1,


4, 3) = -0.183156388887341787e-1, (2, 1, 1) = 


-.333333333333333370, (2, 1, 2) = -1., (2, 1, 3) = 


-0.148504754148298982e-1, (2, 2, 1) = -.333333333333333370, (2,


2, 2) = -.777777777777777900, (2, 2, 3) = 


-0.293880523126925295e-1, (2, 3, 1) = -.333333333333333370, (2,


3, 2) = -.555555555555555580, (2, 3, 3) = 


-0.364715543353660022e-1, (2, 4, 1) = -.333333333333333370, (2,


4, 2) = -.333333333333333370, (2, 4, 3) = 


-0.237474217937020254e-1, (3, 1, 1) = .333333333333333259, (3, 


1, 2) = -1., (3, 1, 3) = 0.148504754148298928e-1, (3, 2, 1) = 


.333333333333333259, (3, 2, 2) = -.555555555555555692, (3, 2, 


3) = 0.364715543353660022e-1, (3, 3, 1) = .333333333333333259, 


(3, 3, 2) = -.111111111111111216, (3, 3, 3) = 


0.354857348491053552e-2, (3, 4, 1) = .333333333333333259, (3, 







Kapitel 9: Plotning


Kapitel 9 side 16


4, 2) = .333333333333333259, (3, 4, 3) = 


0.237474217937020116e-1, (4, 1, 1) = .99999999999999988, (4, 1,


2) = -1., (4, 1, 3) = 0.183156388887341752e-1, (4, 2, 1) = 


.99999999999999988, (4, 2, 2) = -.333333333333333481, (4, 2, 3)


= 0.292885709017474370e-1, (4, 3, 1) = .99999999999999988, (4, 


3, 2) = .333333333333333148, (4, 3, 3) = 


0.292885709017473954e-1, (4, 4, 1) = .99999999999999988, (4, 4,


2) = .99999999999999988, (4, 4, 3) = 0.183156388887341891e-1}, 


datatype = float[8], storage = rectangular, order = C_order)), 


AXESLABELS(x, y, ``), AXESSTYLE(BOX))


Den 3-dimensionale Array hentes ud af mitplot3d som før:


A:=op(indets(mitplot3d,Array));


Vi laver en illustration ganske analog med den vi lavede for GRID-strukturen:


L:=[seq(seq([seq(A[i,j,k],k=1..2)],i=1..4),j=1..4)]:


M:=[seq(seq([seq(A[i,j,k],k=1..3)],i=1..4),j=1..4)]:


p1:=plot(L,style=point,symbolsize=20,symbol=solidbox,


color="DarkCyan"):


p0:=plot3d(0,x=-1..1,y=-1..x,color="LightGray",style=


patchnogrid,transparency=.6):


p2:=plots:-pointplot3d(M,symbolsize=30,axes=boxed,symbol=


solidsphere):


plots:-display(tr(p1),p2,p0,mitplot3d);

MESH-strukturen bruges også ved parametriske plot.


Der er endnu en tredie PLOT3D-struktur: ISOSURFACE. Her bruges en 3-dimensional grid 
og en dertil svarende 4-dimensional Array, hvis elementer A(i, j, k, 1), A(i, j, k, 2) og A(i, j, k, 
3) udgør x-, y- og z-koordinaterne for punktet svarende til grid-punktet (i, j, k), og A(i, j, k, 4) 
er funktionsværdien i gridpunktet. 
En sådan PLOT3D-struktur fås f.eks. ved brug af implicitplot3d, der kan tegne løsningen til 
en ligning af formen f(x,y,z)=h(x,y,z).  Den nævnte funktion er så F = f-h.
Ved gengivelsen af strukturen findes punkter ved interpolation mellem gridpunkterne for at 
finde punkter hvor F(x,y,z) = 0. 


mitplot3d:=plots:-implicitplot3d(g(x,y)=z,x=-1..1,y=-1..1,


z=-0.03..0.03,axes=boxed,grid=[3,3,3],transparency=.4):


lprint(mitplot3d);


PLOT3D(ISOSURFACE(Array(1 .. 3, 1 .. 3, 1 .. 3, 1 .. 4, {(1, 1,


1, 1) = -1., (1, 1, 1, 2) = -1., (1, 1, 1, 3) = 


-0.299999999999999990e-1, (1, 1, 1, 4) = 


0.116843611112658202e-1, (1, 1, 2, 1) = -1., (1, 1, 2, 2) = -1.


, (1, 1, 2, 4) = -0.183156388887341787e-1, (1, 1, 3, 1) = -1., 


(1, 1, 3, 2) = -1., (1, 1, 3, 3) = 0.299999999999999990e-1, (1,


1, 3, 4) = -0.483156388887341776e-1, (1, 2, 1, 1) = -1., (1, 2,


1, 3) = -0.299999999999999990e-1, (1, 2, 1, 4) = 


0.299999999999999990e-1, (1, 2, 2, 1) = -1., (1, 2, 3, 1) = -1.


, (1, 2, 3, 3) = 0.299999999999999990e-1, (1, 2, 3, 4) = 


-0.299999999999999990e-1, (1, 3, 1, 1) = -1., (1, 3, 1, 2) = 1.


, (1, 3, 1, 3) = -0.299999999999999990e-1, (1, 3, 1, 4) = 


0.116843611112658202e-1, (1, 3, 2, 1) = -1., (1, 3, 2, 2) = 1.,


(1, 3, 2, 4) = -0.183156388887341787e-1, (1, 3, 3, 1) = -1., 


(1, 3, 3, 2) = 1., (1, 3, 3, 3) = 0.299999999999999990e-1, (1, 


3, 3, 4) = -0.483156388887341776e-1, (2, 1, 1, 2) = -1., (2, 1,


1, 3) = -0.299999999999999990e-1, (2, 1, 1, 4) = 


0.299999999999999990e-1, (2, 1, 2, 2) = -1., (2, 1, 3, 2) = -1.


, (2, 1, 3, 3) = 0.299999999999999990e-1, (2, 1, 3, 4) = 


-0.299999999999999990e-1, (2, 2, 1, 3) = 


-0.299999999999999990e-1, (2, 2, 1, 4) = 


0.299999999999999990e-1, (2, 2, 3, 3) = 


0.299999999999999990e-1, (2, 2, 3, 4) = 


-0.299999999999999990e-1, (2, 3, 1, 2) = 1., (2, 3, 1, 3) = 


-0.299999999999999990e-1, (2, 3, 1, 4) = 


0.299999999999999990e-1, (2, 3, 2, 2) = 1., (2, 3, 3, 2) = 1., 


(2, 3, 3, 3) = 0.299999999999999990e-1, (2, 3, 3, 4) = 


-0.299999999999999990e-1, (3, 1, 1, 1) = 1., (3, 1, 1, 2) = -1.


, (3, 1, 1, 3) = -0.299999999999999990e-1, (3, 1, 1, 4) = 


0.483156388887341776e-1, (3, 1, 2, 1) = 1., (3, 1, 2, 2) = -1.,


(3, 1, 2, 4) = 0.183156388887341787e-1, (3, 1, 3, 1) = 1., (3, 

1, 3, 2) = -1., (3, 1, 3, 3) = 0.299999999999999990e-1, (3, 1, 


3, 4) = -0.116843611112658202e-1, (3, 2, 1, 1) = 1., (3, 2, 1, 


3) = -0.299999999999999990e-1, (3, 2, 1, 4) = 


0.299999999999999990e-1, (3, 2, 2, 1) = 1., (3, 2, 3, 1) = 1., 


(3, 2, 3, 3) = 0.299999999999999990e-1, (3, 2, 3, 4) = 


-0.299999999999999990e-1, (3, 3, 1, 1) = 1., (3, 3, 1, 2) = 1.,


(3, 3, 1, 3) = -0.299999999999999990e-1, (3, 3, 1, 4) = 


0.483156388887341776e-1, (3, 3, 2, 1) = 1., (3, 3, 2, 2) = 1., 


(3, 3, 2, 4) = 0.183156388887341787e-1, (3, 3, 3, 1) = 1., (3, 


3, 3, 2) = 1., (3, 3, 3, 3) = 0.299999999999999990e-1, (3, 3, 


3, 4) = -0.116843611112658202e-1}, datatype = float[8], storage


= rectangular, order = C_order)), TRANSPARENCY(.4), AXESLABELS


("x", "y", "z"), AXESSTYLE(BOX))


A:=op(indets(mitplot3d,Array));


Vi laver en illustration analog med den vi lavede for GRID- og MESH-strukturerne. Der er 
dog en forskel: A indeholder kun gridpunkterne (x, y, z) og de tilsvarende funktionsværdier F
(x, y, z). Som nævnt skal der interpoleres for at finde de punkter, hvor F(x, y, z) = 0.


Med de 3 første indices (1,1, 1) fås 


x1,y1,z1,F1:=seq(A[1,1,1,k],k=1..4);


Dette er altså i vores tilfælde (x, y, z, g(x,y) - z): 


g(x1,y1)-z1-F1;
0.


Med de 3 første indices (1,1, 2) fås tilsvarende


x2,y2,z2,F2:=seq(A[1,1,2,k],k=1..4);


g(x2,y2)-z2-F2;
0.


Her vises så gridpunkterne sammen med resulatet af interpolationen: fladen.


M:=[seq(seq(seq([seq(A[i,j,k,m],m=1..3)],i=1..3),j=1..3),


k=1..3)]:


p2:=plots:-pointplot3d(M,symbolsize=30,axes=boxed,symbol=


solidsphere):


plots:-display(p2,mitplot3d);

Specielle plotte-procedurer: plots-pakken


Mange af Maple's mere specielle procedurer er anbragt i pakker. En af disse pakker hedder plots. 
Maple kender ikke navnene på procedurerne i plots-pakken før kommandoen with(plots);  
udstedes. Men så kommer Maple med en liste af navnene på de procedurer, der ligger i pakken. 
Hvis ikke denne liste ønskes bragt, bruges blot kolon i stedet for semikolon.


with(plots);


Vi kan nu bruge alle de anførte procedurer. Man kan også bruge procedurerne uden først at have 
sagt with(plots);  men så hedder eksempelvis proceduren logplot i stedet plots[logplot]  eller
plots:-logplot (man taler i Maple om procedurens lange form). 


Enkelt-logaritmisk afbildning fås ved proceduren logplot. Tilføjelserne labels=["x",
"logaritmisk"] og labeldirections=[horizontal, vertical] er valgfri. 


logplot( 3*exp(-x), x=-2..2,labels=["x","logaritmisk"],


labeldirections=[horizontal,vertical], gridlines=true);

Se også hjælpen om semilogplot og loglogplot. 
Et plot i polære koordinater kan (som tidligere vist) fås ved som ekstra argument til plot at føje
coords=polar. Der findes imidlertid også en speciel procedure til formålet:


polarplot(1+cos(t), t=0..2*Pi, title="Cardioiden",titlefont=


[TIMES,ITALIC,12]);


0 1
0

Cardioiden


Det eksakt samme plot kan fås ved kommandoen (men output er her undertrykt):


plot(1+cos(t), t=0..2*Pi, coords=polar,axiscoordinates=


polar,title="Cardioiden",titlefont=[TIMES,ITALIC,12]):


Bemærk hvordan én kommando godt må fylde flere linier:


polarplot([1+cos(t), sin(t)], t=0..2*Pi, 


title="Cardioide med cirkel",titlefont=[COURIER,12]); 

Cardioide med cirkel


Er der givet en ligning for en kurve (eksempelvis  ) og kan hverken x eller y 


isoleres, så kan kurven alligevel tegnes, nemlig ved brug af implicitplot. De to ekstra argumenter
filled=true og coloring=.......  er valgfrie.


implicitplot( y*x^2-sin(x+y)=0, x=-1..1, y=-3.5..3.5,filled=


true,coloring=["CadetBlue","Yellow"]);


Der er et væld af valgfri tilføjelser. Se detaljerne i hjælpen. Her et eksempel: 
Med tilføjelsen gridrefine = 3 forbedres kurven nær det i dette tilfælde problematiske punkt (0, 0):


implicitplot(sin(x^3)=y^4, x=0..1, y=-1..1,gridrefine=3);

Man kan følge lidt med i hvad der foregår i denne animation:


for k from 0 to 4 do 


   p[k]:=implicitplot(sin(x^3)=y^4, x=0..1, y=-1..1,


          grid=[3,3],gridrefine=k,outlines=true,title=cat


("gridrefine = ",k));


end do:


display([seq(p[k],k=0..4)],insequence=true);


x


1


y


0


1


gridrefine = 1


Af hensyn til en statisk version af dette dokument bringer vi også de første fire billeder i 
animationen.
display kan vise en Matrix/Array af plots.  
Array'en eller matricen kan angives med den sædvanlige syntaks (se kapitlet om Lineær Algebra).
Her kan f.eks. skrives display(Array([[p[0],p[1]],[p[2],p[3]]]));
Men også den forenklede syntaks:


display(<p[0],p[1];p[2],p[3]>);

Med kommandoen display(Vector[row]([p[0],p[1],p[2]])); fås de 3 første plot på række.


Datastruktur for implicitplot
Datastrukturen for et plot produceret af implicitplot er ikke meget anderledes end det, der 
kommer fra plot. Begge har formen PLOT(CURVES(A, ...),..).
Her er A i tilfældet implicitplot en 2-dimensional Array, altså en matrix, hvor A ved plot er en
listlist.


p:=implicitplot( y*x^2-sin(x+y)=0, x=-1..1, y=-3.5..3.5,


grid=[3,3]);


p;


x


0 1


y
1


2


lprint(p);


PLOT(CURVES(Array(1 .. 5, 1 .. 2, {(1, 1) = -1., (1, 2) = 


-.553703296932638866, (2, 1) = -.294218484625694332, (2, 2) = 

-2.47023530381006974, (4, 1) = .294218484625694332, (4, 2) = 


2.47023530381006974, (5, 1) = 1., (5, 2) = .553703296932638866}


, datatype = float[8], storage = rectangular, order = C_order, 


attributes = [source_rtable = Array(1 .. 9, 1 .. 2, {(1, 1) = 


5., (2, 1) = -1., (2, 2) = -.553703296932638866, (3, 1) = 


-.294218484625694332, (3, 2) = -2.47023530381006974, (5, 1) = 


.294218484625694332, (5, 2) = 2.47023530381006974, (6, 1) = 1.,


(6, 2) = .553703296932638866}, datatype = float[8], storage = 


rectangular, order = C_order)]), COLOUR(RGB, 1., 0., 0.)), 


AXESLABELS(x, y))


Vi konverterer Array'en til en listlist:


evalindets(p,Array,s->convert(s,listlist));


x


0 1


y
1


2


indets(%,listlist);


implicitplot kan tegne flere kurver på én gang. De tilhørende ligninger kan da anbringes i 
mængdeklammer eller listeklammer. De sidste anbefales pga. den deraf følgende mulighed for at 
give kurverne forskellige farver m.m. :


implicitplot([y*x^2-sin(x+y)=0, y^2-5*x*(x^2-1)=0],x=-1..1,


y=-3.5..3.5,color=["DarkOliveGreen","NavyBlue" ],linestyle=


[SOLID,DASHDOT],thickness=[2,5]);

Der findes også en procedure, der hedder implicitplot3d for ligninger af typen f(x,y,z)=h(x,y,z).
Her er et eksempel med en del valgfrie tilføjelser:


implicitplot3d(x^2+2*y^2-z^2=2,x=-3..3,y=-3..3,z=-3..3,


axes=framed,orientation=[40,60],shading=ZHUE);


Niveaukurver for en funktion af to variable kan enten fås ved manipulation på skærmen af 
resultatet af en plot3d-kommando eller ved brug af proceduren contourplot fra plots-pakken. 
Proceduren contourplot3d tegner niveaukurverne i de højder (niveauer), de svarer til. Der fås 
altså en tredimensional tegning. Ses denne lige ovenfra, er resultatet det samme som for
contourplot. Overraskende nok er contourplot3d hurtigere end contourplot. Med orientation=
[-90,0] ses landskabet fra luften, men det kan drejes med musen:


contourplot3d( sin(x+y)*cos(y), x=-Pi..Pi, y=-Pi..Pi, axes=


normal, orientation=[-90,0],color="Black"); 

contourplot(sin(x+y)*cos(y), x=-Pi..Pi, y=-Pi..Pi,filled=


true,coloring=["Turquoise","Moccasin"]); 


Ved tilføjelsen contours=[-2,-1,0,1,2,3]  får man niveaukurverne med de angivne niveauer. Man 
kan i stedet bede om et vist antal, eksempelvis 40, ved tilføjelsen contours=40. 


inequal, complexplot, complexplot3d
Proceduren inequal fra plots-pakken kan tegne områder i planen givet ved lineære uligheder:


inequal( { x+y>0, x-y<=1,y<2}, x=-3..4, y=-2..3,


optionsexcluded=(color="Yellow"),optionsfeasible=(color=


"Red"));

Proceduren complexplot fra plots-pakken kan tegne en kurve eller en liste af punkter i den 
komplekse plan. 


I eksemplet nedenfor afbilder vi løsningerne til den binome ligning  i planen. Derefter 


tegner vi den cirkel, , hvorpå rødderne ligger og viser de to plot sammen. Vi benytter 


også lejligheden til at vise brugen af proceduren textplot.


ligning:=z^6+1=0;


L:=evalc([solve( ligning, z)]);


complexplot(L, style=point, symbol=soliddiamond,


symbolsize=25,color="Blue");


0


1


punktplot:=%:


cirkelplot:=complexplot( exp(I*t), t=0..2*Pi,linestyle=2,


color=gray):

Strukturen i listerne nedenfor er [x, y, tekst]. Teksten i 3. element er typisk en tekststreng, men
kan også være et algebraisk udtryk, der printes flot, hvis typeset bruges.
Polær form for rødderne til printning. Her er brugt pi i stedet for Pi for at forhindre evaluering 
til rektangulær form:


Lp:=seq(exp(pi*I*(1/6+k/3)),k=0..5);


tekst:=textplot([[0.6,0.5,typeset(Lp[1])],[0,0.85,typeset


(Lp[2])],[-0.6,0.5,typeset(Lp[3])],[-0.6,-0.5,typeset(Lp


[4])],[0,-0.8,typeset(Lp[5])],[0.6,-0.5,typeset(Lp[6])]]):


display([punktplot,cirkelplot,tekst],caption=typeset


("Rødderne til  ",z^6+1=0),tickmarks=[[-1=-1,-1/2=-1/2,


1/2=1/2,1=1],[.5=I/2,1=I,-1=-I,-0.5=-I/2]]); 


e


e


e


e


e


e


1


2


1


I


Rødderne til  


complexplot kan tegne flere kurver på en gang ligesom plot.
Prøv også proceduren complexplot3d. Det, der tegnes i eksemplet nedenfor, er grafen for 
funktionen


 


complexplot3d( cos(z), z=-Pi-2*I..Pi+2*I,view=0..2,axes=


boxed,style=hidden,color="Black");

Et eksempel på brugen af conformal:


a:=Pi/2:  b:=2:


conformal(z,z=-a-b*I..a+b*I,caption="z-planen"):


conformal(sin(z),z=-a-b*I..a+b*I,scaling=constrained,


caption=typeset("w-planen  ",w=sin(z))):


display(Array([%%,%]));


1


1


2


z-planen


1 2 3


1


3


w-planen  


a:='a': b:='b':


Animation


Har man for enhver værdi af en parameter a et udtryk u(a), der kan plottes af en plotteprocedure, 
så kan animate lave en række plots, der vises som en animation i animationsparameteren a.  
Første argument til animate er den plotteprocedure, som man vil animere, andet argument er en 
liste bestående af argumenterne til plotteproceduren, tredie argument er animationsparameteren. 
Yderligere argumenter er valgfrie.
I de første eksempler er plotteproceduren plot.


animate(plot,[sin(x^3)*exp(-(x/a)^2)/a, x=-3..3], a=0.5..3);

Default for antal billeder er frames=25. Maple laver altså i dette tilfælde 25 grafer, hver svarende 
til sin a-værdi i intervallet [0.5, 3]. Filmen kan nu afspilles på forskellig måde ved først at klikke 
med musen på figuren og derefter klikke på "videoafspillerens" knapper. 


En animation kan forsynes med en baggrund. Her bruger som eksempel frembringelsen af en 
hypocykloide. Vi definerer først baggrunden, som her blot er en cirkel med centrum i (0,0) og 
radius 4:


baggrund:=plot(4,theta=0..2*Pi,coords=polar,color=


"OrangeRed",thickness=2):


Her følger så animering med baggrund. Det vi ser, er en lille cirkel, der ruller på indersiden af den
store cirkel:


nf:=50:


animate(plot,


[[3*cos(a)+cos(t),3*sin(a)+sin(t),t=0..2*Pi],color=


"SeaGreen",thickness=3], 


a=0..2*Pi,frames=nf, background=baggrund ,paraminfo=false);

Vi gemmer denne animering i variablen p1, da vi ønsker at vise den sammen med to andre 
animeringer:


p1:=%:


Vi lader nu et fast punkt på den lille cirkel følge med bevægelsen:


p2:=animate(plot,


[[[3*cos(a)+cos(3*a),3*sin(a)-sin(3*a)]],style=point,


symbolsize=25,symbol=solidcircle,color="DarkMagenta"], 


a=0..2*Pi,frames=nf,paraminfo=false):


Vi kan se p2 blot ved kommandoen:


p2; # Ikke vist


Vi kan vise de to animeringer p1 og p2 sammen således:


display(p1,p2); # Ikke vist


Her animerer vi det faste punkts banekurve:


p3:=animate(plot,[[3*cos(aa)+cos(3*aa),3*sin(aa)-sin(3*aa),


aa=0..a],color="DarkMagenta",thickness=2], a=0..2*Pi,frames=


nf,paraminfo=false ):


Her viser vi så alle tre animeringer sammen:


display(p1,p2,p3,title="Hypocykloide");

Hypocykloide


Farven kan afhænge af animationsparameteren. Bagrunden kan også blot være et billede, der 
svarer til en animationsparameterværdi. Her svarende til a = 1.
Plotteproceduren er polarplot.


animate(polarplot,


[1+cos(a*theta),theta=0..2*Pi,color=COLOR(RGB, 1-a/2, a^2/4,


a/2),thickness=2], 


a=0..2,frames=50,background=1);

Her er animationer af implicitplot og plot3d:


animate(implicitplot,[y*x^2-sin(a*x+y)=0, x=-4..4,y=-4..4, 


gridrefine=2,thickness=2], a=0..3);


x

animate(plot3d,[x^2+y^2,x=-s..s,y=-s..s,axes=boxed],s=0..2);

En forudsætning for, at animate virker er, at indholdet i andet argument giver Maple-mening selv 
for symbolske værdier af animationsparameteren. Grunden er den, at andet argument evalueres 
før animationsparameteren får tildelt konkrete talværdier. Her følger animate den sædvanlige 
Maple-regel om fuld evaluering af argumenterne før proceduren (her animate) går i gang. 
Da animate leverer decimalbrøksværdier af animationsparameteren til plotteproceduren, er det 
også en forudsætning, at denne virker på decimalbrøksværdier.
Disse to forudsætninger kan man dog komme udenom ved at definere sin egen plotteprocedure 
(nedenfor kaldet p), der til enhver konkret værdi af en animationsparameter (i eksemplet n) leverer
et plot.
I eksemplet er det klart, at antallet af rammer (frames) skal være 9, når animationsparameteren 
gennemløber intervallet 1..9.
Man kan bede om at et udvalg af billederne forbliver på skærmen efter fremkomst ved tilføjelsen
trace=antal.


bg:=plot(exp(x),x=-1..3,color=blue):


p:=n-> plot(mtaylor(exp(x),x=0,round(n)+1),x=-1..3,color=


COLOR(HUE,n/10),thickness=2,


             caption=typeset(P[round(n)](x)=mtaylor(exp(x),


x=0,round(n)+1))):


animate(p,[n], n=1..9,frames=9, trace=8,background=bg);


x

Animering kan også opnås ved en anden metode, hvor man først definerer en række plot, og derpå
viser dem i rækkefølge. 
Man opbevarer hvert plot i en variabel. Flere plot kan defineres på én gang ved simpel 
programmering. 


I eksemplet nedenfor plottes udtrykkene  osv. Vi gør brug af 


adderingsproceduren add:


add(x^k,k=0..10);


For overskuelighedens skyld definerer vi først følgen bestående af de 12 udtryk, vi vil plotte. Vi 
slutter kommandoen med kolon, da output er temmelig stort. Prøv selv at sætte et semikolon i 
stedet:


f:=seq( add(x^k,k=0..n), n=0..11):


Nu defineres følgen af plot:


p:=seq( plot(f[n], x=-1..0,y=0..1,title=cat("Antal led = ",


n)) ,n=1..12): 


Vil vi se et enkelt af plottene, behøver vi blot nævne det:


p[3]; 


x


0


y


1


Antal led = 3


Først viser vi alle plottene oven i hinanden:


display( p,title="Alle på én gang");

Alle på én gang


Vi har bedt display om en titel, da den ellers vælger den første af de 12 plot-titler, hvilket ville 
være vildledende.
display accepterer en følge af plot eller en liste af plot, altså enten display(p1,p2,p3);  eller 
display( [p1,p2,p3] );
Med tilføjelsen insequence=true fås en animation:


display( p, insequence=true); 


x


0


y


1


Antal led = 5


Vi gemmer resultatet til senere brug:


tegnefilm:=%: 


Ofte ønskes en animation set på en fast baggrund. I dette tilfælde vil vi gerne have et plot af 


udtrykket     i baggrunden. Der gælder nemlig, at


for |x| < 1. Gem (som vist) animationen i en variabel (hér tegnefilm). Definér baggrunden:

baggrund:=plot(1/(1-x),x=-1..0,color="Green"):


Følgende kommando giver nu filmen med den faste baggrund (hér kaldet baggrund):


display(tegnefilm,baggrund);


x


0


1


Antal led = 4


Der er her vist et af billederne.


Vil man have vist alle tegnefilmens billeder på én gang (men ikke oven i hinanden) kan dette 
gøres ved at anbringe plottene i en to-dimensional array:


A:=Array([[p[1..4]],[p[5..8]],[p[9..12]]]):


display(A);

Variabelfri syntaks og for tidlig evaluering


Maple's plot-procedure tillader som nævnt i indledningen både at funktionsvariable inkluderes og 
ekskluderes. Man får sinusfunktionens graf ved begge de to kommandoer:

plot( sin(x), x=-Pi..Pi,caption="Med variabel x. Default er,


at variabelnavnet ses under x-aksen."):


plot( sin, -Pi..Pi,caption="Variabelfri syntaks"):


display(<%%|%>);


x


1 2 3


1


Med variabel x. Default er, at 


variabelnavnet ses under x-aksen.


1 2 3


1


Variabelfri syntaks


Ved plotningen af visse funktioner (procedurer) er den variabelfri syntaks næsten en 


nødvendighed. I det følgende eksempel defineres f(t) som løsningen  til ligningen  , 


beliggende i intervallet ]0, ].  f er altså den omvendte funktion til funktionen  defineret


på intervallet ]0, ]. For ethvert konkret givet  i intervallet [0, 1[ må vi løse ligningen numerisk. 


Dette sker vha. fsolve. Denne procedure kan man give et søgeinterval, her [0, ].


f:= t-> fsolve(sin(x)/x=t,x=0..Pi);


f(.5);
1.895494267


Vi prøver at plotte grafen for f  med følgende syntaks:


plot(f(t),t=0..1);


Error, (in fsolve) t is in the equation, and is not solved for


Vi får en lidt kryptisk fejlmeddelelse. Det der sker er dette: Maple prøver først at evaluere f(t), 


hvor t blot er bogstavet t. Men dette betyder, at fsolve skal løse ligningen   numerisk 


med hensyn til x. Dette er umuligt. En numerisk løsning kan kun findes, når t har en konkret 
talværdi.
Der er to løsninger på dette problem. Den ene er at omgærde f(t) med forhalende apostroffer. Den 
første evaluering fjerner da blot apostrofferne fra 'f(t)'. 
En mere elegant løsning er at bruge variabelfri syntaks. 
Vi viser begge metoder.


Først med forhalende apostroffer:


plot('f(t)',t=0..1,caption="Med forhalende apostroffer om f

(t)"):


plot(f,0..1,caption="Variabelfri syntaks"):


display(<%%|%>);


t
0 1

Med forhalende apostroffer om f(t)


0 1


1


2


3


Variabelfri syntaks


Bemærk iøvrigt, at det x, der blev brugt ved definitionen af f, er en global variabel. Dette kan have 
uheldige konsekvenser:


f(x);
0.8767262154


Så tror man jo, at f er konstant. Der sker det, at f(x) bliver fsolve( sin(x)/x=x, x=0..Pi). Ligningen 


 har en løsning, nemlig .


Det ideelle ved definitionen af f er selvfølgelig, at x blev erklæret lokal. Dette er imidlertid ikke 
muligt, når der bruges piledefinition, men kan gøres ved egentlig proceduredefinition. Samtidig 
sørger vi for, at proceduren kun accepterer numerisk input:


ff:=proc(t::numeric) local x; fsolve(sin(x)/x=t,x=0..Pi) end


proc;


ff(x);


Error, invalid input: ff expects its 1st argument, t, to be of 


type numeric, but received x


ff(.5);
1.895494267


En lidt mere avanceret version er følgende:


ff2:=proc(t) local x; 


   if not type(t,numeric) then 


      'procname'(t) 


     else 


      fsolve(sin(x)/x=t,x=0..Pi) 


   end if 


end proc;

Proceduren ff2 er lavet, så den returnerer uevalueret, når input ikke er af typen numeric (rationale 
tal eller decimaltal). Dette betyder også, at vi igen kan benytte plot-proceduren med variabel:


ff2(x);


ff2(.5);
1.895494267


Som omvendt funktion til funktionen  kan grafen for f selvfølgelig meget lettere 


plottes parametrisk:


plot(ff2(x),x=0..1,caption="Plot af ff2(x)"):


plot([sin(x)/x,x,x=0..Pi],caption=typeset("Det parametriske 


plot  ",[sin(x)/x,x,x=0..Pi])):


display(<%%|%>);


x
0 1


1


2


3


Plot af ff2(x)


0 1


3


Det parametriske plot  

10. Algebraisk Manipulation


Maple har en lang række faciliteter til algebraiske manipulationer. I dette kapitel nævnes nogle af de 
vigtigste. I en vis forstand kan det dog siges, at næsten hele bogen handler om algebraisk 
manipulation, idet et computeralgebraprograms primære opgave netop er algebraisk manipulation.


Proceduren factor


Betragt følgende 7. grads polynomium:


p:=4*x^7+24*x^6+44*x^5-8*x^4-132*x^3-184*x^2-108*x-24;


Proceduren factor faktoriserer p:


factor(p);


men bemærk, at der ikke er opnået en fuldstændig faktorisering indenfor de reelle tal, idet  
ikke er blevet faktoriseret. Det er fordi factor af koefficienterne til p udleder, at man ønsker en 
faktorisering indenfor de hele (eller rationale) tal. Hvordan dette brydes vises lidt senere.


De følgende to polynomier afviger kun fra hinanden ved, at koefficienten til   i  p1 er brøken 


 , mens den tilsvarende koefficient for p2 er decimaltallet  . Proceduren factor behandler


de to polynomier forskelligt, idet den faktoriserer indenfor det talområde, der fremgår af 
koefficienterne:


p1:=x^4-10/21*x^3+419/84*x^2-50/21*x-5/84;


p2:=x^4-10.0/21*x^3+419/84*x^2-50/21*x-5/84;


I p1 er koefficienterne rationale tal, så her faktoriseres indenfor disse:


factor(p1);


I p2 er koefficienterne "flydende" decimalbrøker (floating point numbers), så her faktoriseres 
indenfor disse:


factor(p2);


Hvis ikke Digits er sat til andet, så har den værdien 10. Vi sætter den til 15, så Maple regner med 

15 betydende cifre:


Digits:=15:


Vi vil igen faktorisere p2. Proceduren factor husker imidlertid allerede udførte faktoriseringer, 
hvilket ses af dens husketabel:


op(4, eval(factor));


så vi må først bede Maple om at fjerne husketabellen for factor:


forget(factor);


op(4, eval(factor));


(der kom intet).


factor(p2);


Vi sætter Digits tilbage til 10:


Digits:=10:


 
Denne skelnen mellem et helt tal (eller rationalt tal) og dettes (nødvendigvis endelige) 


decimalbrøksversion må man være opmærksom på overalt i Maple. Mener man , så skriver 


man 1/2 og ikke 0.5.
Default ved faktorisering er altså, at koefficienterne bestemmer talområdet (tallegemet). For at 
bryde dette kan man tilføje et ekstra argument til proceduren factor. Derved kan man fortælle 
Maple, at man tillader et større tallegeme i faktoriseringen end det, der aflæses af koefficienterne 
til polynomiet. Vi giver nogle eksempler.


factor(p1,real);


factor(p1,complex);


identify(%);


factor(x^2-3, sqrt(3));


factor(p1,{I,sqrt(5)});

Bemærk iøvrigt, at factor ikke faktoriserer hele tal. Vil man have tallet 12 skrevet som  skal 
man bruge proceduren ifactor:


ifactor(12);


Om behandlingen af tal, se i øvrigt kapitel 11.


Proceduren expand


Betragt udtrykket


u1:=(a+b-2*c)^2*(c+a);


expand vil gange udtrykket ud:


expand(%);


Betragt nu udtrykkene


u2,u3,u4:=sin(a+b), exp(a+b), ln(a*b);


expand splitter op efter gældende regler:


expand(u2);


expand(u3);


I udtrykket u4  må vi gøre en forudsætning, f.eks. at  a eller b er positiv, ellers sker der ikke noget:


expand(u4) assuming a>0;


Betragt nu udtrykket


u:=(a+b)^2*exp(c-d);


Vi ønsker udtrykket u ganget ud:


expand(u);


Virkningen af expand er en totalopsplitning. Lad os nu sige, at vi faktisk kun ønskede  


ganget ud, og intet gjort ved .  Så tilføjer vi som andet argument til expand det, den ikke skal 
røre ved. Her kan vi enten tilføje exp(c-d) eller blot exp:


expand(u, exp);

Et eksempel mere:


expand( u2*u3,u2);


Proceduren simplify
restart;


Følgende udtryk ønskes simplificeret.


u1:= ln(2)-ln(7)+exp(a)*exp(b-a)+sin(x)^3+2*cos(x)^2-cos(2*


x);


Ved simplifikation kan forstås mange ting. Maple har adskillige procedurer, der kan bruges til 
simplifikationer (omskrivninger), heriblandt simplify.   


simplify(u1);


simplify giver ikke nødvendigvis det, man selv mener er simplest!


Man kan anvende et ekstra argument (som f.eks. ln, exp, power, radical, sqrt eller trig) i  
proceduren simplify, og derved opnå en vis styring af simplificeringen, som i følgende eksempel:


simplify( u1, trig);


simplify(u1,exp);


For flere detaljer herom se hjælpen under simplify eller hjælpen under simplify med et af de 
ekstra argumenter (hjælpen for simplify med exp mangler):


?simplify


?simplify,trig


Betragt udtrykket


u2:=sqrt(x^2);


simplify(u2);


Den komplekse fortegnsfunktionen csgn har værdien 1 i den åbne højre halvplan og -1 i den åbne 
venstre halvplan. På den imaginære akse er den 1 på den positive del og -1 på den negative:
Vil man lokalt gøre brug af en antagelse, kan dette gøres ved brug af proceduren  assuming, der 
har en lidt anderledes syntaks. Den virker således:

simplify(u2) assuming x<0;


Her gøres under simplifikationen den antagelse, at x < 0.  
Man kan nøjes med


simplify(u2) assuming negative;


Under simplifikationen vil Maple nu antage, at samtlige ubekendte (samtlige variable) er 
negative.
En tilsvarende effekt opnås ved som andet argument til simplify at anføre en tilføjelse af formen:
assume=egenskab. 


simplify(u2, assume=[x<0]);


eller blot


simplify(u2,assume=negative);


Endnu et eksempel. Betragt udtrykket


u3:=sqrt((a-7)^2*(x^2+6*x+9));


simplify(u3);


Hvis vi ved, at  og , kan vi enten bruge assuming:


simplify(u3) assuming a>=7, x<-3;


eller anbringe en antagelse som andet argument:


simplify(u3,assume=[a>=7, x<-3]);


Af disse to måder at gøre lokale antagelser på, er den sidste den ældste. 


En global antagelse kunne være gjort før brugen af simplify således:


assume(a>=7,x<-3);


simplify(u3);


Tilde-tegnet på a og x skal minde os om, at der er gjort antagelser:


about(a); about(x);


Originally a, renamed a~:


  is assumed to be: RealRange(7,infinity)







Kapitel 10: Algebraisk Manipulation


Kapitel 10 side 6


Originally x, renamed x~:


  is assumed to be: RealRange(-infinity,Open(-3))


Vi fjerner de globale antagelser:


a:='a':   x:='x':


about(a); about(x);


a:


  nothing known about this object


x:


  nothing known about this object


At antagelser kan gøres inden i kaldet til simplify er ikke blevet overflødig efter introduktionen af
proceduren assuming, som følgende eksempel viser:


restart;


f:=x->a*x:


simplify(sqrt(f(a)),assume=[a<0]);


simplify(sqrt(f(a))) assuming a<0;


Grunden til den forskellige opførsel er følgende: 


I den første version evalueres f(a) til  før simplify kommer igang. Med antagelsen a < 0 får
simplify derefter resultatet - a.
I den anden version bliver f(a) ikke evalueret før antagelsen er gjort om det a, der står som 
argument til f. Det a, der optræder i selve definitionen af f , bliver der ikke gjort antagelser om.


simplify kan så ikke gøre noget ved  da den intet ved om a uden tilde. Til sidst før
assuming er færdig med sit arbejde bliver a~ erstattet af a. 
Denne opførsel kan dokumenteres ved brug af kommandoen debug(`assuming`) (efter restart).


Proceduren simplify kan også bruges med det ekstra argument symbolic. Så bekymrer Maple sig 
ikke om, hvilken gren den skal vælge af en flertydig funktion (for detaljer, se hjælpen til simplify)
:
 


simplify( arcsin(sin(x)), symbolic);
x


simplify( sqrt((x-5)^2), symbolic);


simplify( ln(a*b)+sqrt((x-5)^2), sqrt,symbolic);


simplify( ln(a*b)+sqrt((x-5)^2), symbolic);

Iøvrigt kan kvadratrodsfunktionen sqrt selv kaldes med det ekstra argument symbolic:


sqrt( x^2+4*x+4, symbolic);


Simplifikation med sidebetingelser:


simplify(a+a^2*b^2,{a*b=2});


assume-faciliteten


Under omtalen af proceduren simplify ovenfor har vi set eksempler på brugen af lokale og globale
antagelser i forbindelse med simplifikation.


Proceduren assuming har kun lokal virkning. De gjorte antagelser bruges kun i den pågældende 
regning. Bagefter er de glemt. Proceduren kan også bruges sammen med andre procedurer end 
lige simplify. 
Eksempler:


int(1/x,x=1..a);


Warning, unable to determine if 0 is between 1 and a; try to use 


assumptions or use the AllSolutions option


regnes ikke ud, men med antagelsen a > 0 fås


int(1/x,x=1..a) assuming a>0;


Her tages stilling til alle værdier for a (hvilket også gøres hvis _EnvAllSolutions sættes til true):


int(1/x,x=1..a,AllSolutions);


% assuming a>0;


limit(x^n*exp(a*x),x=0) assuming n>0;
0


limit(x^n*exp(a*x),x=0,right) assuming n<0;


limit(x^n*exp(a*x),x=infinity) assuming a<0;
0


Globale antagelser gøres ved brug af proceduren assume:


assume(n>0, a<0); 







Kapitel 10: Algebraisk Manipulation


Kapitel 10 side 8


Bemærk, at der intet output kom. Der bliver heller ikke anbragt noget nyt i %.


Limit(x^n*exp(a*x),x=infinity): %=value(%);


Vi ser, at der til n og til a nu er føjet en tilde (~) for at erindre om, at antagelser er gjort -  indtil de 
hæves.
Tilde-tegnet kan (i Standard Worksheet Maple) skjules ved at gå ind i Tools/Options/Display/ 
Assumed variables. Her kan man så vælge No Annotation. 
Det kan også gøres ved følgende kommando, der har samme virkning. Se iøvrigt hjælpen for 
information om interface.


interface( showassumed=0);
1


a,n;


Vi får tilderne tilbage således:


interface( showassumed = 1); 
0


a,n; 


Vha. proceduren about kan man se, hvilke antagelser, man har gjort om sine variable:


about(a,n);


Originally a, renamed a~:


  is assumed to be: RealRange(-infinity,Open(0))


Originally n, renamed n~:


  is assumed to be: RealRange(Open(0),infinity)


Vil man fjerne selve antagelsen om n (uanset om tilden vises eller ej),  kan det gøres som følger 
(bemærk apostrofferne):


n:='n'; 


about(n);


n:


  nothing known about this object


Vi fjerner også antagelsen om a:


a:='a';

combine, normal, radnormal


Vi betragter igen følgende udtryk, som vi ønsker simplificeret


u:= ln(2)-ln(7)+exp(a)*exp(b-a)+sin(x)^3+2*cos(x)^2-cos(2*x)


;


simplify kan være udmærket at prøve, men initiativet overlades dermed til Maple. Man har ikke 
den samme kontrol med omformningerne, som man har med eksempelvis combine, factor, 
normal eller collect:


Ønskes udtrykket forsøgt "samlet sammen" (kombineret) kan bruges combine:  


combine(u);


Hvis det ønskes, at kun trigonometriske funktioner skal kombineres, så tilføjes trig. Nogle af de 
andre mulige tilføjelser er exp, ln, power, radical, abs. Tilføjelserne kan være flere på én gang. 
De skrives så i en liste, som [ln, exp] i tredie eksempel nedenfor. 


combine(u,trig); 


combine(u, ln);


combine(u,[ln,exp]);


Proceduren normal sætter (bl.a.) på fælles brøkstreg og forkorter fælles faktorer:


A:=a^2/(b-a)-b^2/(b-a);


normal(A); 


B:=c/(a-b)-1/(a+b);

Der sættes på fælles brøkstreg, men fællesnævneren ganges ikke ud, hvilket heller ikke altid er 
nogen fordel:


normal(B); 


Forsøg på direkte anvendelse af expand har ikke den ønskede virkning:


expand(%); 


Men tilføjelsen expanded gør, at nævneren ganges ud:


normal(B, expanded); 


Proceduren radnormal er egnet til udtryk med rødder. I dette eksempel bruges det med tilføjelsen
rationalized, hvorved nævneren gøres rational:


(sqrt(2)+1)/(sqrt(2)-1);


radnormal(%,rationalized);


Bemærk i den forbindelse, at Maple ved evalueringen af , hvor n er hel og , skaffer 


rational nævner:


1/sqrt(5),1/sqrt(-2);


5^(-1/3), (-2)^(-1/5);


Der er ikke tale om såkaldt "automatisk simplifikation" (se senere i kapitlet), hvorved forstås en 
simplifikation, der sker inden evaluering. Dette kan ses ved at anbringe apostroffer om udtrykket:


'1/sqrt(2+3)'=1/sqrt(2+3);


2 + 3 blev automatisk simplificeret til 5, men mere skete der ikke på venstre side.


collect, coeff


Lad p være udtrykket:

p:= 7*x*y^2+a*x*y^2+3*y^3*x^2-a*y*x^2+4*x+a*y^3*x+7*y;


Vi kan samle leddene i  p efter faldende x-potenser således:


q:=collect(p,x);


Koefficienten til  kan fås på en af følgende to måder:


coeff(q,x^2); 


coeff(q,x,2);


Den sidste metode kan benyttes til at bestemme koefficienten til :


coeff(q,x,0);


Der findes også en procedure coeffs, der kun giver koefficienterne: Se hjælpen.


I pakken PolynomialTools findes CoefficientVector og CoefficientList, der ved input af et 
polynomium i én variabel returnerer koefficienterne i form af en vektor eller en liste, henholdsvis:


PolynomialTools:-CoefficientVector(p,x);


Skal koefficienterne til potenserne af x behandles på en eller anden måde, tilføjes den dertil 
svarende procedure, eksempelvis er det ofte en god idé at tilføje factor:


collect(p,x,factor); 


Skal leddene samles med hensyn til y og koefficienterne faktoriseres, skriver vi:


collect(p,y,factor);


Betragtes  p som et polynomium i de 2 variable x og y kan leddene samles således:


collect(p, [x,y],distributed);

Sammenlign resultatet med følgende to uden tilføjelsen distributed: 


collect(p, [x,y]);


collect(p, [y,x]);


collect kan også samle led med hensyn til funktionsnavne (f.eks. ln eller exp) og mht. uevaluerede 
funktionsudtryk som f.eks. ln(x) og exp(x):


w:=(1/x+ln(x))^3 +5*ln(x)+a*ln(x)/(x+1)+7*ln(x)^2+a*x;


Begge af følgende former kan bruges:


collect(w,ln);


collect(w, ln(x));


Koefficienten til  kan så fås således:


coeff(%,ln(x),2); 


Her samles igen mht. ln(x), men nu bruges simplify på hver af koefficienterne:
  


collect(w, ln(x), simplify); 


eval og subs


Det kan være nyttigt i et udtryk, der indeholder en operand x, at erstatte x med enten et tal eller et 
andet udtryk. Dette kan ske vha. proceduren eval:


u:= ln(2)-ln(7)+exp(a)*exp(b-a)+sin(x)^3+2*cos(x)^2-cos(2*x)


;

Vi erstatter x med :


eval(u,x=Pi); 


Udtrykket u ændres ikke derved


u; 


Proceduren subs kan også bruges, men bemærk, at der ikke bliver evalueret, kun substituteret. 
Bemærk også, at argumentrækkefølgen er omvendt:


subs(x=Pi,u);


 
Evaluering sker ved den blotte reference til ovenstående:
 


%;


Skal flere substitutioner udføres samtidigt, skal disse indkapsles i mængdeklammer, både for eval 
og for subs:


eval(u,{sin(x)=x^7,x=Pi});


subs({sin(x)=x^7,x=Pi},u);


%;


I begge procedurer kan bruges listeklammer i stedet for mængdeklammer.
Proceduren subs kan udføre en følge af substitutioner successivt eller en følge af mængder af 
substitutioner successivt. Nedenfor udføres først den første substitution. I det fremkomne udtryk 
udføres dernæst den anden substitution:


subs(sin(x)=x^7,x=Pi,u);


%;


Med omvendt rækkefølge fås (her med to kommandoer på samme linie):


subs(x=Pi,sin(x)=x^7,u):   %;

Som det ses, fik substitutionen sin(x) = x^7 slet ingen virkning. Grunden er den, at efter 


substitutionen  , er der ingen  tilbage i udtrykket.


eval kan også bruges til at fremtvinge fuld eller delvis evaluering af et udtryk:


s3:=s2: s2:=s1: s1:=s0:


Fuld evaluering er det normale i Maple. Så nævnes s3:


s3;
s0


fås den fuldt evalueret. Ønsker man kun evaluering til første trin, kan man skrive:


eval(s3,1);
s2


Funktioner, procedurer og tabeller evalueres kun til deres navne:


f:=x->x^2;


f;
f


Fuld evaluering fås ved kommandoen


eval(f);


Se også om algsubs, subsop, applyop, applyrule, subsindets, evalindets i Kapitel 14.


Opløsning i stambrøker


Opløsning af en rational funktion i stambrøker (engelsk: partial fraction decomposition) foregår 
ved en af de talrige varianter af proceduren convert:


ratfunk:=(x^2+h*x-11)/(x^4-4*x^3+7*x^2-12*x+12);


convert( ratfunk, parfrac, x);


Da factor ikke automatisk faktoriserer , vil der ikke ske noget i følgende kommando:


convert(1/(x^2-3)^2,parfrac,x);
1

Men der findes en anden version:


convert(1/(x^2-3)^2,fullparfrac,x);


Summen er en sum over samtlige rødder i polynomiet  , altså  og  . Et eksplicit 
udtryk fås ved endnu en convert-kommando:


convert(%,radical);


Forsøges ovenstående to kommandoer på vores første udtryk ratfunk fås en stambrøksopløsning 


indenfor de komplekse tal, da  også vil blive faktoriseret.
Hvis der optræder decimalbrøker i koefficienterne, så angives resultatet på decimalbrøksform:


convert(1/(x^2-3.0),parfrac,x);


Automatisk simplifikation m.m.


Ved automatisk simplifikation forstås en simplifikation, der foretages inden evaluering. Om en 
simplifikation af et udtryk er automatisk eller ej, afsløres ved at sætte apostroffer om udtrykket 
således:


'2+2'=2+2,'1/sqrt(2)'=1/sqrt(2);


En evaluering af et udtryk omgærdet af apostroffer består blot af fjernelsen af disse. Apostroffer 
kan altså bruges til at forhale evaluering. Trods forhaling af evaluering blev ovenfor '2+2' til 4. 
Der er altså tale om en automatisk simplifikation. En sådan kan ikke forhindres. Derimod 


simplificeres  ikke automatisk til .


Ved definitionen af en procedure eller en funktion evalueres ikke, men der foretages automatiske 
simplifikationer:


f:=x->2*3*x;


Hvad enten man kan lide det eller ej, vil Maple foretage følgende automatiske simplifikationer, 
når x ikke er tilordnet nogen talværdi:

'x-x',  '2*x+5*x',  'x+0',  'x/x',  'x*1',  'x^0',  'x^1';  


Naturligvis fås samme resultater uden apostroffer, men med apostrofferne afsløres det, at disse 
simplifikationer ikke kan undgås. 
Den første simplifikation er ikke god, hvis . Den fjerde og den sjette simplifikation er ikke 
god, når .
Men den automatiske simplifikation reagerer korrekt, når den får konkrete tilfælde: 


'infinity-infinity';
undefined


'0/0';


Error, numeric exception: division by zero


undefined i forbindelse med operatorerne /, +, * håndteres automatisk i Maple 13:


'undefined/undefined','undefined-undefined';


'undefined+5', 'undefined*5', 'undefined*0';


Men her er simplifikationen ikke automatisk:


'undefined+g(x)'=undefined+g(x),'sin(undefined)'=sin


(undefined);


Da Maple intet ved om g(x), gøres intet. Maple kender imidlertid sinusfunktionen, så der sker 
noget.


Maple sætter automatisk  til 1, som det ses her:


'0^0';
1


Til nogles irritation vil Maple altid gange et konkret reelt tal ind i en parentes. Sammenlign:


77/3*(a+b); 3.1*(a+b); c*(a+b); Pi*(a+b); sqrt(2)*(a+b); 


(2+3*I)*(a+b);


Man kan prøve at sætte forhalende apostroffer om de to første udtryk, men det hjælper ikke: 
'Simplifikationen' er automatisk.

Maple er meget tilbageholdende med at bruge , der jo heller ikke gælder altid. Vi 


betragtede eksemplet  ovenfor.


simplify(  (a^b)^c );
c


is( (a^b)^c=a^(b*c) );
false


Ifølge hjælpen skal is returnere false, hvis Maple kan godtgøre, at der findes tilfælde, hvor 
udsagnet er falsk: Tilfældet a = -1, b = 2 og c = 1/2 viser, at simplifikationen ikke altid er gyldig:


eval((a^b)^c=a^(b*c),{a=-1,b=2,c=1/2});


Vi prøver med en lokal antagelse om at c er et helt tal. Bemærk det dobbelte kolon, der efterfølges
af en Maple-type, her integer:


simplify( (a^b)^c ) assuming c::integer;


a


Vi prøver en global antagelse:
 


assume(c,integer);  simplify( (a^b)^c );


a


Her kunne også være skrevet assume(c::integer);

11. Tal i Maple


Maple regner eksakt (dvs. bruger ikke decimalbrøkstilnærmelser), med mindre den får anden besked:


1+1/2+1/3+1/4;
25


12


sqrt(2);


Det største hele tal Maple kan håndtere er (på en 32-bit computer) omkring , altså et tal 


med ca. 268 millioner cifre. Om den brugte computer så har hukommelse nok, er en anden sag. Et 
stort antal cifre er altså i praksis intet problem: 


p:=2013*(3^30+7^25)/(26!-5^38);


Ønskes p tilnærmet ved et decimaltal benyttes proceduren evalf:


evalf(p);
0.06835469252


Under beregningen af udtrykket p arbejdes med 10 betydende cifre. Ønskes større nøjagtighed, f.eks. 
50 betydende cifre, er der følgende variant, hvor man specielt skal bemærke de kantede parenteser: 


evalf[50](p);
0.068354692517336090230478097611916293711133612550216


Ønskes beregninger generelt udført med p cifre, kan Digits sættes til p. Ved start har Digits værdien 
10:


Digits;
10


Næste linie indeholder fire kommandoer. To af dem slutter med kolon. De bliver udført, men ikke 
udskrevet:


Digits:=40: evalf(p); Digits:=10: evalf(p);
0.06835469251733609023047809761191629371113


0.06835469252


Miljøvariable

Digits er en såkaldt miljøvariabel. Dette betyder, at hvis Digits sættes til en værdi inde i en 
procedure, så har denne tilordning ingen effekt udenfor.
Som eksempel definerer vi 2 procedurer, der blot skal udregne input med 5 og 30 betydende cifre,
henholdsvis:


p1:=proc(x) Digits:=5; evalf(x) end proc;


p2:=proc(x) Digits:=30; evalf(x) end proc;


Vi finder  med 5 og 30 betydende cifre:


p1(Pi),p2(Pi);


Digits, der inde i procedurerne blev sat til 5 og 30, er uden for proceduren uændret 10:


Digits;
10


Variablen Order, der bestemmer antal led i rækkeudviklinger, er også en miljøvariabel. Det 
samme gælder variable med navne startende med _Env, eksempelvis _EnvExplicit og
_EnvAllSolutions.


anames(environment);


UseHardwareFloats,Rounding,printlevel;


For flere detaljer se hjælpen:


?_Env


Decimaltal kan også skrives på følgende måde, hvor det er ligegyldigt, om der bruges stort eller lille
e:


7.3e-6 = 7.3E-6;  7.3e-7; 7.3e4;  7.3e5;  


73000.


Decimaltal kan også tastes ind ved brug af Float. Maple repræsenterer decimaltal som to hele tal: 
mantissen og eksponenten:


Float(52,23), Float(52,-23), Float(-52, 3), Float(52, -4);


Operanderne i et decimaltal er


op(123.456789);

Float(%);
123.456789


op(-123.456789);


Antal cifre i et decimaltal kan fås således:


length(op(1,123.456789));
9


Sammenlign Maple's håndtering af decimaltallet 123456789.0 og det hele tal 123456789:


123456789.0^2;


Da Digits har værdien 10, regner Maple med 10 betydende cifre, mens den med hele tal regner 
eksakt:


123456789^2; 
15241578750190521


Vær opmærksom på, at kommandoen evalf[n]( udtryk)  ikke betyder, at der garanteres n betydende 
cifre i resultatet af beregningen af udtryk, men kun, at der benyttes n betydende cifre ved 
beregningen. Således giver følgende to kommandoer samme resultat:


evalf(sqrt(2)-14142135*10^(-7));


evalf(sqrt(2))-evalf(14142135*10^(-7));


Resultatet indeholder kun 2 betydende cifre. Resultatet er da også heldigvis skrevet på en måde, der 
antyder 2 betydende cifre.
Vil man have flere betydende cifre, må man ved beregningen bruge flere cifre end her 10:


evalf[15](sqrt(2)-14142135*10^(-7));


Maple kender en række konstanter:


constants;


Tre af disse 'konstanter' kan ikke med vores tastatur skrives, som Maple gør i output, men de fås 
således:


gamma, infinity, Pi;  

 er Eulers konstant, der er defineret som grænseværdien  .  Catalan er 


Catalans konstant. Den er givet ved 


evalf[20](gamma);
0.57721566490153286061


evalf[30](Catalan);
0.915965594177219015054603514932


evalf[150](Pi);
3.1415926535897932384626433832795028841971693993751058209749445923078164062862089\


9862803482534211706798214808651328230664709384460955058223172535940813


Bemærk i det sidste svar linieskiftsymbolet \ . 


Om beregningen af 
Maple opbevarer   med 1000 cifre i den globale variable _bigPi. og med 10 000 cifre i


`evalf/constant/bigPi`. Hvis man beder om med mere end end 1000 cifre, men mindre end 10 


000, så opdateres _bigPi til `evalf/constant/bigPi`. Hvis man beder om med mere end end 10 


000 cifre, så må Maple regne og _bigPi opdateres til den beregnede værdi.


restart;


_bigPi;
3.14159265358979323846264338327950288419716939937510582097494459230781640628620\


899862803482534211706798214808651328230664709384460955058223172535940812848\


111745028410270193852110555964462294895493038196442881097566593344612847564\


823378678316527120190914564856692346034861045432664821339360726024914127372\


458700660631558817488152092096282925409171536436789259036001133053054882046\


652138414695194151160943305727036575959195309218611738193261179310511854807\


446237996274956735188575272489122793818301194912983367336244065664308602139\


494639522473719070217986094370277053921717629317675238467481846766940513200\


056812714526356082778577134275778960917363717872146844090122495343014654958\


537105079227968925892354201995611212902196086403441815981362977477130996051\


870721134999999837297804995105973173281609631859502445945534690830264252230\


825334468503526193118817101000313783875288658753320838142061717766914730359\


825349042875546873115956286388235378759375195778185778053217122680661300192\


7876611195909216420199


Antal cifre i _bigPi efter restart.


length(op(1,_bigPi));
1000


Tidsstudier:

time(evalf[1000](Pi));
0.


time(evalf[1001](Pi));
0.


Antal cifre i _bigPi efter ovenstående kommando:


length(op(1,_bigPi));
10000


time(evalf[10007](Pi));
0.015


length(op(1,_bigPi));
10007


Beregningen af  sker ved følgende korte procedure, der dog som det ses kalder proceduren


`evalf/hypergeom/kernel`:


showstat(`evalf/constant/Pi`);


`evalf/constant/Pi` := proc()


local a, r, t, bp;


global _bigPi;


   1   if Digits <= 51 then


   2     evalf


(3.14159265358979323846264338327950288419716939937511)


       elif Digits <= length(op(1,_bigPi)) then


   3     evalf(_bigPi)


       elif Digits <= 10000 then


   4     _bigPi := `evalf/constant/bigPi`;


   5     evalf(_bigPi)


       else


   6     Digits := Digits+2;


   7     a := evalf(1823176476672000^(1/2));


   8     t := 13591409/545140134;


   9     r := 13591409*`evalf/hypergeom/kernel`([1/6, 1/2, 5/6, 


t+1],[1, 1, t],-1/151931373056000);


  10     bp := a/r;


  11     _bigPi := evalf(bp,Digits-2)


       end if


end proc


Proceduren ̀ evalf/hypergeom/kernel` er indbygget (builtin): 


print(`evalf/hypergeom/kernel`);

PI:=1823176476672000^(1/2)/13591409/hypergeom([1/6, 1/2, 


5/6, 13591409/545140134+1],[1, 1, 13591409/545140134],


-1/151931373056000);


evalf[20000](PI)-evalf[20000](Pi);
0.


Beregningen ovenfor tager lidt tid, men den viser, at påstanden er korrekt i det mindste med 20 
000 betydende cifre.
Maple synes ikke selv umiddelbart at kunne indse, at PI er eksakt lig med Pi, idet den opgiver at 
udregne PI:


value(PI);


Pas på: Den matematiske konstant   SKAL staves  Pi, altså hverken  pi,  PI eller  pI:


Pi, pi;


Nok er Pi og pi fuldstændig ens af udseende i output, men de er helt forskellige. pi er blot det græske 


bogstav 


sin(Pi),sin(pi);


Grundtallet for den naturlige logaritme, e, fås i Maple ved brug af eksponentialfunktionen exp:


exp(1);
e


evalf[50](exp(1));
2.7182818284590452353602874713526624977572470937000


Bemærk, at Maple betegner exp(1) med e i output. Bogstavet e i input behandles som en ubekendt 


variabel. Udtrykket  skrives tilsvarende exp(2).


Maple kender de sædvanlige elementære funktioner, men også mange andre. Blandt  de gængse kan 
nævnes  ln, exp, sin, cos, arcsin, arccos, arctan, arcsinh (NB: arcsinh, ikke arsinh), abs ( = 
absolutværdi = numerisk værdi) og sqrt ( square root = kvadratrod). For en liste over funktioner, som
Maple umiddelbart kender, tryk ENTER på følgende linie:

?inifcns


(inifcns = initially known functions). Se iøvrigt Kapitel 13, Funktioner.


Tredjeroden ville man mene, at man kunne få ved opløftning til 1/3. Men pas på! For positive tal er 
det helt fint, men for negative tal fås ikke den negative reelle rod, men den komplekse rod i første 
kvadrant. En tilsvarende bemærkning gælder andre rødder med ulige eksponent. I stedet kan man 
bruge funktionen surd, hvis brug vi viser nedenfor.


(-8)^(1/3);


simplify(%);


Bogstavet I er Maples symbol for den imaginære enhed, som vi normalt betegner med i eller j.


surd( -8,3);


For lige rødder af positive tal er der ingen forskel (se også under komplekse tal nedenfor):


simplify( 16^(1/4) ); surd( 16, 4);  
2


2


Primtalsopløsning


Procedurerne numer (numerator = tæller) og denom (denominator = nævner) kan bruges på alle 
slags brøker. 


Vi laver en brøk til at øve os på. Vi tager  med 30 betydende cifre og omdanner dette decimaltal 


til et rationalt tal:


convert(evalf[30](Pi),rational,exact);
39269908169872415480783042291


12500000000000000000000000000


taeller:=numer(%);


naevner:=denom(%%);


Vi opløser i primtalsfaktorer. Her skal bruges ifactor ( integer = helt tal), ikke factor, som kun kan
bruges på polynomier.


ifactor(taeller); 

ifactor(naevner);


Tælleren var altså et produkt af kun 4 primtalsfaktorer. Maple har fire procedurer, der har med 
primtal at gøre:


isprime(22463351);   # Er tallet et primtal?
true


ithprime(1000000);   # Primtal nummer 1 000 000
15485863


nextprime(200);      # Det første primtal efter 200
211


prevprime(200);      # Det sidste primtal før 200
199


Den meget nyttige procedure seq giver her sammen med ithprime en følge (= sequence) af 
primtallene med numrene 1 til 25:


seq( ithprime(k), k=1..25); 


I pakken numtheory (number theory) findes bl.a. procedurerne mersenne og pi.


pi(k) finder antallet af primtal mindre end eller lig det hele tal k:


numtheory:-pi(200);
46


plots:-logplot([seq([k,numtheory:-pi(k)],k=2..20000)],


gridlines,caption=typeset("Logaritmisk plot af ",pi(k)),


labels=[k,typeset(pi(k))]);


k


5000 10000 15000 20000


500


1000


1500


2000


Logaritmisk plot af 


Et Mersenne-primtal p er et primtal, der kan skrives som .

mersenne(n) undersøger om  er et primtal og returnerer da dette primtal, ellers returneres
false eller FAIL:


seq(numtheory:-mersenne(k),k=1..20);


mersenne([n]) finder det n'te Mersenne-primtal:


seq(numtheory:-mersenne([k]),k=1..10);


Vi lægger 1 til hvert af Mersenne-primtallene:


[%]+~1;


og kontrollerer, om resultatet kan skrives på formen :


ifactor~(%);


Ovenfor blev brugt elementvise operationer ̀ +~` og `ifactor~` på den givne liste. Disse er til 
rådighed i Maple 13.


Proceduren igcd finder største fælles mål (greatest common divisor) mellem to eller flere hele tal, 
altså det største positive hele tal som går op i dem alle (gcd er beregnet til polynomier):
 


igcd(35,55); igcd(123,456,789);
5


3


Mindste fælles mangefold (= least common multiple), dvs. det mindste positive hele tal, som alle 
tallene går op i:


ilcm(6,9,7);
126


Vil man finde 25 (mod 3) , altså resten efter division af 25 med 3, kan man gøre således:


25 mod 3;
1


eller således:


irem(25,3);
1


Se iøvrigt hjælpen for mod, irem og iquo.

Komplekse tal


Maple regner med komplekse tal. Faktisk er den sommetider gladere for komplekse tal end vi 
andre. Den imaginære enhed betegnes med symbolet I (NB: Stort I). 


(3+5*I)^4/(4-7*I);


Den foretagne udregning var faktisk en automatisk simplifikation. Dette afsløres ved at forsøge 
forhalende apostroffer:


'(3+5*I)^4/(4-7*I)';


Apostrofferne hjalp ikke!


I næste udtryk kan evalueringen af  forhales ved hjælp af apostroffer:


A:=(2+2*I)*'exp(I*Pi/6)';


A;


For at komme videre kan enten bruges expand eller evalc. Proceduren evalc forsøger at skrive det 
komplekse udtryk på rektangulær form, altså på formen x+iy, hvor x og y er reelle. Bemærk 
forskellen på de to resultaters form:


evalc(A); expand(A);


Real- og imaginærdel fås ved brug af Re og Im:


Re( a+I*b), Im( a+I*b); 


Der skete ikke meget. Maple antager ikke generelt, at variable er reelle. Det gør dog proceduren
evalc:


evalc(Re( a+I*b)), evalc(Im( a+I*b)); 

Det samme resultat opnås ved eksplicit at antage, at a og b er reelle:


Re( a+I*b) assuming a::real,b::real;
a


Med  mener Maple en bestemt af de n rødder (lighedstegnet er kun til udstillingsbrug):


(-1)^(1/4) = evalc( (-1)^(1/4) );


Med  menes i Maple den rod i den binome ligning  , der har det numerisk mindste 


hovedargument. Er der 2 sådanne, vælges den rod, der ligger i øvre halvplan. Anderledes sagt: For


 vælges den principale n'te rod af  a, dvs. den rod, hvis hovedargument er    af 


hovedargumentet for a.  Dette gælder også for kvadratroden.


sqrt(3+4*I), sqrt(3+5*I); # Bemærk forskellen i output


evalc( sqrt(3+5*I));


Lighedstegnene i det følgende er kun til udstillingsbrug:


(-I)^(1/3)=evalc( (-I)^(1/3)); 


(1-I)^(1/2)=evalc( (1-I)^(1/2)); 


Som alternativ til (-1)^(1/4) kan bruges


root(-1,4); 


evalc(root(1-I,2));


Maple foretrækker i næste eksempel at bruge cosinus og sinus:


(-1)^(1/5)=evalc((-1)^(1/5));


men vi kan omdanne dette til et rodudtryk ved en convert-kommando:

convert(rhs(%),radical);


Kommandoen surd(z,n) giver dén n'te rod, hvis hovedargument er tættest på hovedargumentet for
z . Sammenlign følgende to kommandoer:
 


simplify( (-16)^(1/4) ); evalc(surd( -16, 4));


Dette betyder, at eksempelvis functionen  er diskontinuert på intervallet ] 


 ], hvilket illustreres i animationen nedenfor, hvor placeringen af de 4 rødder i den binome 


ligning  vises sammen med e  og 


with(plots):


n:=4: eps:=.00001: itvl:=-Pi+eps..Pi-eps: 


p1:=animate(complexplot,


[[surd(exp(I*t),n)],style=point,symbolsize=25,symbol=


circle],


t=itvl):


p2:=animate(complexplot,


[[seq(exp(I*((t+p*2*Pi)/n)),p=0..n-1)],style=point,symbol=


cross,symbolsize=20,color="Navy"],


t=itvl):


p3:=animate(complexplot,


[[exp(I*t)],style=point,symbol=soliddiamond,symbolsize=20,


color="Black"],


t=itvl):


p4:=animate(textplot,


[[[cos(t),sin(t)+.1,typeset(''exp''(I*t))]]],t=itvl):


p5:=animate(textplot,


[[[Re(surd(exp(I*t),n)),Im(surd(exp(I*t),n))+.1,surd]],


color=red],t=itvl):


display(p1,p2,p3,p4,p5);

e


surd


0 1


1


Prøv at ændre surd til root. Prøv også andre værdier for n end 4.


Modulus og argument fås ved brug af henholdsvis abs og argument, men kan også fås ved polar 
eller en convert-kommando:


abs(-3+4*I), argument(-3+4*I); # Modulus og argument


polar(-3+4*I);


convert(-3+4*I, polar);


evalc(polar(2,Pi/3));


polar kan bruges på to måder:


polar(z);


polar(r,v);


evalc(polar(r,v));


Vi ser, at polar(r,v) returnerer uevalueret, men med evalc omskrives til rektangulær form.

Brugeren kan definere en printprocedure, der fortæller, hvordan resultater fra polar skal printes til
skærmen: 


`print/polar`:=proc(r,v) r*e^(I*v) end proc;


polar(2,Pi/2);


evalc(%);


polar(1+I);


Bemærk, at dette kun har med printet til skærmen at gøre, hvilket ses af følgende kommandoer:


polar(2,Pi/2);


op(0,%),op(%);


 Output er altså det samme som før, hvilket understreges ved følgende absurde printprocedure:


`print/polar`:=proc(r,v) bøh end proc;


polar(2,Pi/2);
bøh


evalc(%);


Den imaginære enhed er implementeret som Complex(1), men printes på skærmen som I: 


Complex(1);
I


Complex(4,7);


Complex(4); 


Bemærk outputtet fra følgende input, der egentlig blot er 4 komplekse versioner af tallet 2: 


2+0.*I,2-0.*I,2.+0*I,2+0*I;


Vi ser, at de første 2 tal ikke automatisk simplificeres til decimaltallet 2., mens de to sidste 
simplificeres til henholdsvis decimaltallet 2. og det hele tal 2. 
Output af form som de to første møder man undertiden fra forskellige af Maples procedurer. Man 
kan da ønske at fjerne den imaginære del, som jo er nul. Hertil kan bruges simplify med det ekstra

argument zero. Som eksempel lader vi variablen u være udtrykket


u:=2+.0*I+(3*x+5-0.*I)*(cos(x)^2-sin(x)^2);


simplify(u,zero);


Brug af simplify uden tilføjelsen zero virker også i dette eksempel, men simplify vil også forsøge 
andre simplifikationer, der måske ikke var ønskede:


simplify(u);


 For flere detaljer se:


?Complex


?simplify,zero


Pakken RealDomain


Vil man udelukkende regne reelt, kan man forsøge sig med pakken RealDomain.
 


with(RealDomain);


Resultaterne af følgende kommandoer er kun korrekte, hvis de variable er reelle:


simplify(sqrt(x^2));


Re( a+I*b), Im( a+I*b); 


Nu accepteres   ikke:
 


sqrt(-1);
undefined


Ved løsning af den binome ligning   kommer nu kun de 2 reelle løsninger:


solve(x^6=1,x);


Den sædvanlige procedure solve kan stadig fås frem, men nu må sættes tegnet :- foran:


:-solve(x^6=1,x);

Vi ser her de to reelle løsninger sammen med de 4 imaginære.

12. Strenge, Variabelnavne, Evaluering


Definition af streng og variabelnavn


En streng (kort for tekststreng) i Maple fås ved brug af "...". Eksempel:


str:="Der var engang en mand, der boede i en spand.";


whattype(str);
string


Strenge kan ikke bruges som navne på variable:


"Min skattepligtige indkomst":=123456;


Error, invalid left hand side of assignment


Strenge bruges f.eks. i plot-titler og i fejlmeddelelser i procedurer.
Der findes i pakken StringTools en lang række procedurer, der kan bruges ved manipulation af 
strenge:


?StringTools


Et eksempel:


StringTools:-UpperCase(str);
"DER VAR ENGANG EN MAND, DER BOEDE I EN SPAND."


Da StringTools som ny pakke er et modul, kunne ovenstående lange form for UpperCase bruges.
Versionen med [ ] kan dog også bruges:


StringTools[Capitalize](str);
"Der Var Engang En Mand, Der Boede I En Spand."


En tilfældig streng valgt fra alfabetet ident = {a-z, A-Z, 0-9, _ }:


StringTools[Random](8,ident);
"8dvpw_7b"


Et tilfældigt 6-cifret tal fra 16-talssystemet


StringTools:-Random(6,"0123456789ABCDEF");
"DD3A74"


konverteres her til decimaltal:


convert(%,decimal,hex);
14498420


En tilfældig streng med 20 karakterer


StringTools:-Random(20);
"m'%hD	 � ®� � ç‹¢Î� ÏâHI"


konverteres til en liste med karakterernes byte-værdier:


convert(%,bytes);

og tilbage igen:


convert(%,bytes);
"m'%hD	 � ®� � ç‹¢Î� ÏâHI"


Tilde på tastaturet har hexværdien 7E:


convert("7E",decimal,hex); convert([%],bytes);
126


"~"


I Maple findes to slags variabelnavne:


(1)  Et simpelt navn kan bestå af et eller flere af følgende tegn: Bogstaver,  talcifre (0, 1, ... , 9), 
eller understregninger. Navnet skal begynde med et bogstav. 
 Eksempel: Meget_langt_udtryk_nr_32.  
Mellemrum må ikke forekomme.


Maple selv bruger i udskriften undertiden navne, der begynder med understregning, som f.eks. 
_C1 (se Kapitel 6: Differentialligninger for et eksempel). Sådanne variabelnavne bør man derfor 
selv undgå. Undgås skal også de ord, Maple reserverer til eget brug, som f.eks. factor eller 
restart. 


Der skelnes mellem store og små bogstaver, så navnet  udtryk1  er forskellig fra  Udtryk1. 


(2)  Et variabelnavn kan også fås ved at omgærde en række karakterer (tegn) med accent grave'er. 
Et mellemrum regnes også for en karakter.     


`Min skattepligtige indkomst`:=123456;


`Min skattepligtige indkomst`;
123456


I Maple findes mange procedurer med navne af formen `navn1/navn2`. Her er accenterne 
nødvendige for at forhindre, at skråstregen opfattes som et divisionstegn.
Et eksempel er


eval(`evalf/sin`);


Som det ses er ̀ evalf/sin  ̀navnet på en procedure. Default er, at indholdet i Mapleprocedurer ikke 
printes. Man kan få indholdet at se ved at udstede kommandoen interface(verboseproc=2); eller 
ved at bruge showstat i stedet: showstat(`evalf/sin`); Det har vi af pladshensyn ikke ønsket at gøre
her.


Navne kan inkludere et indeks:


x[2],  U[235],  abc[Frederik,Peter],  A[3][2];


Bemærk resultaterne af følgende typekontroller:

type(str,string), type( U[235], string),   type( U[235], 


name),  type(U[235], indexed);


Et symbol er et navn uden index:


type(qwerty,symbol), type( U[235], symbol), type(`Min 


øllebrød`,symbol) ;


Advarsel mod brug af navn og samme navn indiceret
Det vil i mange sammenhænge forekomme naturligt at have en variabel med navnet x og en 
eller flere variable med navne som .


Dette kan dog i visse tilfælde give alvorlige problemer, så det bør undgås.


Grunden ses af de følgende linier:


x[0]:=7;


Variablen x er nu implicit blevet defineret til at være en tabel (om tabeller se kapitel 14):


eval(x);


Da normal evaluering af et tabelnavn blot resulterer i navnet selv:


x;
x


er det ofte intet problem, at x er en tabel. Eksempel:


int(x^2,x=0..1);
1


3


Men bemærk forskellen mellem multiplikation af vektoren v med s og x, henholdsvis:


v:=<a,b,c>:  s*v, x*v;


Bemærk den umiddelbart ubegribelige fejlmelding fra dsolve: 


dsolve(diff(x(t),t)=x(t));


Error, (in ODEtools/odsolve) unable to handle ODEs of undefined


differential order


Vi prøver debug:


debug(dsolve);
dsolve


dsolve(diff(x(t),t)=x(t));


{--> enter dsolve, args = diff((table( [( 0 ) = 7 ] ))(t), t) =

(table( [( 0 ) = 7 ] ))(t)


false


<-- ERROR in dsolve (now at top level) = unable to handle ODEs 


of undefined differential order}


Error, (in ODEtools/odsolve) unable to handle ODEs of undefined


differential order


Vi afslutter fejlfindingen:


undebug(dsolve);
dsolve


Havde vi skrevet differentialligningen ud før brugen af dsolve, ville vi have opdaget 
problemet:


diff(x(t),t)=x(t);


Skulle vi i øvrigt få den idé, at sætte x til et eller andet:


x:=56:


så er x dermed omdefineret og værdien af   er glemt, og resultatet er i stedet det 


mærkværdige:


x[0];
56


Da det ikke på forhånd er klart, hvor problemer kan opstå, bør man afholde sig fra brug af et 
navn og samme navn indiceret. 


restart;


For en detaljeret hjælpeside om navne prøv:


?names


Variabel, Evaluering


En Maple-variabel er blot et navn. Den peger enten på sig selv (dvs. på sit eget navn) eller på et 
andet Maple-objekt. Vi kan også sige, at variablen enten har sit eget navn eller noget andet som 
værdi. Ved tilordningen


x:= 4711;


peger x nu på 4711. Vi siger, at x har værdien 4711. 

I dette worksheet er y ikke tidligere givet nogen værdi. Variablen y har altså sit eget navn som 
værdi. Definér nu p som summen af x og y:


p:= x+y;


Maple evaluerer først højresiden fuldstændigt. Derpå får p den evaluerede værdi, altså 4711+y. 
Dvs. at p peger direkte på  4711+y, ikke på  x+y. Tildeler vi nu x en ny værdi


x:= 2;


vil p ikke evaluere til en anden værdi, som det ses:


p;


Ønsker vi igen at benytte x som en ubekendt variabel, skal vi give x sit eget navn som værdi, 
hvilket sker ved brug af forhalende apostroffer (altså ikke accenter!):


x:= 'x'; 


Omdefinerer vi p ved tilordningen


p:= x+y;


vil p nu pege på summen af de to navne x og y. Skriver vi først:


x:= 2;


bliver p evalueret til:


p;


Tillægger vi x en ny værdi


x:= 4711;


bliver p evalueret til 


p;

Vær opmærksom på, at Maple skelner imellem apostroffer og accenter.  Accent aigu (som vi 
kender fra ordet  idé ) har ingen speciel betydning, men kan benyttes som et tegn. Accent grave 
kan som nævnt benyttes ved definition af et navn. Ved tilordningen


A:=`Min købmandsregning`;


peger A på `Min købmandsregning`. Vi fortsætter med


`Min købmandsregning`:=47;


A;
47


Det tomme navn fås med to accenter:  `` . Det bør undgås som variabelnavn af indlysende grunde.
Det har dog sin anvendelse ved navnesammenslutninger, se nedenfor. En anden anvendelse er vist
i slutningen af kapitel 13.


Når Maple første gang læser en kommando, bliver udtryk som nævnt normalt evalueret helt, men 
for udtryk omgivet af apostroffer sker kun, at disse fjernes. Evalueres én gang til, bliver variablen 
evalueret:


a:=cos(Pi/4)*'sin(Pi/2)';


Nu evalueres også :


a; 


Vi renser a for indhold ved brug af forhalende apostroffer:


a:='a';


a peger herefter på sit eget navn. Således renses variable for tilordninger, hvis man ikke vil bruge 
restart-kommandoen eller unassign (se hjælpen). 


Undtagelserne til fuldstændig evaluering er: Navne på procedurer (og specielt funktioner), 
moduler, tabeller, arrays og matricer (knyttet til den gamle linalg-pakke). Navne på Matricer (stort
M) knyttet til LinearAlgebra-pakken evalueres fuldstændigt. 


Sammensætning af navne og strenge


Strenge og navne kan sammensættes med hinanden eller indbyrdes vha. proceduren cat eller ved 
brug af to lodrette streger || . Men Maple fraråder selv brugen af || , se hjælpen:  

?||


Her sættes abc sammen med defg og hijk:
 


cat(abc,defg,hijk);
abcdefghijk


abc||defg||hijk;
abcdefghijk


Når der bruges || ved sammensætningen, evalueres første element ikke. Hvis også evaluering af 
første element ønskes, anbringes den tomme streng (eller det tomme navn) forrest. Så er det 
nemlig den, der ikke bliver evalueret, men det gør jo ikke noget. 
Proceduren cat evaluerer alle argumenter.


A:=Et_simpelt_navn;


B:=` og et accentnavn`;


Bemærk nedenfor, at A nedenfor ikke evalueres:


A||B; 
A og et accentnavn


Vi anbringer det tomme navn forrest:


``||A||B; 
Et_simpelt_navn og et accentnavn


Længden af B (= antal karakterer i navnet):


n:=length(B); 


Et længere eksempel:


``||A||B||` med `||n||` karakterer`;
Et_simpelt_navn og et accentnavn med 17 karakterer


Sammenlign med brugen af cat:
 


cat(A,B,` med `,n,` karakterer`);
Et_simpelt_navn og et accentnavn med 17 karakterer


Der kan tilordnes til A||B og til cat(A,B):


A||B:=54;


cat(A,B):=54;


Vi danner igen navnet:

cat(A,B);
Et_simpelt_navn og et accentnavn


og dette evaluerer til


%;
54


cat og || kan også bruges med strenge:


str:="Der var engang en mand, der boede i en spand.";


str2:="Spanden var af ler,";


str3:=" konen vasked' bleer. ";


I vores fjerde streng inkluderer vi et linieskiftsymbol "\n":


str4:="\nNu ka' jeg ikke mer'.";


str; cat(str2,str3,str4);
"Der var engang en mand, der boede i en spand."


"Spanden var af ler, konen vasked' bleer. 


Nu ka' jeg ikke mer'."


Strenge og navne kan sammensluttes. Første element bestemmer, om resultatet bliver en streng 
eller et navn.
Først en streng:


""||A||str; 
"Et_simpelt_navnDer var engang en mand, der boede i en spand."


Så et navn:


``||A||str;
Et_simpelt_navnDer var engang en mand, der boede i en spand.


Og ved brug af cat:
Første en streng:


cat("",A,"   ",str);
"Et_simpelt_navn   Der var engang en mand, der boede i en spand."


Så et navn:


cat(A,str);
Et_simpelt_navnDer var engang en mand, der boede i en spand.


evaln, convert(..., string), parse


I forbindelse med præsentation af resultater, kan proceduren evaln (evaluate to the last name) 
undertiden være nyttig. Den kan bruges på udtryk, der (hvis de ikke evalueres til bunds) kan 

optræde på venstre side af et tilordningstegn (:=). Proceduren evaluerer evt. underudtryk, men 
standser lige før det evaluerede ikke længere er et udtryk, der kan tilordnes en værdi:


k:=77: t[k]:=48*Pi;


evaln bevirker at k evalueres, men t[77] gør ikke:


evaln(t[k]); 
t


t[77] evalueres i næste omgang:


%; 


Evaluering kan selvfølgelig forhales med apostroffer, men så evalueres k heller ikke, hvis 
apostrofferne omgærder hele udtrykket:


't[k]'; 
t


Hvis apostrofferne kun omgærder t, så opnås heller ikke samme effekt som for evaln: 


't'[k];


evaln brugt til udstilling:


evaln( t[k] ) = t[k];


evaln( sin(Pi) ) = sin(Pi);


Forhalende apostroffer virker i sidste tilfælde ligeså godt:


'sin(Pi)' = sin(Pi); 


Sammenlign følgende tre kommandoer:


evaln( sin( Pi+cos(Pi/2) ));


'sin( Pi+cos(Pi/2) )';


'sin'( Pi+cos(Pi/2) );


For at forstå output fra evaln( sin( Pi+cos(Pi/2) ));  skal man huske, at evaluering stopper lige før 
der ikke længere er et objekt, der kan tilordnes til.


Accenter har en helt anden virkning end apostroffer:


x:=3:   `sin( x*Pi/2)`=sin( x*Pi/2); 

Der er stor forskel på udtrykkene  `sin(x*Pi/2)` og sin(x*Pi/2). Maple opfatter det første som kun 


et navn, mens det andet opfattes som sinusfunktionen anvendt på  . 


Dette må dog ikke misforstås. Hvis et navn omgærdet af accenter også er et gyldigt navn uden 
accenter, så er accenterne overflødige. Der er forskel på while og `while` fordi while er et 
reserveret ord, altså ikke et gyldigt navn. Der er ingen forskel på S1 og `S1  ̀i følgende eksempel:


S1:=47; `S1`;


47


Proceduren parse forsøger at læse et navn eller en streng som en Maple-kommando:


str:="sin( x*Pi/2)";


parse(str); 


Da x ovenfor fik tillagt værdien 3, fås videre:


%;


Tilføjelsen statement er nødvendigt i følgende eksempel:


parse("for k to 3 do k^3 end do",statement);
1


8


27


27


parse virker ligeså godt på navne defineret ved brug af accenter:


nvn:=`sin( x*Pi/2)`; 


parse(nvn);


%;


Ved en convert-kommando kan vi gå den modsatte vej:


str:=convert( 'sin(x*Pi/2)',string); 


nvn:=convert( 'sin(x*Pi/2)',name);


parse(str)=eval(parse(str)); 

parse(nvn)=eval(parse(nvn));


Beskyttede navne (reserverede ord)


Som nævnt i begyndelsen af kapitlet er der visse navne, der er reserverede af Maple til eget brug. 
Der er dels de såkaldte nøgleord (keywords), dels er der navnene på de mange Mapleprocedurer. 
Nøgleordene er mestendels de ord, der bruges til programmering, så som if, then, else, while. 
Maple vil selv gøre vrøvl, hvis man forsøger at bruge et reserveret ord:


expand:=opvarm+luft;


Error, attempting to assign to `expand` which is protected


Vil man undersøge på forhånd om et navn er beskyttet (protected), kan man gøre som følger:


type(expand, protected);
true


type( sin, protected);
true


type( sinus, protected);
false


Man kan beskytte sine egne navne vha. proceduren protect (se hjælpen).

1. 1. 


3. 3. 


2. 2. 


13. Funktioner


Definition af funktion i Maple


En funktion f er en afbildning, der til ethvert element x i definitionsmængden lader svare et 
element (billedet af x ved f) i en anden mængde. Dette billede af x ved f  betegnes f(x).


Når elementet x ikke er konkretiseret, men kun er et bogstav, da er f(x) et udtryk i x. Funktionen er
f, udtrykket i x er f(x).


Maple opfatter en funktion f som en procedure, dvs. en opskrift, der beskriver, hvad f skal gøre 
ved et argument x, ifald f anvendes på det. Der kan skrives procedurer, der laver mange forskellige
ting. Den type procedure, der mest intuitivt svarer til funktionsbegrebet, vil vi undertiden kalde en
pileprocedure. 


En funktion kan defineres på 3 måder i Maple:


Ved syntaksen      f :=  x  KONKRET formel i x;
Pilen laves af et minus og et `større end`-tegn. 
Intet evalueres ved definitionen. Først når funktionen kaldes (dvs. bruges på et argument, et 
input), evalueres der. (Automatiske simplificeringer udføres dog). 
Ved syntaksen      f := unapply( udtryk, x);
når et udtryk i x (hér kaldet udtryk) er givet (evt. defineret i forvejen ved kommandoen
udtryk:=konkret formel i x;). 
udtryk evalueres fuldt ved definitionen. 
Resultatet er i begge ovennævnte tilfælde en pileprocedure.
Ved direkte brug af generel procedure-definition:    f := proc(x) opskrift end proc;  Intet 
evalueres ved definitionen. Først når proceduren kaldes, evalueres der.


Eksempel


Metode 1:  . Funktionen er givet direkte (konkret) ved sin forskrift og vi er 
klar til at taste den ind. 
Her bør man bruge piledefinitionen:


f:= x -> a*x^2*sin(x); 


Et eksempel på et kald til funktionen f:


f(Pi/3);


Metode 2: Vi har et udtryk indeholdende en variabel (x). Ud fra dette udtryk ønsker vi dannet 
en funktion, som vi vil kalde g:
Her bruges unapply.


Som eksempel lader vi det givne udtryk være 

 :


udtryk:=int((8-x)/(x-3),x); 


Vi danner en funktion, der ved input af x som output giver værdien af udtryk:


g:=unapply( udtryk, x);


Vi ser, at udtryk blev evalueret ved selve definitionen af g. 
Et eksempel på et kald til funktionen g


g(7);


Bemærk, at piledefinitionen slet ikke virker efter hensigten her:


gpil:=x->udtryk;


Vi ser, at udtryk ikke blev evalueret ved definitionen: Der står stadig udtryk efter pilen i den blå
udskrift.
Vi prøver et kald til funktionen:


gpil(7);


Dette output fås faktisk uanset hvilket input, man giver funktionen. Grunden er, at det x, der 
optræder i udtryk først ses, når funktionen bliver kaldt (dvs. brugt som en funktion), men så er 
det for sent. Her skal altså bruges unapply!
I alle tilsvarende situationer, hvor funktionen f ønskes defineret ved f(x) = udtryk og udtryk er 
(eller indeholder) en variabel, der er tilordnet en værdi indeholdende x, vil brugen af unapply 
være nødvendig.
Derimod går følgende godt, fordi variablen u1 ikke er tilordnet en værdi indeholdende x:


u1:=34: 


f:=x->u1*cos(x);


Ganske vist blev u1 ikke evalueret ved definitionen, men det er ikke et problem, da u1 ikke er 
defineret ud fra x.


f(t);


Metode 3: Output for vores funktion kan først findes efter en eller flere mellemregninger, eller
det er måske mere bekvemt sådan. Vi kunne også have brug for lokale variable til at opbevare 
mellemresultater.
Her bruges generel proceduredefinition.
Vi tager et simpelt eksempel, der i virkeligheden blot definerer den sammensatte funktion 


:


h:= proc(x) local r,s; r:=x^2; s:=sin(r); exp(s)  end 


proc; 


Et eksempel på et kald til funktionen h:

h(u);


e


og et mere konkret:


h(sqrt(Pi));
1


Når det kan lade sig gøre, brug da den direkte pilemetode (1). 


Hvad med syntaksen F(x) : = formel ?
Man kan godt i Maple definere et udtryk i x på følgende måde:


F(x):=sin(2*x)+3*x; 


Men Maple kender så kun F(x). Den vil ikke kunne udregne F(0), F(1) eller F(t):


F(0),F(1),F(t),F(x);


Ved tilordningen F(x):=sin(2*x)+3*x  noterer Maple sig, at F er en funktion, og at ved input af
bogstavet x (og kun det!) skal F levere outputtet sin(2*x)+3*x. Dette sikres ved anbringelse i 
husketabellen for funktionen F. Denne fås som fjerde operand i F:  


op(4,eval(F));


Ellers ved Maple intet om funktionen F:


eval(F);


Det der står i output er, at proceduren F skal huske det, den udregner (option remember). 
Iøvrigt skal den blot returnere uevalueret ('procname(args)'). Mere om procedurer i kapitel 15.
At anbringe specialtilfælde i en funktions husketabel kan undertiden være en god idé: 


F:= x -> sin(x)/x;


Denne funktion er åbenbart ikke defineret for x = 0:


F(0); 


Error, (in F) numeric exception: division by zero


Vi tilføjer den ekstra information til F's husketabel:


F(0):=1:

F(0);
1


F har nu fået en husketabel:


op(4,eval(F));


men er dog ikke derved sat til at huske alt, hvad den udregner, det kræver, at den eksplicit får
option remember (se kapitel 15):


F(Pi/2); op(4,eval(F));


Man må passe på rækkefølgen, idet en ny proceduredefinition sletter alt om den gamle, 
inklusive husketabellen:


G(7):=53: 


G:= x -> x^2: 


G(7);
49


Funktion af flere variable


Som f ovenfor defineres en funktion g af to variable som en pileprocedure enten direkte eller ved 
brug af unapply:


g:= (x,y) -> exp(x+2*y)*tan(x-y^2);


Ved brug af unapply:


udtryk:=exp(x+2*y)*tan(x-y^2):  


g:= unapply( udtryk, x, y);


og vi har:


g(1,2);


evalf(%);
21.15578278


Fuldstændig som for funktioner af én variabel opfatter Maple her g som en procedure, hvilket ses 
ved følgende kommando:

lprint( eval(g));


proc (x, y) options operator, arrow; exp(x+2*y)*tan(x-y^2) end 


proc


Evaluering af funktionsnavn


Bemærk Maples svar i det følgende eksempel, hvor vi først sætter f1 til at være det samme som f:


f:=x->x*ln(1+x);


f1:=f;


f1,f;


Vi ser, at f slet ikke evalueres, men at f1 evalueres til f. Funktioner (procedurer) evalueres kun til 
det sidste navn. Man kan dog få den konkrete definition frem ved følgende kommando:


eval(f1); eval(f);


Normalt er det lettere blot at bede om f(x) (man taler om et kald til proceduren):


f(x);


En procedure evalueres altså først (til andet end et navn), når den bliver kaldt (som i eksemplet 
lige ovenfor), eller når eval anvendes som vist.


Sammensætning af funktioner
Lad f være funktionen fra eksemplet ovenfor. Vi kontrollerer lige, at f(x) er det, som det burde 
være:


f(x);


Lad g være funktionen


g:= x -> sqrt(x); 


Sammensætning af funktioner sker ved brug af tegnet @ (snabel a):

f@g; 


Evalueringen af  f@g sker først, når denne sammensatte funktion bliver kaldt, dvs. bliver anvendt 
på et argument x:


(f@g)(x); 


(f@g)(4);


En funktion kan sammensættes med sig selv, enten således:


(g@g)(x); 


eller således:


(g@@2)(x);


g sammensat med sig selv 8 gange:


(g@@8)(x);


Den inverse funktion til en funktion F findes som følger, men Maple bestemmer den kun i meget 
specielle tilfælde.
Først renser vi dog F for tilordninger:


F:='F':


(F@@(-1))(x);


(ln@@(-1))(x);


e
x


ln@@(-1);
exp


(exp@@(-1))(x);


exp@@(-1);
ln


sin@@(-1);
arcsin


tan@@(-1);

arctan


Sammensætning af en funktion og dennes inverse giver den identiske funktion, dvs. funktionen 
:


G:=F@@(-1);


(G@F)(x);
x


(F@G)(x);
x


Stykkevist definerede funktioner: piecewise


Funktioner givet ved forskellige regneudtryk på forskellige intervaller (Tuborgfunktioner) kan i 
Maple defineres ved brug af den meget nyttige procedure piecewise. Eksempel: 


w:=piecewise(x<-Pi,sin(x),x<Pi,Pi-abs(x),1+cos(x));


Første argument skal være en betingelse, typisk en ulighed, hér  . Hvis denne betingelse er 


opfyldt for det konkrete x, så er resultatet næste argument (hér  ). Hvis derimod uligheden 
ikke er opfyldt, så går Maple videre til tredie argument. Hvis dette er en betingelse (hér er det 


 ) og denne er opfyldt, så er resultatet fjerde argument (hér  ). Sådan kan fortsættes. 


Sluttes (som hér) med et udtryk uden foranstående betingelse, så er det resultatet, når ingen af de 
øvrige betingelser er opfyldt. Ellers sættes værdien til 0.


Vi laver en funktion ud af w:


f:=unapply(w,x);


Vi afprøver funktionen f:


f(-3*Pi/2),f(-Pi/2),f(3*Pi/2);


Følgende går som ønsket:


S:=x->piecewise( x=0, 1, sin(x)/x);

S(0);
1


I tidlige releases (før 8) kom en fejlmelding om division med nul, fordi alle argumenter blev 
evalueret fuldt ud før stilling blev taget til hvilken værdi skulle bruges.
Bemærk forskellen mellem følgende to:


piecewise(0=0,1,0/'sin(0)');
1


piecewise(0=0,1,'sin(0)'/0);


Error, numeric exception: division by zero


Den sidste fejlmelding skyldes, at automatisk simplifikation kommer ind før evaluering.


Tal som konstante funktioner


Konkrete tal kan af Maple opfattes som konstante funktioner:


7(x), 3.14(y), -34(x,y), 47.11(x,y,z,w), sqrt(2)(x), sin(2)


(x,y);


Dette gælder dog ikke tal som , der i denne sammenhæng kun opfattes som et navn: 


Pi(2);


evalf(%);


Glemmer man gangetegnet i udtrykket  vil man se, at noget er galt. 5 bliver opfattet som 
den konstante funktion 5 anvendt på x + y:


5(x+y);
5


 Den konstante funktion 2 sammensat med f :


(2@f)(x);
2


I modsat rækkefølge


(f@2)(x);


Man bør ikke bruge skrivemåden , da Maple opfatter den anderledes end tilsigtet: 


sin^2(x);


Det der først evalueres er 2(x). Herved fås 2. Derefter kommer `^` igang.
Man skal i stedet skrive:


sin(x)^2;

I Maple 8 og tidligere blev en funktion defineret som en numerisk konstant ved definitionen 
automatisk simplificere til konstanten. Da en konstant som vist kan bruges som en funktion, 
havde denne simplifikation da heller ingen katastrofale konsekvenser. Den var imidlertid 
uhensigtsmæssig, så simplifikationen udføres ikke i senere versioner:


g:=x->2;


I Maple 8 og tidligere er output af følgende blot sin:


g:=x->sin(x);


En lille finurlighed: Det tomme navn som funktion


Betragt resultatet af en primtalsopløsning:


p:=ifactor( 1234567890987654321 );


Det interessante er, at Maple ikke ganger ud, selv om resultatet tilsyneladende er et produkt af 
hele tal, der jo ellers altid automatisk bliver ganget sammen.
Sammenlign følgende to input:


p;   (3)^2*(7)*(19)*(928163)*(1111211111);


1234567890987654321


Ved brug af expand får vi primtallene ganget sammen igen:


expand(p);
1234567890987654321


Hvad foregår der her? Lad os pille en af operanderne i p ud. Tag f.eks. den tredie operand:


h:=op(3,p);


Hvorfor parentesen? Det afsløres, hvis vi spørger hvilken type h har:


whattype(h);
function


Med type function mener Maple et uevalueret kald af en funktion, som f.eks. sin(2), sin(x) eller f
(x). Hvad er da navnet på den funktion, som h er et kald til? Navnet fås som den 0'te operand:


op(0,sin(2));
sin


Men den nulte operand i h er ikke synlig:


op(0,h);

men der er tale om et navn:


type(%,name);
true


Navnet er det tomme navn, ``. Lad os prøve:
 


``(1234)*``(5678);


expand véd hvad den skal gøre ved det tomme navn `` :


expand(%); 
7006652


Funktioner kendt af Maple


Den naturlige logaritme betegnes i Maple med ln, men man kan også bruge log. Logaritmen med 
grundtallet b hedder log[b]. Titalslogaritmen hedder derfor log[10], men man kan også i dette 
tilfælde blot skrive log10: 


log( exp(123)), log(a);


log[10](1000);
3


Samme resultat fås ved kommandoen


log10(1000);
3


For en liste over de af Maple kendte funktioner udfør følgende kommando:


?inifcns


Listen er meget lang. Vi nævner kun her, at arctan kan bruges med to argumenter, arctan(y,x). 
Denne angiver hovedargumentet for det komplekse tal x+iy:


arctan(1,1), arctan(1,0), arctan(2,-2), arctan(0,-3), arctan


(-3,-3), arctan(-4,4);


Se også hjælpen til FunctionAdvisor. Her får vi blot FunctionAdvisor til at give os listen med 
elementære funktioner:


FunctionAdvisor(elementary);


The 26 functions in the "elementary" class are:

Typecheck af input til funktion


Hvad enten man vælger at definere sin funktion ved brug af pilenotation, unapply eller egentlig 
procedure-definition, kan man lave typecheck af input før det gives videre til beregning.
Vi giver nogle eksempler:


I det første eksempel defineres en kvadratrod, der kun vil have ikke-negative tal som input:


kvrd:=(x::nonnegative) -> sqrt(x);


kvrd(-4);


Error, invalid input: kvrd expects its 1st argument, x, to be of 


type nonnegative, but received -4


kvrd(4);
2


Typecheck kan også laves på funktioner af flere variable:


f:=(x::positive,n::integer) -> x^n;


f(2.1,-4);
0.05141890467


f(-2.1,-4);


Error, invalid input: f expects its 1st argument, x, to be of 


type positive, but received -2.1


Ved brug af unapply er syntaksen (i samme eksempel):


unapply(x^n, x::positive,n::integer);


Egentlig procedure-definition tillader naturligvis også typecheck før beregningerne i selve 
procedure-kroppen går igang: 


g:=proc(x::string) local L; 


L:=length(x); 


if L<10 then 


   cat(x,". For kort: Vi føjer noget til!") 


  elif L>15 then


   substring(x,1..15)


  else

   x 


end if 


end proc;


g(7);


Error, invalid input: g expects its 1st argument, x, to be of 


type string, but received 7


g("Peter");
"Peter. For kort: Vi føjer noget til!"


g("Der var engang en mand");
"Der var engang "


g("Rønnebærgelé");
"Rønnebærgelé"


Mere om typer i Maple i kapitel 14. Se også programmeringskapitlet, kapitel 15. Prøv evt. 
hjælpen:


?type








Kapitel 14: Tabeller og Arrays, Manipulation af Datastrukturer


Kapitel 14 side 1


14. Tabeller og Arrays, Manipulation af Datastrukturer


Dataopbevaring sker i forskellige strukturer. Vi har allerede stiftet bekendtskab med følger, lister, 
mængder, matricer og vektorer. 


Tabeller og arrays


Tabeller


Tabellen kan være nyttig som dataopbevaringsmiddel bl.a. fordi dens størrelse ikke skal 
fastlægges på forhånd. 
Vi illustrerer brugen. Først viser vi en eksplicit definition af en tabel (en tabel over vægten af 
forskellige personer):


vægt:=table( [Jensen=87, Olsen=59] );


Som det er tilfældet for funktioner, evalueres tabeller kun til deres navn:


vægt; 
vægt


Fuld evaluering kan dog tvinges igennem:


eval(vægt);


Olsens vægt fås ved kommandoen


vægt[Olsen];
59


Vores tabel er temmelig lille. Tilføjelser til tabellen kan gøres på simpel vis:


vægt[Frederiksen]:=72; 


vægt[Jensen, `før slankekur`]:=100;


eval(vægt);

Tabellens højresider fås således:


entries(vægt);


Tabellens venstresider fås således:


indices(vægt);


Man kan også få indholdet i tabellen at se ved brug af print:


print(vægt); 


print printer kun til skærmen. Der er intet output.


En tabel dannes implicit i det øjeblik, der tilordnes en indiceret variabel:


T[guf]:="flødebolle";


eval(T);


Hvis tabel-elementerne ikke gives som ligninger, så bliver indices de naturlige tal:


prioritering:=table([Mikkel,Rasmus,Palle,Kristine]);


indices(prioritering);


Hvis vi ønsker at ændre en tabel, men samtidigt gemme den gamle intakt, vil den umiddelbare 
idé ikke virke:


pri2:=prioritering;


Vi ændrer på indholdet af pri2:


pri2[1]:=Rasmus: pri2[2]:=Mikkel:


eval(pri2);


eval(prioritering);


Vi ser, at begge er ændret. 
Situationen er den samme som for matricer og omtalt i kapitlet om lineær algebra.
Løsningen er analog:


pri3:=copy(pri2);

eller man kunne blot have sagt: pri3:=eval(pri2);


pri3[3]:=Per: 


eval(pri3);


eval(pri2);


Eksempler på tabeller: Pakker i Maple


Enhver af Maple's gamle pakker (til og med Release 5) er en tabel, hvis indices er navnene 
på de i pakken indeholdte procedurer. Som tabelværdi svarende til et bestemt 
procedurenavn har den ikke selve proceduren, men i stedet en readlib-kommando, der 
læser proceduren, når denne bliver kaldt. Vi betragter som eksempel en lille pakke,
inttrans-pakken:


type(inttrans,table);
true


indices(inttrans);


eval(inttrans);


type(laplace,procedure);
false


type( inttrans[laplace],procedure);
true


type(`inttrans/laplace`,procedure);
true


Efter with-kommandoen refererer laplace til samme procedure som inttrans[laplace] og
`inttrans/laplace`:


with(inttrans);


type(laplace,procedure);
true

evalb betyder evaluate boolean og returnerer true hvis input er en sand påstand, ellers 
returneres false:


evalb(eval(`inttrans/laplace`)=eval(laplace));
true


Nyere pakker er udformet som moduler (herom i kapitlet om programmering). Således er 
pakken LinearAlgebra et modul. Gamle pakker vil ved revision blive til moduler. Det er 
således i Maple 9 sket med pakken numapprox. og i Maple 11 med plots-pakken:


type(eval(plots),table);
false


type(eval(LinearAlgebra),table);
false


whattype(eval(LinearAlgebra));
module


type(eval(plots),`module`);
true


At der blev brugt accent grave i ̀ module` var nødvendigt, fordi ordet module bruges til 
definition af et modul.


arrays (med lille a): Bør ikke bruges!


Fra og med Release 6 er der to forskellige slags arrays, nemlig den gamle array og den nye
Array. Vi behandler i dette afsnit den gamle array. Selv om man ifølge hjælpen bør bruge den
nye Array i stedet, er det nødvendigt at kende til eksistensen af hensyn til muligheden for 
skrivefejl og af hensyn til gamle worksheets. 
En array kan betragtes som en speciel form for tabel. Den må defineres som array på forhånd. 
Ved indiceringen kan kun bruges hele tal eller talsæt og størrelsen fastlægges ved definitionen 
(enten eksplicit eller implicit).  En éndimensional array skal have indices ,


hvor  og hvor p og q er hele tal. En flerdimensional array har indices (i, j, k,...), hvor i, j, 


k på samme måde gennemløber intervaller. En todimensional array er altså i realiteten en 
matrix.


Hér er et eksempel på en 2-dimensional array:


A1:=array(0..3,2..3, [ (1,3)=56, (2,2)=-98]);


Ved definitionen af A1 vises den, og ikke-definerede elementer refereres blot til ved deres 
navne som f.eks. . Vil man efterfølgende se indholdet i A1 er det ikke nok kun at nævne


A1. En array evalueres kun til sit navn ganske som for tabeller:


A1;
A1

eval(A1);


Vi bemærker, at ikke definerede elementer nu refereres til ved indicerede spørgsmålstegn.
En to-dimensional array kan også gives som følger, hvor indiceringen ikke er angivet eksplicit,
men størrelsen deduceres ud fra de opgivne data:


A2 := array ( [[1,7,-5] , [3,4]] ); 


Der bruges da standardindicering (dvs. elementerne har række- og søjlenumre fra 1 og opefter)
og arrayen vises som en matrix. 
De enkelte elementer af en array fås som for tabeller:
 


A2[1,2];
7


En array med de samme elementer som A2, men med en non-standard indicering kan fås 
således:


A3 := array (0..1,0..2, [[1,7,-5],[3,4]] );


Proceduren array accepterer (som også table ) et ekstra argument, der er et af navnene sparse, 
symmetric, antisymmetric, identity,  diagonal. Navnet sparse betyder i Maple, at alle 
elementer, der ikke udtrykkeligt defineres, sættes til 0.


array(sparse,1..3,1..4,[(1,2)=7,(3,4)=abc]);


Som nævnt kan en array have vilkårlig dimension. Her er et tredimensionalt eksempel:


array(1..2,4..5,0..1,[(1,4,0)=777, (1,4,1)=888]);

Hvis vi definerer en 1- eller 2-dimensional array og bruger sædvanlig indicering (starter med 1
og går op), så får vi en vektor eller en matrix:


A:= array( [[1,4,9],[16,25,36]]);


c := array ( [2,3,-6] );


Arrays (med stort A)


En Array kan betragtes som en speciel form for tabel (en rtable). Den må defineres som Array på 
forhånd. Ved indiceringen kan kun bruges hele tal eller talsæt og størrelsen fastlægges ved 
definitionen (enten eksplicit eller implicit).  En éndimensional Array skal have indices 


, hvor  og hvor p og q er hele tal. En flerdimensional Array har indices 


(i, j, k,...), hvor i, j, k på samme måde gennemløber intervaller. En todimensional Array er altså i 
realiteten en matrix.


A1:=Array(0..3,2..3, {(1,3)=56, (2,2)=-98});


Bemærk tuborg-parenteserne. Ikke-definerede elementer sættes automatisk til 0: 


A1[2,3];
0


Navnet på en Array evalueres helt:
 


A1;


En Array kan også defineres ved brug af lister. Endvidere kan datatypen angives, default er typen
anything:


A2 := Array ( [ [a,b,c] , [d,e,f] ] ,datatype=name); 


Vi ser, at A2 bliver printet som en matrix. 


Der bruges standardindicering (dvs. elementerne har række- og søjlenumre fra 1 og op) , når ingen

indicering er givet. En Array med de samme elementer, men med en non-standard indicering kan 
fås således:


A3 := Array (0..1,0..2, [ [a,b,c] , [d,e,f] ] ,datatype=name)


;


Array accepterer ekstra argumenter. Bl.a. kan man angive en indiceringsfunktion, eksempelvis
symmetric, antisymmetric, identity,  diagonal. 
Hvis datatypen tillader det (f.eks. hvis den ikke angives) så sættes manglende elementer 
automatisk til nul.


Array(antisymmetric,1..3,1..3,{(1,2)=7,(3,1)=abc});


Array(identity,1..3,1..3);


Nedenfor defineres A4 som en diagonal Array. Da ingen elementer defineres i første omgang, 
sættes alle til 0. 
 


A4:=Array(diagonal,1..3,1..3);


Vi omdefinerer nu (1,1)-elementet:


A4[1,1]:=56;


A4;


Vi forsøger herefter at omdefinere (1,2)-elementet:

A4[1,2]:=43;


Error, attempt to assign non-zero to off-diagonal entry of a 


diagonal Array


Da A4 var defineret som en diagonal Array, får vi ikke lov.


Som nævnt kan en Array have vilkårlig dimension:


Array(1..2,4..5,0..1,{(1,4,0)=777, (1,4,1)=888});


Højre-klik på output giver bl.a. mulighed for at studere elementerne (browse).


Mængder og lister


Mængder har vi tidligere mødt f.eks. ved løsning af flere ligninger samtidigt. Ved definition af 
mængde bruges tuborg-klammer:


A:={Peter, Lise, Morten, Peter}; 


B:={Torben, Ilse, Annette, Lise};


Maple opbevarer ikke nødvendigvis elementerne i en mængde i den rækkefølge, hvori de 
indtastes.
Foreningsmængden, fællesmængden og mængdedifferensen fås således:


A union B;


A intersect B;


A minus B;


Udsagnet A subset B returnerer sandt, hvis A er en delmængde af B:


% subset A;
true


Medlemskab af mængde:

(Peter in A) = evalb(Peter in A);


member(Peter,A);
true


Lister er nyttige, når elementrækkefølgen er væsentlig, eller når et element skal forekomme flere 
gange. Ved definitionen bruges kantede parenteser. 


L:=[ sin , cos , tan , cot, sec];  


De enkelte elementer kan hentes på følgende måde:  


L[2];     # Element nr.2 fra venstre
cos


L[2..4];  # Delliste


L[-2];    # Element nr. 2 fra højre
cot


L[3..-2]; # Delliste


L[];      # Følgen af samtlige elementer


men følgen af samtlige elementer fås også som op(L) og det tredie element ved op(3, L); Se næste
afsnit.
Tilføjelser til en liste kan gøres således:


L:=[op(L),csc];


Hvis listen er lange og den slags operationer skal udføres ofte, er det bedre at bruge en tabel, da 
listen skal dannes på ny ved enhver tilføjelse.
Medlemskab af en liste kan afgøres ganske som for mængder, se ovenfor.


I pakken ListTools findes værktøj til behandling af lister.
Et eksempel:


M:=LinearAlgebra:-RandomMatrix(3,4,generator=-2..2);


LM:=convert(M,listlist);


ListTools:-Flatten(LM);

ListTools:-MakeUnique(%);


Vi sorterer listen (sorteringskriteriet er som default `<`, se også nedenfor om sortering):


sort(%);


Begrebet operand


Alle udtryk i Maple består af operander: For en sum er det leddene, for et produkt er det 
faktorerne, for en mængde eller en liste er det elementerne:


p:=x*y^4+5-7+f(z,w);


Følgen af samtlige operander i et udtryk fås ved brug af op:


op(p); 


L;


op(L); 


Enkelte operander kan pilles ud således:


op(3,L);      # Tredie operand i L
tan


op(2..4,L);   # Operand nr. 2 til 4 i L


op(-1,L);     # Den sidste operand i L
csc


op(-3..-1,L); # De 3 sidste operander i L


Antallet af operander i et udtryk fås ved brug af nops (number of operands):


nops(L); 
6


op(0, udtryk) giver ofte udtrykkets type. For uevaluerede funktionsudtryk (som f(2) ) fås dog 
navnet på funktionen, altså f. (Se iøvrigt hjælpen for op).


op(0, sin(3));
sin

op(0,L); 
list


op(0,p);
`C`


Vil vi omdanne listen L til en mængde kaldet S, kan det f.eks. gøres således:


S:={op(L)};


op(0,S);
set


Men det samme kan også opnås ved en convert-kommando:


convert(L,set); 


convert-kommandoer ændrer ikke det oprindelige udtryk:


L; 


Vi kan tilsvarende gå fra mængde til liste enten ved


[op(A)]; 


eller ved


convert(A,list); 


Operanderne er selv udtryk, der er sammensat af operander:


p;


op(1,p);


op(%);


op(2,op(1,p));


op([1,2],p); # Alternativ til ovenstående


op(%);

op([1,0],p); # Viser, at første led i p er et produkt
`*`


op([1,2,2],p); 
4


Sidste operand i p:


op(-1,p);


Dette udtryks operander er


op(%);


Procedurerne map, map2, zip og elementvise operationer


Vil man gøre et eller andet ved samtlige operander i en liste, mængde eller udtryk, kan det gøres 
ved de meget nyttige procedurer map eller map2. Hvis man kun vil behandle en eller flere 
operander, kan man bruge proceduren applyop (se senere).


Med Maple 13 kan elementvise operationer udføres på containere: tabeller, mængder, lister, 
matricer og arrays ved tilføjelse af en tilde ~ til den givne operator eller procedure. Dette vil blive 
vist ved eksempler parallelt med behandlingen af map.
Hjælpen til elementvise operationer fås ved


?operators[elementwise]


I vores første eksempel ønsker vi at bruge sinus-funktionen på hver af faktorerne i produktet x*y, 


selv om vi godt ved, at  ikke er det samme som . Maple adlyder:


map( sin, x*y); 


Følgende udtryk lader sig ikke faktorisere, men hvert af dets to led kan faktoriseres. Dertil kan
map bruges:


u:=(x^2+x)*exp(x)+(y^2-3*y+2)*(x^3+8);


map( factor, u);


Nedenfor anvendes differentiationsoperatoren D på hver af elementerne i listen L. Vi minder først 
om indholdet i L:


L;


map( D, L);

Differentiationsoperatoren D kan imidlertid godt selv finde ud af at 'mappe':


D(L);


Ønsker vi at evaluere alle funktionerne i L i samme punkt, f.eks.  kan vi gøre således:


map( f -> f(Pi/6), L); 


En seq-kommando kan ofte bruges i stedet. Et alternativ til ovenstående metode er:


seq( f(Pi/6), f=L);


I dette tilfælde er Maple dog i stand til at klare udregningerne helt uden map eller seq:


L(Pi/6);


Næste eksempel viser, at map også kan bruges, når den funktion, der skal mappes kræver flere 


argumenter. Vi ønsker at bestemme  for enhver af funktionerne i listen L. Det ekstra 


argument til int (intervalangivelsen) angives som det sidste argument i map:


map( int, L(x), x=Pi/6..Pi/3);


I stedet for map kan igen bruges seq:


seq( int( f(x), x=Pi/6..Pi/3), f=L);

Her kan den elementvise operation int~ også bruges, idet L(x) er en liste og dermed en container:


int~(L(x),x=Pi/6..Pi/3);


Proceduren map2 kan bruges, når den funktion, der skal mappes, tager et fast første argument, 
som i næste eksempel, hvor Taylor-polynomier for sinus-funktionen bliver fundet ud fra 3 
forskellige udviklingspunkter. Her er også et fast sidste argument, nemlig trunkeringsordenen:
 


map2( taylor, sin(x), [x=0,x=Pi/2,x=Pi], 4);


Vi fjerner O-leddene. Dette sker også ved en map-kommando:


map(convert,%,polynom);


Vi prøver også her de tilsvarende elementvise operationer:


taylor~(sin(x),[x=0,x=Pi/2,x=Pi], 4);


convert~(%,polynom);


Kommer det argument, der skal 'mappes' på først som n'te argument, kan bruges map[n]. 
Her findes Taylorpolynomierne for sinus med udviklingspunkt 0 og med 
trunkeringeringsordenerne 2, 4, 6 og 8: 


map[3](mtaylor,sin(x),x=0,[2,4,6,8]);


Elementvis version:


mtaylor~(sin(x),x=0,[2,4,6,8]);

To lister (eller vektorer/matricer) kan 'lynlåses' sammen til en liste vha. zip. 
I første eksempel skal vi ud fra listerne [a, b, c] og [6, 9, -3] danne en ny liste bestående af 
produkterne af elementerne:


zip( (x,y)-> x*y, [a,b,c], [6,9,-3]);


eller kortere


zip( `*`, [a,b,c], [6,9,-3]);


Elementvis version:


[a,b,c] *~ [6,9,-3];


Bemærk resultatet, når listerne ikke er lige lange. 


zip( `*`, [a,b,c,d,e,f], [6,9,-3]);


Her tilføjes som fjerde argument en værdi som bruges til udfyldning af den korte liste:


zip( `*`, [a,b,c,d,e,f], [6,9,-3],37);


To matricer zippes:


zip((s,t)-> s/t,Matrix(2,2,symbol=a),Matrix(2,2,symbol=b));


Ovenfor kan (s, t) -> s/t  erstattes af den korte version  `/`.
Elementvis version:


Matrix(2,2,symbol=a)/~Matrix(2,2,symbol=b);


Proceduren zip er nyttig til, ud fra en tabel af sammenhørende x- og y-værdier, at danne en liste af 
punkter, der så kan plottes:


xliste:=[1,2,3,7]: 


yliste:=[-1,1,2,-1]:


punkter:=zip( (x,y) -> [x,y], xliste, yliste);


Også her kan en elementvis operation bruges: 

`[]`~(xliste,yliste);


plot( punkter, style=point, symbol=solidcircle, symbolsize=


40,color=red); 


2 3 4 5 6 7
0


1


2


Proceduren applyop


Betragt udtrykket


u:=(a+b/c)^2*(1/x+1/(x+y));


hvor vi ønsker  ganget ud, men intet gjort ved .  Så kan vi gøre således


expand(u, 1/x+1/(x+y));


Hvis vi nu egentlig gerne ville have   udenfor parentes, kan vi jo prøve med factor:


factor(%);


Men det var ikke lige det vi ønskede. En simpel løsning er at bruge proceduren applyop:


applyop( expand, 1, u);

Denne kommando betyder, at expand bliver brugt på første (1-tallet) operand i u. Udtrykket u er et
produkt med faktorerne:


op(u);


Første faktor:


op(1,u);


Her bruger vi normal med tilføjelsen expanded på anden operand i u:


applyop( normal, 2, u,expanded);


Bruger vi normal med tilføjelsen expanded på det oprindelige udtryk, fås resultatet på en anden 
form:


normal(u,expanded);


Vi giver endnu et eksempel.
Betragt udtrykket


w:=x/(x^2+3)+3/(x-4)-7/(x+4);


Vi ønsker, at sætte de to sidste brøker på fælles brøkstreg. Proceduren normal direkte brugt vil 
sætte hele udtrykket på fælles brøkstreg. Vi bruger først applyop til at forsyne de to led med en 
koefficient, som vi kalder k. Så kunne vi bruge selectremove (se senere i kapitlet) til at opdele 
udtrykket i to led: ét, der indeholder k, og et, der ikke gør. Denne fremgangsmåde tillader 
individuel behandling af de to led. I vores eksempel skader det ikke at behandle de to led ens, så 
vi bruger collect i stedet:
Operanderne i w er


op(w);

wa:=applyop(s->s*k,{2,3},w);


collect(wa,k,s->normal(s,expanded));


subs(k=1,%);


En procedure applysome, der tillader anvendelse af en Mapleprocedure på et deludtryk af et givet 
udtryk, er angivet i kapitel 16.


Sortering


Maple har en sorteringsprocedure


L:=[ sin , cos , tan , cot, sec];  


sort(L);


sort(L,lexorder);


Default-sorteringskriteriet er øjensynligt alfabetisk (lexorder) i dette tilfælde.
Sortering efter maskinadresse:


sort(L,address);


Maskinadresserne:


addressof~(%);


Vi sorterer adresserne efter defaultkriteriet `<` og undersøger, om listen var sorteret i forvejen:


sort(%)-%;


L;    # Intet sket med det oprindelige L


sort kan (som vist) tage et andet argument. Det skal være et af sort kendt navn (ovenfor address, 
lexorder) eller en boole'sk funktion af 2 variable, dvs. en funktion, hvis værdimængde er {true,
false}. 
Her sorteres punkterne i  punkter efter voksende andenkoordinat:


punkter;


sort(punkter,(x,y)->evalb(x[2]<y[2]));

Output fra evalb (evaluate boolean) er netop enten true eller false (dog kan evalb returnere 
uevalueret).


For polynomier er sortering destruktiv i den forstand, at det oprindelige polynomium overskrives:
 


p:=56+x^3-x+5*x^2;


sort(p);


p;


Typer


Typekontrol kan være nyttig specielt, hvis man vil skrive sine egne procedurer. Maple opererer 
med mange forskellige typer. Vi giver nogle eksempler, men vær opmærksom på, at et Maple-
objekt kan tilhøre flere typer:


whattype(sin);
symbol


type(sin,procedure);
true


map2(type,sin,[name,symbol,string]);


Elementvis version:


type~(sin,[name,symbol,string]);


whattype(eval(sin));
procedure


A;


whattype(A);
set


Med apostroffer om A ser type et navn:


type('A',name);
true


En alternativ måde til afgørelse af typer benyttes specielt i procedurehoveder (se kapitel 15):


evalb(A::name);
false

L;


whattype(L);
list


type(L, list(procedure));    # Sammensat type
true


type(L, list(name));
true


p;


whattype(p);
`C`


type(p,polynom);
true


type(p, polynom(integer));   # og er koefficienterne hele 


tal?
true


Her undersøges om p er et polynomium med heltallige kofficienter i den variable x:


type(p, polynom(integer,x));
true


whattype( x->x^2);
procedure


whattype(sin(2)); 
function


Sidste svar kan måske undre. Med typen function mener Maple 'uevalueret kald af en funktion'. I 
dette tilfælde er sinusfunktionen anvendt på 2, men ikke udregnet. Sammenlign med de næste:


whattype(sin(Pi/2));
integer


type(sin(2), constant);
true


type(sin, function);
false


type(sin, procedure);
true


Sammenlign de to følgende:


type( sqrt(4), integer);
true


type(4^(1/2),integer);

false


Sagen er, at sqrt(4) ved fuld evaluering giver 2, der jo er et helt tal (integer). Derimod evalueres 4^
(1/2) ikke umiddelbart til 2:


4^(1/2);


whattype(4^(1/2));
`^`


Selvfølgelig vil simplify(4^(1/2) ); give 2.


Der kan undertiden være behov for at undersøge, om et givet udtryk er et uevalueret kald til en 
bestemt funktion: 


type(GAMMA(sqrt(5)),specfunc(anything,GAMMA));
true


type(GAMMA(sqrt(5)),specfunc(name,GAMMA));
false


type(f(x),specfunc(name,f));
true


u:=(t^2+7*t-11)*sin(t)+exp(t);


hastype( u, polynom);
true


Svaret er true fordi koefficienten til sin(t) er et polynomium. Mens hastype undersøger typer, 
leder has efter udtryk blandt operander eller underoperander:


has( u, exp);
true


has(u, t^2);
true


Da   ikke er en underoperand for udtrykket u, fås følgende:


has( u, t^2+7*t); 
false


Yderligere oplysninger om typer fås ved kommandoen:


?type

Procedurerne select, remove, selectremove, subsop, algsubs


select, remove og selectremove


Det er undertiden nyttigt at kunne udvælge (eller fjerne) elementer af en bestemt slags fra en 
mængde, liste, sum, produkt eller uevalueret funktionsudtryk. Det sker ved procedurerne
select og remove: 
Lad der være givet følgende mængde af hele tal:


S:={-2,6,8,-4,7,-3}:


Ønsker vi at udvælge de hele tal fra S, der er delelige med 4, kan vi gøre således:


select(x->evalb( irem(x,4)=0 ), S);


eller lidt kortere (men mindre let forståeligt)


select(rcurry(irem,4)=0, S);


hvor vi har brugt rcurry (right curry):


rcurry(irem,4);


Udvælgeskriteriet skal være en boolesk funktion, altså en funktion med værdimængde {true, 
false}.
Delmængden bestående af de positive tal kan findes således:


select( x->evalb(x>0),S); 


eller således


select( type, S, positive); 


Bemærk, at den booleske funktion type kræver to argumenter. Dens andet argument (ovenfor
positive) anbringes som tredie argument til select.
Nedenfor fjerner vi de ulige tal: 


remove( type, S, odd);  


De kunne også være fjernet ved brug af irem således:


remove(x->evalb( irem(x,2)<>0 ), S);


Fra en sum udvælges den delsum, hvori indgår navne:


select( hastype, 5*aha+56+sqrt(2)-Pi, name); 

Fra et produkt fjernes den (eller de) faktorer, der indeholder y:


remove( has, exp(5*x+y)*c*(z^2+x),y);


Tit er man interesseret i både det, der bliver udvalgt, og det, der bliver fjernet. Så kan man 
mere effektivt bruge proceduren selectremove:


lige,ulige:=selectremove(type,S,even);


lige;


ulige;


subsop, algsubs


Proceduren subsop kan erstatte en operand i et udtryk med noget andet. Vi laver først ved 
integration et udtryk, som vi kan bruge til illustrationen:


udtryk:=int((x^2+3)/(x+1)/(x^2+1)+3+2*x,x);


Operanderne i denne sum er leddene:


op(udtryk); 


Vi erstatter 3. operand med hvadsomhelst:


subsop(3=hvadsomhelst,udtryk);


Underoperand 1, 2 i udtryk er


op([1,2],udtryk);


Vi erstatter dette med noget:


subsop([1,2]=noget,udtryk);


Underoperand 1, 2, 0  i udtryk er


op([1,2,0],udtryk);
ln


Vi erstatter dette med LN:

subsop([1,2,0]=LN,udtryk);


Flere erstatninger ved én kommando:


subsop([1,2,0]=LN,[2,1]=tan(x),udtryk);


Den almindelige subs-procedure undersøger, om et af et udtryks operander eller 
underoperander er lig med det, der skal erstattes. 
Følgende forsøg går ikke godt, fordi a + b ikke er en operand i a + b + c:
  


subs(a+b=s,a+b+c);


Proceduren algsubs er lavet til den slags erstatninger:


algsubs(a+b=s, a+b+c);


algsubs(3*x+x^2=pol,udtryk);


Procedurerne indets, subsindets og evalindets


Navnet indets er en forkortelse af indeterminates. Proceduren virker således:


u:=x^2*sin(x)+y+Pi*exp(x*y)+cos(2);


indets( u);


For at forstå dette output læser vi i hjælpen til indets, at indets betragter sit argument som et 
rationalt funktionsudtryk (dvs. et udtryk dannet udelukkende ved brug af operationerne +, -, *, /  
på givne symboler). Udtryk som sin(x) og exp(xy) bliver dermed også opfattet som 
'indeterminates', men konstanten cos(2) gør ikke.
Vi kan blandt disse udvælge dem, der har type name:


select(type,%,name);


Eller på én gang:

select(type,indets(u),name);


Den følgende version af indets gør næsten det samme, men bemærk at  kommer med:


indets( u, name);


Vi kan fjerne konstanter som  eller Eulers konstant  ved brug af remove:


remove(type,%,constant);


Ved brug af den sammensatte type And(name, Not(constant)) i indets kan vi sikre, at konstanter 
ikke medtages:


indets( u, And(name,Not(constant)));


Da man kan bede om de dele af et givet udtryk, der er af en hvilken som helst given type, skal 
ordet 'indeterminates' ikke tages for bogstaveligt:


indets(u, integer);


indets(u, function);


indets(u, `^`);


indets(u, `*`);


Beslægtet med indets er subsindets, der på et udtryks dele af en bestemt type (fundet ved hjælp af
indets) udfører en ønsket transformation ("substitution"). Ligesom subs har en fætter, der hedder
eval, så har subsindets en fætter med navn evalindets.
I første eksempel erstattes alle forekomster af underoperander af typen `^` med operanden selv 
plus 7:


u;


subsindets(u,`^`,s->s+7);


Den følgende procedure MedStort omdanner et uevalueret funktionsudtryk funk(x) til et 
tilsvarende skrevet med stort begyndelsesbogstav:


MedStort:=(s::function)->convert(StringTools[Capitalize](op







Kapitel 14: Tabeller og Arrays, Manipulation af Datastrukturer


Kapitel 14 side 26


(0,s)),name)(op(s));


MedStort(sin(x));


MedStort(funk(x,y,z));


Resultatet af følgende er et inaktivt integral, altså et Int-integral:


MedStort('int'(sin(x),x=0..Pi));


subsindets(u,function,MedStort);


Printet fra følgende input kunne bruges i det konkurrerende program Mathematica, der skriver alle
funktionsnavne med stort og bruger kantede parenteser ved funktionsanvendelse:


lprint(subsindets(%,function,s->op(0,s)[op(s)]));


x^2*Sin[x]+y+Pi*Exp[x*y]+Cos[2]


Hacking (addressof, disassemble, pointto, dismantle, m.m.)


Lad der nu være givet udtrykket


udtryk:=2*z+4*x*y^2+y+5;


Antag, at vi (af uransagelige grunde) vil fordoble alle heltallene i udtrykket. Vi forsøger os med
subsindets: 


subsindets(udtryk,integer,w->2*w);


Det var ikke lige det, der var hensigten. For at forstå resultatet, finder vi først 


indets(udtryk,integer);


Vi bemærker, at indets ikke anfører tallet 1, som jo heller ikke optræder eksplicit i udtryk. De 


substitutioner, der derfor bliver udført i udtryk (og i den rækkefølge) er :


subs(2=4,4=8,5=10,udtryk);







Kapitel 14: Tabeller og Arrays, Manipulation af Datastrukturer


Kapitel 14 side 27


Kommandoen subsindets[flat] udfører substitutionerne samtidigt:


subsindets[flat](udtryk,integer,w->2*w);


Kommandoen subsindets[flat] virker altså hér på følgende måde, hvor det er tuborgklammerne, 
der bør bemærkes:


subs({2=4,4=8,5=10},udtryk);


Betragt nu hvad der sker, når vi tilføjer substitutionen 1 = a:


subs({1=a,2=4,4=8,5=10},udtryk);


På en eller anden måde optræder der altså ét-taller i Maples interne repræsentation af udtrykket, 
nemlig som koefficienterne til y og 5 og som eksponenten til x i produktet x*y^2. Dette er også lige
præcis hvad der gør. Her kommer lidt hacking:
  


udtryk;


addressof(udtryk);
174932180


d:=disassemble(%);


Vi får et antal positive hele tal. Dette antal er strukturens længde.
 


kernelopts(dagtag=d[1]);
SUM


seq(pointto(d[k]),k=2..nops([d]));


Dette betyder kort fortalt, at Maple internt repræsenterer vores udtryk som en struktur af formen 
SUM( z, 2, x*y^2, 4, y, 1, 5, 1). 


Produktet x*y^2 kan selv opløses:


pointto(d[4]);


g:=disassemble(d[4]);


kernelopts(dagtag=g[1]);

PROD


seq(pointto(g[k]),k=2..nops([g]));


Produktet x*y^2 repræsenteres altså internt i Maple som PROD( x, 1, y, 2). Denne struktur har 
længden 5, da g er en følge med 5 elementer. 


Den interne struktur fås lidt hurtigere ved brug af proceduren dismantle:


dismantle(udtryk);


SUM(9)


   NAME(4): z


   INTPOS(2): 2


   PROD(5)


      NAME(4): x


      INTPOS(2): 1


      NAME(4): y


      INTPOS(2): 2


   INTPOS(2): 4


   NAME(4): y


   INTPOS(2): 1


   INTPOS(2): 5


   INTPOS(2): 1


Tallene i parenteser er længderne af de forskellige strukturer.
I den interne repræsentation af udtryk optræder altså i alt 3 éttaller, men indets ser ingen. Med 
denne forklaring skulle også følgende i første omgang overraskende resultater kunne forstås:
 


nytudtryk:=sin(1)+udtryk;


subsindets(nytudtryk,integer,w->2*w);


subsindets(udtryk,integer,w->2*w);


subsindets[flat](nytudtryk,integer,w->2*w);


subsindets[flat](udtryk,integer,w->2*w);


Vi hjælper dog lidt på vej:

indets(nytudtryk,integer);


Nu optræder éttallet eksplicit i vores nye udtryk. Derfor ses det af indets/integer. Men hvorfor 
blev 5-tallet til 10 ved subsindets, men til 20 ved subsindets[flat]? Vi må forestille os følgende 
kæde af ændringer ved subsindets(nytudtryk,integer,w->2*w); De successive substitutioner er 1 =
2, 2= 4,  4 = 8, 5 = 10:


nytudtryk=SUM(sin(1),1,z,2,PROD(x,1,y,2),4,y,1,5,1)


:#Udgangspunktet


SUM(sin(2),2,z,2,PROD(x,2,y,2),4,y,2,5,2): # Substitutionen 


1 = 2


SUM(sin(2),2,z,2,PROD(x,2,y,2),4,y,2,10,1): # Automatisk 


simplifikation på konstantleddet!


SUM(sin(4),4,z,4,PROD(x,4,y,4),4,y,4,10,1): # Substitutionen


2 = 4


SUM(sin(8),8,z,8,PROD(x,8,y,8),8,y,8,10,1): # Substitutionen


4 = 8


Den sidste substitution 5 = 10 ændrer intet, da 5 ikke længere forekommer i den interne 
repræsentation. Ved brug af subsindets[flat](nytudtryk,integer,w->2*w); sker følgende:


nytudtryk=SUM(sin(1),1,z,2,PROD(x,1,y,2),4,y,1,5,1)


:#Udgangspunktet


SUM(sin(2),2,z,4,PROD(x,2,y,4),8,y,2,10,2): # 


Substitutionerne {1 = 2, 2 =4, 4 = 8, 5 = 10}


SUM(sin(2),2,z,4,PROD(x,2,y,4),8,y,2,20,1): # Automatisk 


simplifikation på konstantleddet!

15. Programmering i Maple


Løkker og betingede sætninger


Antag, at vi ønsker  udregnet for hvert andet n startende med 3 og sluttende, når n er blevet 8 


eller højere. Så kan vi bruge en løkke:   


for n from 3 by 2 to 8 do n^2 end do;


9


25


49


Bemærk, at end do kan erstattes af od.


Hvis  by k  udelades (her by 2) underforstås by 1. Hvis from p  udelades underforstås from 1:


for n to 3 do n^2 end do; 
1


4


9


En meget kort version, hvori kun optræder det antal gange løkken skal køre:


to 3 do "Goddag" end do;
"Goddag"


"Goddag"


"Goddag"


End ikke det antal gange løkken skal køre behøver optræde. Løkken vil køre evigt, med mindre 


der indføres et stopkriterium som i følgende eksempel, hvori udtrykket  differentieres flere 


gange, indtil det forsvinder. 
Bemærk foruden den betingede sætning if ... then ... end if også forekomsten af ordet break: 


u:=x^5:


do u:=diff(u,x); if u=0 then break end if end do; 

Ovenstående kunne (lidt mere elegant) være klaret ved brug af while:


u:=x^5:


while u<>0 do u:=diff(u,x) end do;


Ovenfor så vi en anvendelse af konstruktionen  if ... then ... end if  inde i en løkke. 
Konstruktionen kan udbygges til if ... then ... else ... end if.
I stedet for end if kan bruges fi. 
En gaffelfunktion (bedre kendt som en Tuborg-funktion) kan defineres ved brug heraf :


f:=  t -> if t >=0 then sqrt(t) else 0 end if;


Vi afprøver definitionen:


f(7),f(-7);


Hvis Maple ikke kan afgøre om et input til f er ikke-negativt, så kommer en fejlmelding:


f(s);


Error, (in f) cannot determine if this expression is true or 


false: 0 <= s


Dette problem kunne klares ved at lade f(t) returnere uevalueret, hvis ikke t er af typen numeric 
(altså et konkret reelt tal):


?type,numeric


f:=  t -> if not type(t,numeric) then 'f(t)' elif t >=0 then


sqrt(t) else 0 end if;


Apostrofferne omkring f(t) er meget vigtige. Uden dem vil f(t) ende i en uendelig rekursion, når t 
ikke er af typen numeric.


f(7),f(-7),f(s);


Bemærk sammentrækningen elif, der står for else if.


Definitionen af f er dog stadig ikke perfekt:


f(Pi);

Tallet  regnes nemlig ikke til typen numeric. Vi kunne så erstatte denne type med typen


realcons, 


f:=  t -> if not type(t,realcons) then 'f(t)' elif t >=0 


then sqrt(t) else 0 end if;


f(Pi);


Error, (in f) cannot determine if this expression is true or 


false: 0 <= Pi


Fejlmeldingen siger, hvad problemet er


if Pi>=0 then "Pi er ikke negativ" end if;


Error, cannot determine if this expression is true or false: 0 <=


Pi


evalb er ikke i stand til at afgøre om  er større eller mindre end 0:


evalb(Pi>=0);


Vi prøver med is:


is(Pi>=0);
true


Ny forbedring af f:


f:=  t -> if not type(t,realcons) then 'f(t)' elif is(t >=0)


then sqrt(t) else 0 end if;


f(7),f(-7),f(Pi),f(s);


Det er imidlertid ulige meget lettere at bruge piecewise:


fp:=t->piecewise(t<0,0,sqrt(t));


fp(7),fp(-7),fp(Pi),fp(s);


if ..then .. (elif...then) ..else.. end if  bruges meget ved egentlige proceduredefinitioner, som vi nu 
vender os mod. 


Procedurer


Vi har set adskillige eksempler på pileprocedurer, der jo benyttes ved definition af en funktion. 
Pileprocedurer benyttes, når proceduren er simpel. 

Ved mere indviklede procedurer, specielt procedurer, der kræver brug af lokale variable, må 
gennemføres en egentlig proceduredefinition. Der kan selvfølgelig også defineres procedurer, der 
ikke direkte har noget med funktioner at gøre, såsom mange af de procedurer, der følger med 
Mapleprogrammet, eksempelvis factor, expand, solve og plot.


En procedure er ligesom en funktion en 'maskine', der tager et input, gør noget ved dette, og derpå
leverer et eller andet output. 
Definitionen af en procedure begynder med proc(.....) og slutter med end proc. Imellem disse står
procedurekroppen, der er en opskrift på de beregninger, der skal udføres. 
Ved definitionen af proceduren beregnes intet, dog kan heller ikke her automatiske 
simplifikationer undgås. Det er først når proceduren bliver kaldt (bliver brugt), at beregningerne 
bliver foretaget. 


Syv simple eksempler på procedurer


Vi giver først en række simple eksempler på procedurer. Disse gør intet krav på nyttighed.


I første eksempel skal proceduren ved input x levere  som output, uanset hvad x 


måtte være pladsholder for:


p1:=proc(x) 7*x^2-x+8 end proc;


Bogstavet x i procedurehovedet kaldes den formelle parameter i proceduren. Ved kaldet p1(5); 
erstattes den formelle parameter x overalt i procedurekroppen med den aktuelle 
parameterværdi, hér 5:
  


p1(5);
178


Lad os sige, at vi laver tilordningen


p1a:=p1;


Så er det allerførste, der sker med kommandoen p1a( int( 6*t^2, t=0..1) ); evaluering af p1a til
p1. Derefter vil der ske fuld evaluering af argumentet, altså udregning af integralet, nemlig til 
2. Herefter erstattes den formelle parameter x i proceduredefinitionen for p1 overalt med 2, så 
følgende 4 procedurekald giver samme resultat:


p1a( int(6*t^2,t=0..1) ), p1( int(6*t^2,t=0..1) ), p1a(2),


p1(2) ;


Der er undtagelser til denne generelle regel om fuld evaluering af argumenterne før de gives til
proceduren: piecewise, seq, add og mul er eksempler på procedurer med specielle 
evalueringsregler.


Ofte gør man i en procedure brug af lokale variable, dvs. hjælpevariable til opbevaring af 
mellemresultater i proceduren. I næste eksempel skal proceduren som input kunne tage to 
tekststrenge. Som output skal den svare true, hvis den første tekststreng er kortere end den 
anden, ellers skal den svare false. Vi udnytter her Maple-proceduren evalb (=evaluate 
boolean), der svarer true, hvis den får et sandt udsagn og false, hvis den får et falsk:  







Kapitel 15: Programmering i Maple


Kapitel 15 side 5


evalb(3<4);
true


Desuden udnytter vi Length fra pakken StringTools:


StringTools:-Length("Dette er en lang og streng vinter");
33


p2:=proc(a,b) local s,t; 


s:=StringTools:-Length(a); t:=StringTools:-Length(b);


evalb(s<t)


end proc;


Der må ikke være semikolon lige efter proc(....), dvs. i dette tilfælde lige før local. 
Procedurekroppen indeholder her 3 kommandoer. Resultatet af sidste kommando er outputtet 
fra proceduren. Efter sidste kommando (lige før end proc) behøver man ikke sætte semikolon, 
men man må gerne.
Vi foretager et par kald til proceduren:


p2("Pølse","sennep");
true


p2("Frederikke","lkjhg");
false


Bemærk, at værdien af de lokale variable s og t ingen effekt har på de globale variable s og t:


s,t;


Når en procedure bliver kaldt, vil argumenterne (de aktuelle parameterværdier) som nævnt
normalt først blive evalueret helt. Først derefter bliver de formelle parametre i proceduren (hér
a  og b) erstattet med de aktuelle værdier:
 


w:="ymer":


p2(w,"ti");
false


Det er ofte en fordel at lade proceduren undersøge, om inputtet har den rigtige type. I 
proceduren p2 ovenfor er det jo (implicit) forudsat, at de to argumenter begge er strenge. 
Internt i vores procedure vil der kunne opstå en fejl, der udløser en fejlmelding, der ikke 
nødvendigvis er lige så sigende, som tilfældet er i vores simple eksempel:


p2(7,8);


Error, (in p2) invalid input: StringTools:-Length expects its 

1st argument, s, to be of type string, but received 7


Maple-typen er string. 
I p2 ovenfor blev brugt dt lange navn StringTools:-Length. Det er nødvendigt i en procedure. 
Kommandoen with(StringTools); virker ikke.
Et alternativ til lange navne er uses StringTools; som vi har gjort i den følgende version:


p3:=proc(a::string,b::string) local s,t; 


uses StringTools;


s:=Length(a); t:=Length(b);


evalb(s<t)


end proc;


Vi ser, at Maple selv allerede ved proceduredefinitionen har indsat de lange navne. Skal flere 
pakker bruges adskilles disse ved kommaer.
uses pakke; virker som konstruktionen use pakke in .... end use;


Bemærk indholdet i procedurehovedet. Efter hver af de formelle parametre står to koloner 
fulgt af en typebetegnelse, hér string. 


p3("Pølse","sennep");
true


p3(89,"mmmm");


Error, invalid input: p3 expects its 1st argument, a, to be of 


type string, but received 89


I Maple er defineret en lang række typer. Se hjælpen for type:


?type


Eksempler på typer er equation, float, integer, list, name, numeric, posint, procedure, range, 
realcons, series, set, symbol, `+`, `*`, `^ .̀  


Et typecheck kunne også laves i den første procedure p1. Lad os sige, at vi vil lade proceduren 
acceptere dels alle komplekse tal, dels variabelnavne (som f.eks. x eller t). De tilsvarende 
typenavne er complex og name. Proceduren skal altså kontrollere, om input er af en af de to 
typer complex eller name. Denne sammensatte type skrives i Maple som Or( complex, name), 
men kan også skrives {complex, name}. Vi foretrækker den første version. Kontrollen gøres 
derfor således:


p4:=proc(x::Or(complex,name)) 7*x^2-x+8 end proc;


p4("tre");


Error, invalid input: p4 expects its 1st argument, x, to be of 


type Or(complex, name), but received tre

p4(5);
178


p4(3+I);


p4(t);


Proceduren vil med typekontrollen Or(complex,name) ikke acceptere et kald som eksempelvis 
p4(a + b). En mere egnet typekontrol i dette tilfælde er nok snarere x::algebraic. Et udtryk er af
type algebraic, hvis det tilhører en af følgende typer: integer, fraction, float, symbol, indexed, 
`+`,  `*`,  `^`,   series, function, `!`,  `.`,  name,  uneval eller complex.
  


Parentetisk kan det bemærkes, at proceduren p2 (eller versionen med typecheck p3) kan 
bruges, hvis man ønsker at sortere en liste af ord efter deres længde. Her skal bruges Maple-
proceduren sort, der som sit andet argument tillader et kriterium for sorteringen f.eks. i form af
en boolesk procedure, dvs. en procedure, der svarer true eller false. Vi bruger her vores 
procedure p3:


ordliste:=["Ole","svinekotelet","rødkål","is","rødbeder","


   ","å"]:


sort(ordliste,p3);


Hvis sort kaldes med kun ét argument og dette er en liste af strenge, så bruges leksikal 
sortering. For flere detaljer se hjælpen for sort.


En procedure behøver slet ikke at have formelle parametre. Her er et eksempel, hvor der heller
ikke kræves et input:


p5:=proc() "Goddag, hvordan går det?" end proc;


Et kald til proceduren p5 skal have parenteser. Inden i parenteserne behøver der ikke være 
noget, da proceduren jo ikke bruger evt. argumenter til noget. Bemærk, at procedurenavnet 
alene ikke giver andet end procedurenavnet, hvilket er helt generelt for procedurer i Maple:


p5;
p5


p5();
"Goddag, hvordan går det?"


p5(x,56,æbler);
"Goddag, hvordan går det?"

Hvis der ved en proceduredefinition er anbragt formelle parametre i procedurehovedet, der i et 
aktuelt kald til proceduren faktisk ikke bliver brugt i procedurekroppen, så behøver de ikke 
blive nævnt i dette kald til proceduren:
 


p6:=proc(a::integer,b::integer) if a<10 then 3*a else 5*b 


end if end proc;


p6(8);
24


p6(13);


Error, invalid input: p6 uses a 2nd argument, b (of type 


integer), which is missing


p6(13,6);
30


Maple-proceduren time er et eksempel på en procedure, der ikke kræver argumenter. Den giver
den forbrugte CPU-tid:


time();
0.046


Proceduren kan bruges til måling af beregningstider:
 


t0:=time():


evalf[10001](Pi):


time()-t0;
0.032


Men får time et argument, så finder proceduren den tid, der forbruges til beregning af 
argumentet. Vi prøver udregningen af  med 10000 betydende cifre:


time(evalf[10000](exp(1)));
0.015


At time virker på denne måde betyder, at den ikke følger den generelle regel, der siger fuld 
evaluering af argumenterne før proceduren i øvrigt går igang.


Hvis vi nu prøver time igen, så overraskes vi måske:


time(evalf[10000](exp(1)));
0.


Det tog (næsten) ingen tid. Grunden er, at Maple-proceduren evalf husker de udregnede 
resultater. Den har i proceduredefinitionen - umiddelbart efter evt. erklæringer om lokale og 
globale variable - erklæringen option remember; Alt hvad der bliver leveret af resultater fra







Kapitel 15: Programmering i Maple


Kapitel 15 side 9


evalf bliver gemt i dens husketabel. 
Husketabellen for en procedure p fås ved kommandoen op(4, eval(p)); mens procedurens
options fås ved op(3, eval(p));
Vil man se husketabellen for evalf kan man altså prøve kommandoen op(4,eval(evalf));
Som eksempel på en brugerdefineret procedure med option remember omdefinerer vi 
proceduren p4 til en ny procedure p7 med option remember og samtidigt ændrer vi 
typekontrollen til x::algebraic:


p7:=proc(x::algebraic) option remember; 7*x^2-x+8 end 


proc;


Procedurens options fås som nævnt således:


op(3,eval(p7));
remember


Før et kald til proceduren er foretaget, er husketabellen endnu ikke oprettet (output er den 
tomme følge, NULL):


op(4, eval(p7));


p7(54);
20366


Nu er tabellen oprettet og der er én indgang i tabellen:


op(4, eval(p7));


p7(a+b);


Nu er der to:


op(4, eval(p7));


Husketabellen kan ændres af brugeren direkte ved tilordning. I eksemplet nedenfor indføjes 


oplysningen, at p7(10) skal være 17 (der ganske vist ikke passer med formlen , 


men det er måske netop pointen):


p7(10):=17;


p7(10);
17


op(4, eval(p7));

Bemærk, at dette viser, at Maple kigger i husketabellen lige efter evaluering af argumenterne, 
men før noget som helst andet foretages. Finder Maple det, den skal bruge, kommer resultatet 
ud direkte, uden at proceduredefinitionen overhovedet bliver brugt - ej heller typekontrollen i 
procedurehovedet bliver foretaget. Meningen med option remember er jo også at minimere 
tidsforbruget.
En procedure uden option remember vil få en husketabel, hvis en direkte tilordning gøres som i
eksemplet ovenfor. 


Husketabellen kan slettes af proceduren forget:


forget(p7);


op(4, eval(p7));


Husketabellen slettes også ved en ny eksekvering af proceduredefinitionen.


Der er andre mulige options, se hjælpen til option.  


De følgende nummererede eksempler er mere komplicerede, men indeholder også en række nye 
begreber.


Eksempel 1 (maks)


Den endelige procedure skal hedde maks og skal finde det største tal i en følge af reelle tal. 
Således skal maks( 1 , Pi , -6 , sqrt(10) , 2, 3.2 ); returnere 3.2.  
I første omgang vil vi forlange inputtet givet som en liste af tal, altså f.eks. således  [1 , Pi , -6 ,
sqrt(10) , 2, 3.2 ]. Vores første forsøg hedder maks1: 


maks1 := proc(data) local n,i,x; 


            n := nops(data);            


            x := data[1];            


            for i from 2 to n do 


              if is(x < data[i]) then x := data[i] end if 


            end do;


            x


         end proc;


I proceduredefinitionens hoved befinder sig én formel parameter, som vi har kaldt data. Når 







Kapitel 15: Programmering i Maple


Kapitel 15 side 11


proceduren bliver kaldt (altså brugt som i maks( [1 , Pi , -6 , sqrt(10) , 2, 3.2] ); ), så vil data 
overalt i proceduren blive erstattet af den aktuelle parameterværdi, nemlig listen [1 , Pi , -6 , 
sqrt(10) , 2, 3.2]. Efter procedurehovedet følger en erklæring om, at tre variable skal anses for 
lokale, nemlig n, i og x. De bliver tillagt værdier indenfor proceduren, men udenfor vil disse 
værdier være ukendte. 
Ved kommandoen n:=nops(data); bliver n sat til antallet af operander i data, dvs. antal 
elementer i listen data, i det aktuelle tilfælde altså 6. 
Ved kommandoen x:=data[1]; bliver x sat til det første tal i listen data, i det aktuelle tilfælde 1.


Kommandoen is( x < data[i] ); returnerer enten true, false eller FAIL. Vi kunne have udeladt
is i den betingede sætning således: if  x < data[i] then x := data[i] end if; Vi har medtaget is, 


da Maple ellers ikke kan afgøre om f.eks. . 


Output fra en procedure er kun resultatet fra den kommando, der står umiddelbart før end 
proc med mindre proceduren indeholder en speciel return-kommando (herom senere).
Vi prøver om maks1 virker:


L := [1 , Pi , -6 , sqrt(10) , 2, 3.2];


maks1(L) ;
3.2


Da proceduren blev skrevet, gik programmøren ud fra, at input skulle være en liste. Det kan 
derfor ikke undre, at der kan ske underlige ting, hvis man glemmer det:


maks1(3,4);
3


Ved en typekontrol i procedurehovedet kan vi sikre, at proceduren gør vrøvl, hvis den bliver 
kaldt med andet end en liste. Vores anden version maks2 har en sådan typekontrol. En 
typekontrol i procedurehovedet har altid formen  variabelnavn::typenavn, i vores tilfælde skal 
vi altså skrive data::list(algebraic).


maks2 := proc(data::list(algebraic)) local n,i,x; 


            n := nops(data);            


            x := data[1];            


            for i from 2 to n do  


             if is(x<data[i]) then x := data[i] end if 


            end do;


            x


         end proc:


Vi afprøver fejlmeldingen:


maks2(3,4);


Error, invalid input: maks2 expects its 1st argument, data, to 


be of type list(algebraic), but received 3

Strengen "abc" er ikke af typen algebraic:


maks2([3,"abc"]);


Error, invalid input: maks2 expects its 1st argument, data, to 


be of type list(algebraic), but received [3, "abc"]


I denne Maple-session er navnet a ikke blevet tilordnet nogen værdi. Vi undersøger hvordan 
vores procedure reagerer på et input, der også indeholder navnet a:
 


maks2([a,3,4]);
a


maks2([3,4,a]);
4


is-kommandoen svarer false på spørgsmålet a < 3. Dette skyldes, at is( påstand) ifølge hjælpen 
skal svare false, hvis der findes tilfælde, hvor den pågældende påstand er falsk:
 


is(a<3);
false


Vi laver proceduren om til en tredie version maks3, der ved kaldet maks3( [a, 3, 4] ); 
returnerer maks3( [a, 4] );
Vi får brug for at kunne udvælge elementer af en vis type fra en liste. Hertil kunne bruges
select. Vi ønsker samtidigt at gemme de fravalgte elementer, hvortil remove kunne bruges. Vi 
vælger i stedet at bruge proceduren selectremove, der laver begge ting på én gang. Bemærk, at 
output fra selectremove består af:  det udvalgte, det fravalgte  altså af en følge af længden 2:


selectremove(type,[a,3,4,b,c,Pi],realcons);


Vi benytter lejligheden til også at indsætte kommentarer i proceduren.


maks3 := proc(data::list(algebraic)) local R,B,n,m,i,x;   


   


# Vi skiller reelle tal fra resten (typisk variabelnavne):


R,B:=selectremove(type,data,realcons);


# Antal tal og ikke-tal i input


n := nops(R); m:=nops(B);


# Hvis der kun er ikke-tal i input, så returneres 


uevalueret.


# Ellers behandles tallene i R som i de tidligere 


versioner.


if n=0 then return 'maks3'(data) else


            x := R[1];            


            for i from 2 to n do  


             if is(x<R[i]) then x := R[i] end if 

            end do


end if;


# Var der ikke-tal i input vil output være delvist 


uevalueret: 


if m=0 then x else 'maks3'([x,op(B)]) end if


end proc:


Vi afprøver proceduren:


maks3([-6,3,4]);
4


maks3([a,3,4]);


maks3([a,7,Pi,b,1.5]);


Maple har sin egen max. Den virker dog på en følge - ikke en liste - så vi erstatter L med sine 
operander:


op(L);


max( op(L) ) ; 
3.2


Vores procedure skal nu omskrives, så den (i det ydre) virker som max. Vores fjerde og sidste 
version maks skal ikke have formelle parametre, for den skal jo kunne kaldes med et vilkårligt
antal argumenter. Vi refererer derfor i proceduren til _passed (tidligere kaldet args), der er 
følgen af argumenter, som proceduren bliver kaldt med (når den bliver kaldt). Antal 
argumenter i procedurekaldet hedder _npassed (tidligere nargs), men den kommando får vi 
ikke brug for i dette eksempel. 
Da typekontrollen af input ikke kan foretages i procedurehovedet, er der nu indsat en kontrol i 
procedurekroppen.
For ikke at lave for meget om i forhold til maks3 indfører vi data som en lokal variabel. Den 
skal indeholde listen af argumenter. 
I maks3-proceduren refereres inde i procedurekroppen 2 gange til 'maks3'. Det er praktisk at 
erstatte sådanne referencer med Maple-ordet 'procname'. Ved ændringer af procedurenavnet 
behøver man da ikke at ændre navnet i procedurekroppen, idet procname af Maple ved et 
procedurekald bliver erstattet af det navn proceduren har fået.


maks := proc() local data,R,B,n,m,i,x;


#Vi gemmer argumenterne i en liste kaldet data:


data:=[_passed];


#Typekontrol af input:


if not type(data,list(algebraic)) then error "Input skal 


have typen algebraic" end if;       

# Vi skiller reelle tal fra resten:


R,B:=selectremove(type,data,realcons);


# Antal tal og ikke-tal i input


n := nops(R); m:=nops(B);


# Hvis der kun er ikke-tal i input, så returneres 


uevalueret.


# Ellers behandles tallene i R som i de tidligere 


versioner.


if n=0 then return 'procname'(args) else


            x := R[1];            


            for i from 2 to n do  


             if is(x<R[i]) then x := R[i] end if 


            end do


end if;


# Var der ikke-tal i input vil output være delvist 


uevalueret: 


if m=0 then x else 'procname'(x,op(B)) end if


end proc:


Vi afprøver proceduren:


L;


maks(op(L));
3.2


maks(a,3,4);


maks(a,7,Pi,b,1.5);


Afprøvning af typekontrollen:


maks("abc",8,9);


Error, (in maks) Input skal have typen algebraic


Det er generelt en god idé også at afprøve sin procedure på ekstreme tilfælde: Hvad gør vores 
procedure, hvis den intet input får?


maks();


En fornuftig reaktion. Om den er ønskelig eller ej, kan diskuteres. Maple's indbyggede max 
giver:


max();

Eksempel 2 (ekstremum)


Vi vil bestemme mindste- og størsteværdien for en funktion af én variabel på et interval ved at 
bede Maple om at tegne grafen på dette interval. Vi vil så bede Maple om at finde de 
plottepunkter, der har mindst og størst y-koordinat.
Når Maple skal tegne grafen for en funktion, finder den et antal punkter stort nok til at den kan
tegne et rimeligt billede. 
Vi tager et eksempel, hvor vi for overskuelighedens skyld tvinger Maple til kun at bruge 5 
punkter. Dette sikres ved de valgfri argumenter numpoints = 5 og adaptive = false:
 


p:=plot(cos(t),t=0..2*Pi,numpoints=5,adaptive=false);


lprint(p);


PLOT(CURVES([[0., 1.], [1.64346670229216496, 


-0.726064305421969676e-1], [3.07343829238787292, 


-.997678390392996884], [4.68159090151039514, 


-0.307932103310345658e-1], [6.28318529460999998, 


.99999999999999988]], COLOUR(RGB, 1.00000000, 0., 0.)), 


AXESLABELS(t, ""), VIEW(0. .. 6.283185308, DEFAULT))


Vi skal bruge listen af punkter inden i CURVES. Denne liste af punkter er en liste af lister (en
listlist). Den kan lettest fiskes ud således:


indets(p,listlist);


Bemærk, at output fra indets altid er en mængde. Vi fjerner mængdeklammerne ved at bede 
om operanderne (der er kun én, nemlig listen af lister):


L:=op(%);


Vi sorterer nu listen efter voksende y-koordinat. y-koordinaten er jo andenkoordinaten, derfor 
2-tallet i det følgende:


SL:=sort(L, (s,t)->evalb(s[2]<t[2]) );

Punktet med mindste y-koordinat er nu det første punkt i SL, og punktet med størst y-koordinat
er det sidste:


SL[1],SL[-1];


Vi vil nu skrive ovenstående ind i en procedure, der skal hedde ekstremum. Den skal som 
input tage et funktionsudtryk og en intervalangivelse på formen variabelnavn = tal..tal, f.eks.  
x = 3..5. 
Proceduren ekstremum kan derfor se således ud:


ekstremum:=proc(fx::algebraic,itvl::name=realcons..


realcons)


description "Finder globalt minimumspunkt og 


maksimumspunkt (x,y) for y=f(x) på et givet interval";


local p,L,SL;


p:=plot(_passed);


L:=op(indets(p,listlist));


SL:=sort(L,(s,t)->evalb(s[2]<t[2]));


SL[1],SL[-1]


end proc:


Proceduren udnytter ikke de formelle parametre fx og itvl. Disse er dog taget med for at få 
fornuftige fejlmeldinger, hvis input er af gal type. En procedure behøver ikke at have formelle 
parametre. I vores procedure kunne vi blot have startet således: ekstremum:=proc() local .... 
osv., men så ville evt. fejlmeldinger komme fra plot. 
I stedet for at referere til de formelle parametre fx og itvl i linien p:=plot(_passed); bruger vi
_passed, der - når proceduren bliver kaldt - bliver erstattet af følgen af de aktuelle parametre. 
Derved tillader vi, at ekstra argumenter udover de to første automatisk gives videre til 
plotteproceduren. Eksempelvis kan det være en god idé at forøge det antal punkter, som bruges
af plot, ved tilføjelsen numpoints=200 (eller et andet tal). 


Vi afprøver proceduren:


ekstremum( cos(t), t=0..2*Pi); 


Følgende ville plot ikke beklage sig over, men ekstremum skal:


ekstremum( [sin(t),cos(t)], t=0..2*Pi);


Error, invalid input: ekstremum expects its 1st argument, fx, 


to be of type algebraic, but received [sin(t), cos(t)]


ekstremum(cos(t), t=0..2*Pi,numpoints=500); 


ekstremum(x^2*sin(x),x=0..2*Pi,numpoints=200);

Vi sammenligner med Maple's minimize og maximize:


minimize(x^2*sin(x),x=0..2*Pi,location=true);


evalf(%);


maximize(x^2*sin(x),x=0..2*Pi,location=true);


evalf(%);


Det kunne være en god idé, at plottet af grafen blev vist, hvis man bad om det. Dette kan f.eks.
gøres ved at forsyne proceduren med en userinfo-linie:


ekstremum:=proc(fx::algebraic,itvl::name=realcons..


realcons)


description "Finder globalt minimumspunkt og 


maksimumspunkt (x,y) for y=f(x) på et givet interval";


local p,L,SL,m,M;


p:=plot(_passed);


L:=op(indets(p,listlist));


SL:=sort(L,(s,t)->evalb(s[2]<t[2]));


m:=SL[1];


M:=SL[-1];


userinfo(2,procname,print(plots[display](p,plot([m,M],


style=point,color="Blue",symbolsize=25))));


m,M


end proc:


Brugeren af proceduren kan nu sætte infolevel[ekstremum] til et niveau på 2 eller derover. Så 
vises plottet sammen med de to punkter. Nu giver det også mening at gøre rent grafiske 
tilføjelser, så som farve eller stil.


infolevel[ekstremum]:=2;


ekstremum(x^2*sin(x),x=0..2*Pi,style=point,symbol=circle);

ekstremum:


x


1 2 3 4 5 6
0


Undersøgelse af Maple's egne procedurer


Procedurerne i Maple's bibliotek kan man få at se. Dette gælder eksempelvis for proceduren 
for sinusfunktionen. Denne er nemlig skrevet i Maples eget sprog i modsætning til de 
indbyggede, som man ikke kan få at se. Vi beordrer Maple til at være ordrig, hvad procedurer 
angår (verbose =  unødigt ordrig):


interface(verboseproc=2); 
1


eval(sin); # Det lange resultat er ikke vist (prøv selv at


trykke ENTER)


Et eksempel på en indbygget (builtin) procedure er max-proceduren:


eval(max); 


De indbyggede procedurer er nummererede, men nummeret ændres fra release til release. I 
releases før 11 blev nummeret vist (for max var det 249). Nu vises navnet i stedet. Nummeret 
fås nu således:


op(3,%);


Også procedurerne i Maples pakker kan man få at se. Hér eksempelvis kommandoen, der vil 
vise proceduren PolynomialInterpolation fra CurveFitting-pakken. Da den fylder en del 
viser vi den dog ikke (prøv selv):
 


eval(CurveFitting:-PolynomialInterpolation);


interface(verboseproc=1); # 'Default'-værdien
2







Kapitel 15: Programmering i Maple


Kapitel 15 side 19


    


    


    


Fejlfinding i procedurer


Skriver man sine egne procedurer, kan det ikke undgås, at fejl opstår. Ja, endog Maples egne 
procedurer kan indeholde fejl. 
For at finde sådanne fejl kan det undertiden hjælpe midlertidigt at indsætte en print-kommando
på de steder i proceduren, hvor man har mistanke om, at størrelser ikke beregnes korrekt. 
Maple har imidlertid også forskellige faciliteter til brug ved fejlfinding:


1. Man kan sætte printlevel til et højt niveau.


2. Man kan bruge proceduren debug ( = trace).


3. Man kan bruge Maple's store  'debugger'. 


Vi illustrerer kort de tre metoder på vores procedure maks, som ganske vist ikke indeholder 
fejl. 


Brug af printlevel


Vi sætter printlevel til et passende højt tal:


printlevel:=5:


maks(a,7,Pi,b,1.5,sqrt(2));


{--> enter sqrt, args = 2


<-- exit sqrt (now at top level) = 2^(1/2)}


{--> enter maks, args = a, 7, Pi, b, 1.5, 2^(1/2)


<-- exit maks (now at top level) = maks(7, a, b)}


Vi sætter printlevel til sin default-værdi:


printlevel:=1:


Brug af debug (= trace)


Når printlevel sættes til en værdi, gælder denne værdi for alle procedurer. Ved at bruge 
kommandoen debug (eller dens synonym trace) kan man i realiteten sætte printlevel højt 
for en bestemt procedure.  


debug(maks);
maks


maks(a,7,Pi,b,1.5,sqrt(2)); 


{--> enter maks, args = a, 7, Pi, b, 1.5, 2^(1/2)

<-- exit maks (now at top level) = maks(7, a, b)}


For at komme tilbage til normal status bruger vi undebug (eller untrace):


undebug(maks);
maks


Brug af Maple's store debugger


Maple's store debugger er ret omfattende. Vi nøjes med at gøre opmærksom på eksistensen 
ved at give et eksempel på anvendelsen af to af dens procedurer, showstat og stopat. 
showstat viser proceduren og giver linierne numre, hvilket også kan være særdeles nyttigt 
ved samtale med andre om evt. fejl i proceduren. 


showstat(maks); 


maks := proc()


local data, R, B, n, m, i, x;


   1   data := [args];


   2   if not type(data,list(algebraic)) then


   3     error "Input skal have typen algebraic"


       end if;


   4   R, B := selectremove(type,data,realcons);


   5   n := nops(R);


   6   m := nops(B);


   7   if n = 0 then


   8     return ('procname')(args)


       else


   9     x := R[1];


  10     for i from 2 to n do


  11       if is(x < R[i]) then


  12         x := R[i]


           end if


         end do


       end if;


  13   if m = 0 then

  14     x


       else


  15     ('procname')(x,op(B))


       end if


end proc


stopat sætter et breakpoint, der gør, at når proceduren kaldes, så standser udførelsen ved 
det angivne linienummer. Hvis intet linienummer anføres, standser programmet før første 
linie eksekveres. 
 


stopat(maks,7);


Ved kaldet 


maks(2,3,8); 


Warning,  computation interrupted


kommer nu ( i Standard Worksheet Maple) et vindue op, der tillader en række debug-
kommandoer. Der fører for vidt at beskrive mulighederne i dette vindue.


Vi fjerner nu det satte breakpoint:


unstopat(maks);


Vi har kun lige gjort opmærksom på debuggerens eksistens. For mere information se 
hjælpen:


?debugger


Eksempel 3 (Besselfunktioner. En rekursiv procedure)


Besselfunktionerne  er funktioner, der opfylder differentialligningen


restart;


ligning:=x^2*diff(y(x),x,x)+x*diff(y(x),x)+(x^2-n^2)*y(x)=


0;


som det også fremgår af følgende resultat:


dsolve(ligning);

Maple's betegnelse for  er BesselJ(n,x).


Vi skal her antage, at x > 0. 


Det kan vises, at  opfylder rekursionsformlen . 


Det ved Maple godt:


simplify(BesselJ(n+1,x)+BesselJ(n-1,x));


For  og for   er Besselfunktionerne specielt simple:


 


'BesselJ(1/2,x)'=BesselJ(1/2,x); 'BesselJ(-1/2,x)'=BesselJ


(-1/2,x);


Vi kan udnytte rekursionsformlen til udfra disse to simple udtryk at finde  for alle  


, hvor p er ulige (positiv eller negativ). Ved at løse for  i rekursionsformlen ovenfor 


fås formlen  , hvormed vi kan finde   ud fra  og


. Herefter kan vi finde  , osv.


Vi kan også gå baglæns ved at løse for  i rekursionsformlen. Herved fås formlen  


.


De to formler kan sammenskrives til én formel:  , 


hvor s = 1, når n > 0 og s = -1, når n < 0. 
Dette tal s (fortegnet for n) kan i Maple findes ved brug af signum:
 


'signum(3/2)'=signum(3/2);'signum(-3/2)'=signum(-3/2);


Vores procedure til beregning af Besselfunktionerne  for n  lig med halvdelen af et ulige 

tal ser nu således ud:


BJhalv:=proc(n,x::algebraic) local SQRT,s; option 


remember;


if not type(2*n,odd) then error "Første argument skal være


halvdelen af et ulige helt tal" end if;


s:=signum(n);


SQRT:=sqrt(2/Pi/x);


if n=-1/2 then SQRT*cos(x) elif n=1/2 then SQRT*sin(x) 


else SQRT*collect(2*(n-s)/x*BJhalv(n-s,x)/SQRT-BJhalv(n-2*


s,x)/SQRT,[sin,cos],expand)


end if


end proc:


Det mest bemærkelsesværdige er, at proceduren BJhalv kalder sig selv, dvs. at den inde i 
proceduren refererer til sig selv - endda to gange. Man siger, at proceduren er rekursiv.
En rekursiv procedure kan blive meget dyr i beregningstid, hvis ikke den bliver instrueret om 
at gemme allerede udregnede resultater. Dette kan gøres med option remember. 
Første formelle argument n typekontrolleres ikke i procedurehovedet. Det gøres derimod i 
procedurekroppens første linie.
Den lokale variable SQRT er introduceret for at sikre et smukkere output. Af samme grund er 
brugt collect.


Vi afprøver proceduren:


BJhalv(1/2,x);


BJhalv(5/2,x);


BJhalv(-7/2,x);


Maple er enig med os, når x > 0:


simplify( BesselJ(11/2,x)-BJhalv(11/2,x)) assuming x>0;
0


simplify( BesselJ(-21/2,x)-BJhalv(-21/2,x)) assuming x>0;
0


Kontrol af fejlmeldingen:

BJhalv(4,x);


Error, (in BJhalv) Første argument skal være halvdelen af et 


ulige helt tal


Et typecheck kunne indsættes i procedurehovedet ved brug af satisfies. Kontrollen kunne være
n::satisfies(s->type(2*s,odd)).
Kontrollen i procedurekroppen kan så fjernes, men fejlmeldingen ved forkert input ville ikke 
være så letforståelig.


evalb((7/2)::satisfies(s->type(2*s,odd)));
true


evalb(7::satisfies(s->type(2*s,odd)));
false


p:=proc(n::satisfies(s->type(2*s,odd))) "Ja, halvdelen af 


et ulige tal" end proc;


p(3/2);
"Ja, halvdelen af et ulige tal"


p(3);


Error, invalid input: p expects its 1st argument, n, to be of 


type satisfies(proc (s) options operator, arrow; type(2*s, odd)


end proc), but received 3


Procedure inden i procedure


Betragt følgende procedure, der som output har en anden procedure (en funktion):


G:=proc(a) local b; b:=3; x->a*x+b end proc;


G(4);


Det b, der optræder, er det b, der er lokal for proceduren G, hvilket ses ved et kald til output-
funktionen: 


G(4)(-2);


Grunden til at kommandoen G(4);  ikke besvares af Maple med  er, at intet 


evalueres ved proceduredefinitioner. Der sker kun automatiske simplifikationer. Evaluering 
sker først, når en procedure bliver kaldt. Dvs. b i den indre procedure bliver først evalueret, når
denne procedure bliver kaldt. Men hvorfor er a så blevet erstattet af 4? Når G bliver kaldt med 
argument 4, så bliver den formelle parameter a overalt i proceduren erstattet af 4. 


Hvor den inderste procedure er en funktion, kan den bedste løsning være at bruge unapply:


G2:= proc(a) local b; b:=3; unapply(a*x+b,x) end proc;

G2(4);


G2(4)(-2);


Eksempel 4 (Newtons metode)


Newtons metode er et bedre eksempel på definition af en procedure inden i en anden 
procedure. Proceduren newton skal løse en ligning med én ubekendt ved Newtons metode. 
Som formelle parametre i proceduren bruges ligningen (ligning), rodgættet i form af en ligning
(x = x0) og - som valgfrit argument - antallet af iterationer givet på formen iterationer = n. 
Hvis ikke antallet af iterationer gives, så bruges iterationer = 5 ( = default ). Proceduren giver 
som output listen af iterationer (inklusive startgættet).
Det valgfri argument kapsles inde i mængdeklammer: {iterationer::posint:=5}. Når proceduren
kaldes, kan man eksempelvis skrive iterationer=32. I proceduren optræder iterationer så 
overalt med værdien 32. Hvis det valgfri argument ikke benyttes, optræder iterationer med 
værdien 5 (default-værdien). Valgfri argumenter af denne type kan anbringes i vilkårlig 
rækkefølge, men efter de tvungne argumenter og efter eventuelle positionelle argumenter. Vi 
kunne have valgt at gøre antallet af iterationer positionelt, men stadig valgfrit. Det skulle i så 
fald komme som argument nr. 3 og kunne have formen  iterationer::posint:=5. Ved et kald til 
proceduren skrives da som tredie argument eksempelvis 32, altså blot det ønskede antal 
iterationer.


Mere information om valgfri argumenter mm. fås i hjælpen:


?parameter_classes


Vores færdige procedure ser således ud:


newton:=proc(ligning::equation, start::name=realcons,


{iterationer::posint:=5}) 


           local x,x0,valgfri,X,f,F,xnewt,k;


#Den ubekendte og startgættet gemmes i x og x0


x,x0:=op(start);     


# Kontrol af at ligningen har (og kun har) den ubekendte 


x:


X:=select(type,indets(ligning),name);


if X <>{x} then error "Ligningen skal indeholde netop 


variablen %1, men indeholder %2",x,X end if;


# Ligningen skrives på formen f(x)=0 


f:=unapply((lhs-rhs)(ligning),x);


# Den egentlige Newton-algoritme


F:= x -> x-f(x)/D(f)(x);


xnewt[0]:=x0;


for k to iterationer do 


xnewt[k]:=evalf(F(xnewt[k-1]));

end do;


# Her kommer så output


[seq(xnewt[k], k=0..iterationer)]


end proc;


Vi prøver om proceduren virker efter hensigten.
Først uden angivelse af antal iterationer, hvorfor default-værdien på 5 bruges:


newton( 2*sin(x)=x, x=1.8); 


Dernæst med eksplicit angivelse af antal iterationer:


newton( 2*sin(x)=x, x=-1.7,iterationer=4);


Vi afprøver også fejlmeddelelserne:


newton( 2*sin(x)=y, x=-1.7, iterationer=4);


Error, (in newton) Ligningen skal indeholde netop variablen x, 


men indeholder {x, y}


newton( 2*sin(x)=x, y=-1.7, iterationer=4);


Error, (in newton) Ligningen skal indeholde netop variablen y, 


men indeholder {x}


newton( 2=1,x=7);


Error, (in newton) Ligningen skal indeholde netop variablen x, 


men indeholder {}


Det bemærkes, at vi har brugt den lokale variable xnewt. Den bruges som en tabel. Ligeså 
nærliggende kunne det være at bruge betegnelserne xnewt0, xnewt1, xnewt2, o.s.v.. Vi får da 
brug for navnesammenslutteren. Proceduren ovenfor virker faktisk, hvis xnewt[k] erstattes af
cat(xnewt,k) overalt. Den virker IKKE, hvis linien xnewt[0]:=x0; skrives som xnewt0:=x0;  

kun hvis den skrives cat(xnewt,0):=x0;  Det der sker i første tilfælde er, at xnewt0 af Maple 
opfattes som en lokal variabel, da den optræder på venstre side af et tilordningstegn. Maple 
giver da også ved kørslen en advarsel herom. De andre cat(xnewt,k) forekomster bliver opfattet
som globale variable, inklusive cat(xnewt,0), der optræder i løkken, når k=1. I det andet 
tilfælde optræder kun de globale variable cat(xnewt,k) . De sammensluttede variable vil i 
procedurer altid blive opfattet som globale. Hvis dette ikke ønskes, bør man undgå brugen af 
navnesammenslutteren i procedurer. 
Det samme gælder for brugen af || i stedet for cat :


p:=proc() local a,b,c,d,ab; c:=cat(a,b); d:=a||b; type(c,


`global`),type(d,`global`) end proc:


p();


Evaluering i procedure


Kaldes en procedure, evalueres enhver af de aktuelle parametre, hvormed den er kaldt, på 
sædvanlig måde, dvs. generelt evalueres de til bunds. Derefter erstattes de formelle parametre 
overalt i proceduren med disse aktuelle værdier. Herefter bliver de formelle parametre ikke 
mere evalueret. Lokale variable evalueres kun til den værdi, de fik ved sidste tilordning. 


Betragt følgende procedure, der blot skal udregne :


p:=proc(x) local a,b,c; a:=b+1; b:=c^2; c:=x-Pi; a end 


proc:


p(34);


Først nu evalueres fuldt ud:


%; 


Sammenlign med følgende kommandoer:


x:=34: a:=b+1; b:=c^2; c:=x-Pi; a;


Fuld evaluering af en lokal variabel kan opnås ved brug af eval:


p1:=proc(x) local a,b,c; a:=b+1; b:=c^2; c:=x-Pi; eval(a) 


end proc:


p1(34);

En ændret rækkefølge af kommandoerne i proceduren p ville også have det ønskede resultat:


p2:=proc(x) local a,b,c; c:=x-Pi; b:=c^2; a:=b+1; a end 


proc:


p2(34);


Betragt følgende kommandoer:


restart;


f:=g:


g:=proc(x::uneval) x,eval(x),procname end proc:


a:=7:


f(a);


'f'(a);


%;


g(a);


Det følger af det ovenstående, at det første, der sker ved evalueringen af et procedurekald som 
f(a), er evaluering af procedurenavnet, her til g, hvormed altså f(a) bliver til g(a) (der her er 
lavet, så argumentet ikke evalueres før det sendes til behandling i proceduren).


Eksport af lokal variabel


I procedurer er der en afgørende forskel på lokale og globale variable. Følgende eksempel 
illustrerer dette på dramatisk vis. 


restart;


tugthus:=proc() local x; x end proc;


ud:=tugthus();


type(ud,`local`);


true


ud-x;


Det x, der kommer ud af proceduren, er forskelligt fra det vi skriver udenfor, og prøver vi et 
nyt kald til proceduren tugthus:


ud_igen:=tugthus();

ser vi nu, at de to lokale x'er også er forskellige:


ud-ud_igen;


cat(`Det borende `,ud);
Det borende x


cat(`Det borende `,ud) - cat(`Det borende `,x);
0


Ved navnesammenslutningen mistes de betydninger karaktererne bærer med sig, jvf. en 
tidligere bemærkning om, at navnesammensluttede variable i procedurer ikke opfattes som 
lokale.
Ovenstående er ikke kun en kuriositet. Undslupne lokale variable har vi set brugt af Maple i 
kapitel 5, hvor en lokal variabel ved navn _Z~  blev brugt ved angivelsen af samtlige løsninger


til ligningen 


Moduler


Et modul er som en procedure, men tager intet input og giver intet direkte output. Modulet kan 
have et antal variable, der eksporteres, meget lig eksporterede lokale variable fra procedurer. Dog 
har man direkte adgang til disse eksporterede variable. Desuden kan modulet have lokale og 
globale variable ganske som procedurer. 
Et modul kan skabe et miljø afgrænset fra omverdenen i hvilket beregninger kan foretages og evt. 
eksporteres til omverdenen.
I første eksempel bruges modulet blot til opbevaring af information.


ABC:=module() export a,b,c; 


a:="a er første bogstav i alfabetet";


b:="b er andet bogstav i alfabetet";


c:="c er tredie bogstav i alfabetet";


end module;


Adgang til den eksporterede variable a fås således:


ABC:-a;
"a er første bogstav i alfabetet"


Alternativt kan man bruge følgende konstruktion:


use ABC in


a,b,c;


end use;

Som et andet eksempel tager vi et modul, hvor en af de variable, der eksporteres, er navnet på en 
procedure, nemlig newton-proceduren fra et tidligere eksempel. Desuden eksporteres variablen
syntaks, der vil indeholde en tekststreng.
For at undgå evt. forvirring med den tidligere brug af newton begynder vi med et restart:


restart;


NEWTON:=module() export newton, syntaks;


option package;


#### Her følger proceduren newton


newton:=proc(ligning::equation, start::name=realcons,


{iterationer::posint:=5}) 


local x,x0,valgfri,X,f,F,xnewt,k;


x,x0:=op(start);     


X:=select(type,indets(ligning),name);


if X <>{x} then error "Ligningen skal indeholde netop 


variablen %1, men indeholder %2",x,X end if;


f:=unapply((lhs-rhs)(ligning),x);


F:= x -> x-f(x)/D(f)(x);


xnewt[0]:=x0;


for k to iterationer do 


xnewt[k]:=evalf(F(xnewt[k-1]));


end do;


[seq(xnewt[k], k=0..iterationer)]


end proc;


#### Her følger definitionen af variablen syntaks


syntaks:="newton( ligning, x = startværdi, iterationer=antal


(valgfrit))";


end module;


NEWTON:-syntaks;
"newton( ligning, x = startværdi, iterationer=antal (valgfrit))"


NEWTON:-newton(cos(x)=x/4,x=-Pi/2);


use NEWTON in


newton(cos(x)=x/4,x=-4);


newton(cos(x)=x/4,x=1);


end use;


exports giver de eksporterede variables navne, dog i deres globale versioner, som det ses:

syntaks:=7:


exports(NEWTON);


%;


NEWTON:-syntaks;
"newton( ligning, x = startværdi, iterationer=antal (valgfrit))"


syntaks:='syntaks': # "Nulstiller"


Nye pakker i Maple implementeres som moduler. De i en pakke indeholdte procedurer 
eksporteres. Pakken LinearAlgebra er et modul:
 


type( LinearAlgebra, `module`);
true


Det samme er nu tilfældet med plots-pakken:


type( plots, `module`);
true


Accent grave er nødvendig om module, da Maple ellers tror, at en moduldefinition starter og 
derfor giver en fejlmelding, da denne moduldefinition ikke afsluttes, ja knap kommer igang.
member kan bruges til kontrol af om et navn eksporteres af et modul:


member(Determinant, LinearAlgebra);
true


A:=<1,2|3,4>;


Determinant(A);


LinearAlgebra:-Determinant(A);


Alternativ skrivemåde:


LinearAlgebra[Determinant](A);


Ved kommandoen with(modulnavn); gøres navnene globale:


with(LinearAlgebra):

Determinant(A);


with(NEWTON);


syntaks;
"newton( ligning, x = startværdi, iterationer=antal (valgfrit))"


De eksporterede variable fra et modul har man umiddelbar adgang til. Men et modul kan som sagt
også have lokale variable. Nogle af disse kunne være procedurer. Det er f.eks. tilfældet i Maple-
pakkerne LinearAlgebra, RealDomain og CurveFitting. Man kunne måske have lyst til at kigge 
på en af disse procedurer. 
For at vise hvordan dette gøres bemærker vi først, at et moduls første operand er følgen af 
eksporterede variable:


op(1,eval(CurveFitting));


Modulets anden operand er moduldefinitionen:


op(2,eval(CurveFitting));


Endelig er tredie operand følgen af lokale variable:


op(3,eval(CurveFitting));


Nogle af disse lokale variable er navne på procedurer. Vi undersøger hvilke det er:


select(type,[op(3,eval(CurveFitting))],procedure);

remove(type,[op(3,eval(CurveFitting))],procedure);


Vil vi se definitionen af en af procedurerne, f.eks. den første GetXYData, der er lokal variabel 
nummer 2, kan vi gøre vi således:


showstat(op(3,eval(CurveFitting))[2]); #Output ikke vist


Vil vi i stedet se den fjerde, erstattes [2] med [4].  Bemærk, at showstat( GetXYData);  ikke 
virker (hverken med eller uden with(CurveFitting); ), da GetXYData er en variabel, der er lokal 
til modulet. 
debug kan bruges på procedurer, der er lokale for et modul, ved en noget speciel syntaks. Det 
specielle er dobbeltkolonnet, der jo ellers bruges ved typedeklarationer:


debug(CurveFitting::GetXYData);
GetXYData


CurveFitting:-PolynomialInterpolation([1,2,3],[7,9,13],x);


{--> enter GetXYData, args = [1, 2, 3], [7, 9, 13]


<-- exit GetXYData (now in CurveFitting:-PolynomialInterpolation)


= [1, 2, 3], [7, 9, 13], 3, 3}


undebug(CurveFitting::GetXYData);
GetXYData

16. Egne Biblioteker og Pakker


Gemme egne procedurer


Har man skrevet en procedure, som man ønsker at gemme til senere brug i forskellige 
sammenhænge, kan den selvfølgelig gemmes i et sædvanligt worksheet, men den kan også 
gemmes på den måde, som nu skal beskrives.


I nyhedsgruppen comp.soft-sys.math.maple kunne man læse følgende lille elegante procedure
applysome skrevet af Robert Israel, http://www.math.ubc.ca/~israel. Den skal gøre det muligt at 
anvende en procedure på nogle af operanderne i et udtryk. Der er her foretaget et par uvæsentlige 
småændringer:


applysome:= proc(fn, S::set, expr)


  local yes,no,i,n,f0,a,Sp;


  # Overskydende aktuelle parametre (hvis fn kræver ekstra 


argumenter):


  a:=_rest;  


  n:= nops(expr);


  (*Da operandnumrene kan være negative sørger vi for, 


  at de nu nummereres fra 1 og til n *)


  Sp:=map(`modp`,S,n+1); 


  f0:= op(0,expr);


  yes:= NULL; no:= NULL;


  for i from 1 to n do


    if member(i,Sp) then


      yes:= yes,op(i,expr)


    else no:= no, op(i,expr)


    end if


  end do;


  f0(no,fn(f0(yes),a))


end proc:


Vi afprøver proceduren på summen


u:=x^2+2*x*y+y^2+1/(a+b)+1/(a-b)+1/c;


hvor vi ønsker at skrive de første 3 led som et produkt:


applysome(factor,{1,2,3},u);

Ønsker vi nu at sætte de første 2 led på fælles brøkstreg med nævneren ganget ud, gør vi således:


applysome(normal,{1,2},%,expanded);


I modsætning til den fremgangsmåde, der blev beskrevet i kapitel 14, virker applysome også på 
produkter (i princippet på hvad som helst):


v:=sin(x)*cos(x)*cos(4*x);


Her bruges combine kun på første og sidste faktor i v:


applysome(combine,{1,-1},v);


combine alene vil arbejde på hele udtrykket:


combine(v);


Proceduren applysome synes at virke efter hensigten. Vi vil gemme proceduren ved brug af save. 
Proceduren bliver gemt i internt Maple format, hvis den gives et navn, der ender med .m ellers 
gemmes den som ren tekst. Man skal angive den fulde sti. Anvendes tegnet \ , må det fordobles, 
da det ellers blot opfattes som et linieskift. Husk også gåseøjne (eller accent grave):


save applysome,"c:/temp/applysome.m";  


Vi ønsker at prøve, om det har virket. Så genstart. Indlæs derefter applysome.m ved hjælp af
read -kommandoen:


restart;


read "c:/temp/applysome.m";


u:=23+a+b+c;


Som afprøvning ser vi, om vi kan få fordoblet det første og det sidste led, men ikke de andre:


applysome(x->2*x,{1,-1},u);


Vi beder om at få proceduren at se:


eval(applysome);

Alternativt gemmes her proceduren som ren tekst:


save applysome,"c:/temp/applysome.mpl";  


Indlæsning som før, men nu vises proceduren allerede ved read-kommandoen::


restart;


read "c:/temp/applysome.mpl";


Oprettelse af eget bibliotek


En pakke bør være et modul med option package. De eksporterede variable (normalt navne på 
procedurer i dette tilfælde) er tilgængelige efter kommandoen with(pakkenavn);


Skriver man mange procedurer og pakker, kan man ønske sig sit eget bibliotek. 
I Maple findes en pakke, LibraryTools, der kan bruges til formålet. Denne pakke vil blive omtalt 
ganske kort nedenfor.


Et bibliotek kan imidlertid også oprettes på følgende måde:


1. Opret mappe (katalog) til biblioteket. I det følgende bruges som eksempel mappen mitbib, 
der hér er anbragt på følgende sted:   C:\temp\mitbib


Oprettelsen af mitbib kan evt. foretages fra Maple selv som følger:


mkdir("c:/temp/mitbib");


2. Indret mappen (kataloget) som bibliotek vha. march. Dette kan gøres på én af to måderl: 

Ved første metode dannes en fil kaldet maple.lib (og desuden en fil kaldet maple.ind):


march('create', "c:/temp/mitbib", 100);


100-tallet angiver det antal .m-filer, der oprindeligt gøres plads til.


Ved anden metode dannes en fil med et navn, man selv bestemmer, men det skal have formen 
navn.mla. Her bruges filnavnet minfil.mla:


march('create', "c:/temp/mitbib/minfil.mla", 100);


3. Definér i Maple variablen savelibname:


savelibname:="c:/temp/mitbib";


4. Gem nu proceduren (eller hvad det nu måtte være for et Maple-objekt) i biblioteket. Vi 
gemmer hér proceduren applysome:


savelib('applysome');


Bemærk, at hvis ikke savelibname er defineret, så vil proceduren blive gemt i det bibliotek, der 
ligger først i libname. Herved risikerer man at anbringe proceduren i Maples eget bibliotek. 
Dette er ikke en god idé, idet man risikerer at ødelægge det bibliotek, der følger med Maple. 
Er der flere Maple objekter, der skal gemmes, kan dette gøres på én gang, ved at angive en 
følge bestående af objekternes navne hver omgærdet af apostroffer. Har vi i stedet en pakke 
med navn  PAK (et modul med option package), der eksporterer flere variable, så er 
kommandoen blot savelib( 'PAK');   


Vi tager som eksempel følgende pakke:


NEWTON:=module() export newton, syntaks;


option package;


#### Her følger proceduren newton


newton:=proc(ligning::equation, start::name=realcons,


{iterationer::posint:=5}) 


           local x,x0,valgfri,X,f,F,xnewt,k;


x,x0:=op(start);     


X:=select(type,indets(ligning),name);


if X <>{x} then error "Ligningen skal indeholde netop 


variablen %1, men indeholder %2",x,X end if;


f:=unapply((lhs-rhs)(ligning),x);


F:= x -> x-f(x)/D(f)(x);


xnewt[0]:=x0;


for k to iterationer do 

xnewt[k]:=evalf(F(xnewt[k-1]));


end do;


[seq(xnewt[k], k=0..iterationer)]


end proc;


#### Her følger definitionen af variablen syntaks


syntaks:="newton( ligning, x = startværdi, iterationer=antal


(valgfrit))";


end module;


Vi gemmer NEWTON:


savelib(NEWTON);


5. Det er en god idé at skrive en hjælpeside til sin procedure. Den bør struktureres som Maple's 
egne hjælpesider. Disse er fra og med Release 4 skrevet som Maple worksheets. 


Der følger skabeloner til hjælpesider med Maple-programmet. Se i øvrigt hjælpen:


?helppages


6. Vi åbner nu et nyt worksheet med et restart; 


Før brugen af vor procedure applysome eller vores pakke NEWTON og hjælpen dertil, må vi 
definere libname:


restart;


libname:="c:/temp/mitbib",libname;


Proceduren applysome kan nu benyttes. Pakken NEWTON kan bruges efter kommandoen
with( NEWTON ); 


u:=45+Pi+exp(-1)+gamma;


applysome(evalf,{3,4},u);


with(NEWTON);


syntaks;
"newton( ligning, x = startværdi, iterationer=antal (valgfrit))"

newton(sin(x)=x/2,x=2,iterationer=3);


march kan vise indholdet i biblioteket mitbib. Der kommer også nogle filer af formen -:N.m, hvor
N er et positivt helt tal. Disse bruges internt af Maple.
Er biblioteket i maple.lib bruges


march( 'list', "c:/temp/mitbib/maple.lib" ) ;


Er biblioteket i minfil.mla bruges i stedet:


march( 'list', "c:/temp/mitbib/minfil.mla" ) ;


Her er antallet af filer i det bibliotek, der følger med Maple:


nops(march('list',"c:/Programmer/Maple 13/lib/maple.mla"));
36219


De første 5 filer:


march('list',"c:/Programmer/Maple 13/lib/maple.mla")[1..5];


Man kan evt. lave en fil kaldet maple.ini og anbringe denne i mappen Users i 
Mapleprogrammet. Denne fil skrives i almindelig tekst (i Notesblok). I filen skrives linien
libname:="c:/temp/mitbib", libname; (hvis ellers biblioteket hedder mitbib og ligger i 
C:/temp). Så er man fri for at definere libname hver gang man får brug for sine egne 
hjemmelavede procedurer.


For mere information se følgende udmærkede hjælpeside i Maple:


?repository,management


Brug af pakken LibraryTools


I pakken LibraryTools ligger procedurerne


with(LibraryTools);


der kan bruges til at oprettelse og administration af et bibliotek. Vi nøjes her med at henvise til
hjælpen for pakken og dens procedurer.
Dog et enkelt eksempel:


ShowContents("c:/temp/mitbib");

